
DCF77 alarm clock with LCD
This application of an ATmega16 provides an alarm clock with the following properties:
• Standard four line and 20 column LCD white on blue background with convenient
adaption of the backlight to ambient brightness.
• Three-key adjustment plus fast and convenient potentiometer selection of date,
time and alarm time without complicated and lengthy adjustment procedure.
• Synchronization of date and time with the atomic clock DCF77 in and around Ger
many with any commercially available receiver module or with the high-perfor
mance DCF77 receiver described here. DCF77 input compatible with any polarity,
comprehensive signal duration, parity and data checking for all parts of the DCF77
time information with detailled error messages.
• Exact timing with an xtal allows reliable operation even without DCF77 and by man
ually adjusting date and time. Five different xtal frequencies can be configured with
only small changes to the source code.
• Can be supplied from (rechargeable) batteries or with a wall plugin with fallback op
eration for net voltage failures. Four rechargeable 1.2V batteries are sourced with a
small current to enable background charging of the fallback supply.
• Wake-up with random selection from one out of eight melodies to be played, soft
ware is prepared to allow up to seven additional self-programmed melodies to se
lect from.
• Widely configurable assembler source code, with comprehensive debug code for all
hardware components on board.
• Comprehensive documentation of the hard- and software allows for own extensions
to the freely available source code.

1 Hardware
The hardware consist of the alarm clock and the DCF77 receiver.

1.1 Alarm clock

The alarm clock consists of the controller ATmega16 and some peripherals.
The controller clock is sourced from an xtal of 3.2768 Mhz on the XTAL1 and XTAL2 pins.
See here for other clock options. From this clock source the device controls date and time.
The in-system-programming adapter ISP6 on the pins RESET, MOSI, MISO and SCK allows
to program the chip.
The four-line LCD is attached to port C with the data bus and controlled by two portpins
on port B. The 10k trim potentiometer controls the contrast of the LCD.
The backlight on pins 15 and 16 of the LCD is driven by the operating voltage and PWMswitched by OC2 of the controller.
To pins ADC0, ADC1 und ADC2 are attached:
0. a potentiometer for adjusting input numbers,
1. a photo transistor for measuring ambient light,
2. the field strength voltage from the DCF77 receiver.
The supply voltage for the AD converter on AVCC is filtered with a 22µH coil and a ceramic
capacitor of 100nF. The analog reference voltage on AREF is blocked with a capacitor of
100nF.
The signal output of the DCF77 receiver is connected to INT0. The polarity of the signal is
irrelevant, the software works with both (see the signal analysis section for details of the
decoding process).
The three keys for operation are connected to portpins PB0, PB1 and PB2.
The OC1A output from timer TC1 drives via an electrolytic capacitor the speaker, which
plays melodies.
The power supply is built with a 9V transformer, two power diodes, an electrolytic capaci
tor and a 5V regulator with two Tantal capacitors. The 5V supply is fed via a Schottky
diode to allow to add a combination of four rechargeable batteries as second supply. The
rechargeable batteries are via a resistor of 3k3 loaded with a small current (3 mA when
completely unloaded, 1 mA with completely loaded batteries) from the 9V source.

1.2 The DCF77 receiver
The DCF77 signal can be supplied by a commercial module or via the self-made receiver
here. This receiver was used in developing the software.

2 Mounting, initial operation, operation
2.1 Mounting
The controller part was mounted on
a double-sided breadboard. The 16pin connector for the LCD is mount
ed towards the soldering-side and
the connections are made with sol
derable enameled copper wire.
The six-pin ISP connection and the
sixteen-pin connector can be seen in
the lower part of the picture. To the
left side the supply connector is lo
cated.

Here the mount
ing of the com
ponents
in
a
plastic box can
be seen. To the
right the power
supply, to the
left the speaker
on the side wall,
below the board
with the con
troller
and
a
small board with
the keys and the
photo transistor.
On the bottom
of the box the
DCF77 superhet
and
the
four
rechargeable
batteries
are
mounted.
Be
cause the super
het is mounted very near to the speaker the receiver is plugged with strong stray signals
in case that music is played on the loudspeaker. When the music finalizes it lasts a few
seconds until the influence of those stray signals diminishes and the reception of DCF77
signals is free of errors.

2.2 Initial operation
When putting the device into
operation the fuses of the
ATmega16 have to be pro
grammed to work with the
external xtal of 3.2768 MHz.
To ease debugging of the
hardware, the source code
has several test modes in
cluded. Those test modes
can be invoked by setting the
debug switches in the source
code to 1, reassembling and
burning the generated hex
code into the controller chip's
flash. All debug options are
in detail described here. Set
ting at least one of the debug
switches to 1 invokes the in
clude
source
file
debug_code_en.inc,
which
normally is not producing ex
ecutable code.
First the LCD shall be tested because all further test steps depend from the LCD and its
correct functioning. The LCD test issues the text shown here and does nothing else than

that.
Further test modes are:
2. dbgAdc: displays the result
of AD conversions,
3. dbgKey: displays the state
of key inputs,
4. dbgSpk: sounds a 1,000 Hz
tone on the speaker,
5. dbgDcfDur: displays the du
ration of signals on the
DCF77 input in milliseconds
(high and low),
6. dbgDcfSig: displays signal
errors on the DCF77 input
with longer text messages,
7. dbgDcfBits: displays the last
eight bits received via
DCF77,
8. dbgMusic: displays diverse parameters when playing music.
With those debug switches all external hardware can be tested.

2.3 Operation
Operating the alarm
clock is rather simple.
The attribution of keys
to actions is simple and
adjusting,
e. g.
the
alarm time, is rather
quick.
All key functions appear
in line 4. A comprehen
sive description of key
functions is given here.

3 Software
3.1 Software in text format
The complete source code can be viewed in html format here in the attachment.

3.2 Download of the assembler source code
To assemble the source code the following files have to be downloaded and stored in the
same directory:
1.
2.
3.
4.

dcf77_clock_m16_v5_en.asm (Main source file),
lcd.inc (the LCD routines),
music_code.inc (Musical notes, melodys and routines),
debug_code_en.inc (Debug routines).

Consult the debug switches (all have to be zero for the final version) and the adjustable
constants in the main source code file to adjust the code to your needs. Then assemble
the code (either with ATMEL's assembler 2, with gavrasm or with the assembler simulator

avr_sim. If you do not know how to assemble, consult the how-to for Linux or Windows.
Do not forget to change the fuses of the ATmega16 and, if your xtal frequency is different
from 3.2768 MHz, to adjust the constant clock in the main source code, otherwise your
time and date will not be correct.

3.3 Description of the software, documentation
The functioning of the software is comprehensively described here, all functions are in the
ory and practice listed there.
Praise, error reports, scolding and spam please via the comment page to me.
To the top of that page

Attachments:
•

A more general description of date and time with AVRs in assembler

•

A home-made DCF77 superhet receiver with high performance

•

Software description of the assembler source code

•

Assembler source code listing of the
◦ main program,
◦ LCD include routines,
◦ music codes,
◦ debugger routines.

Date and time with an AVR in assembler
Sometimes it happens that one has to handle a date or time on an AVR, e.g. if program 
ming an alarm clock. While C programmers have their large and mighty libraries (and do
not understand what those do at all), the assembler programmer has to do it by himself.
And it is pretty simple.

0 Content
1. To measure seconds as correct as necessary
2. Date and time formats
3. Date and time programming

1 To measure seconds as correct as necessary
1.1 Loops for timing
If the AVR has nothing else to do than this (this happens rarely) one can send the AVR
into an endless count to come to seconds, minutes, hours, days and years. Fortunately an
AVR is a rather silent controller and he will not cry or complain about having to do such
nonsense like counting down the seconds in a 28-bit register series from 0x01E13380
down to zero before advancing the year by one (or rather 0x01E28500 if it is a leap year).
Or, if he runs at a clock frequency of 1 MHz, to absolve 0x1CAE8C13E000 useless clock cy
cles using a 48-bit wide down-counter. He will be quit during his job, and will do as the as 
sembler programmer has told him.
For counting down a second at 1 MHz clock frequency a counter is needed that can count
down from 300,000 to zero. For this we need 20 bits, a single byte or a 16-bit counter is
insufficient for that. Such a counter in assembler can be constructed as down-counter or
as up-counter. A down-counter has first to set a delay loop constant, then to down-count
(LSB first, if carry: MSB next, if carry HSB last. Hint: use SUB or SUBI to down-count in 
stead of DEC because DEC does not influence the carry flag when the register is decreased
from zero to 0xFF.
That would produce a source code like this:
ldi R16,BYTE1(cDelay) ; Set start values in R18:R17:R16
ldi R17,BYTE2(cDelay)
ldi R18,BYTE3(cDelay)
DelayLoop:
subi R16,1 ; Down-count LSB
brcc DelayCheck ; No carry: check zero
subi R17,1 ; Down-count MSB
brcc DelayCheck ; No carry: check zero
subi R18,1 ; Down-count HSB
DelayCheck:

tst R16 ; Check
brne DelayLoop
tst R17 ; Check
brne DelayLoop
tst R18 ; Check
brne DelayLoop
DelayEnd:
; One second is

LSB
MSB
HSB
over

When up-counting the source code is similar:
clr R16 ; Start from zero
clr R17
clr R18
DelayLoop:
subi R16,-1 ; up-count LSB, carry flag reversed!
brcs DelayCheck
subi R17,-1 ; Up-count MSB, carry flag reversed!
brcs DelayCheck
subi R18,-1 ; Up-count HSB, carry flag reversed!
DelayCheck:
cpi R16,BYTE1(cDelay)
brne DelayLoop
cpi R17,BYTE2(cDelay)
brne DelayLoop
cpi R18,BYTE3(cDelay)
brne DelayLoop
DelayEnd:
; One second over

Now, which number cDelay should be chosen for a second? For that we need to know how
many clock cycles the whole loops require. As each BRCC/BRCS and BREQ/BRNE needs
two clock cycles in case it jumps and one clock cycle in case it does not jump, a very so 
phisticated formula would be needed to calculate the complete number of cycles from a
constant cDelay, and vice versa.
The only way out of this is to construct the different branches in a way that they consume
the same number of execution cycles. Each line executed is analyzed for its clock cycles
and, with the help of NOP instructions and jumps, delayed so that each branch needs the
same clock cycles. In case of the down-counting the subtraction phase so would look like
this:
DelayLoop: ; 0 clock cycles
subi R16,1 ; + 1 = 1 clock cycle
brcc Delay1 ; + 1 = 2 for not jumping, + 2 = 3 for jumping
subi R17,1 ; + 1 = 3 clock cycles
brcc Delay2 ; + 1 = 4 for not jumping, +2 = 5 for jumping
subi R18,1 ; +1 = 5 clock cycles
brcc Delay3 ; +1 = 6 for not jumping, + 2 = 7 for jumping
rjmp DelayCheck ; + 2 = 8 clock cycles
Delay1: ; 3 clock cycles
nop ; + 1 = 4 clock cycles
nop ; + 1 = 5 clock cycles
Delay2: ; 5 clock cycles
nop ; + 1 = 6 clock cycles
nop ; + 1 = 7 clock cycles
Delay3: ; 7 clock cycles
nop ; + 1 = 8 clock cycles
DelayCheck: ; 8 clock cycles

; (Now check if all zero)

Now the complete execution between DelayLoop: and DelayCheck: needs 8 clock cycles,
where-ever all those branches run along. A predictable execution time for this. Now the
same for the DelayCheck: that checks if all registers are zero:
DelayCheck: ; 8 clock cycles
tst R16 ; + 1 = 9 clock cycles
brne DelayCont1 ; + 1 = 10 for not jumping, + 2 = 11 for jumping
tst R17 ; + 1 = 11 clock cycles
brne DelayCont2 ; + 1 = 12 for not jumping, + 2 = 12 for jumping
tst R18 ; + 1 = 13 for not jumping, + 2 = 14 for jumping
breq DelayEnd ; + 1 = 14 for not jumping, + 2 = 15 for jumping
rjmp DelayLoop ; + 2 = 16
DelayCont1: ; 11 clock cycles
nop ; + 1 = 12 clock cycles
DelayCont2: ; 12 clock cycles
nop ; + 1 = 13 clock cycles
nop ; + 1 = 14 clock cycles
rjmp DelayLoop ; + 2 = 16
DelayEnd: ; 15 clock cycles
nop ; + 1 = 16 clock cycles
; One second is over

So the whole number of clock cycles N needed is
N = 3 + 16 * cDelay
of which the 3 is for the initial phase of setting the registers, and cDelay for one second
can be calculated simply like this:
cDelay = (N - 3) / 16
The result for N = 1,000,000 is 62,499 (rounded down), which corresponds to 3 + 62,499
* 16 = 999,987 cycles. If you need to be more exact (see accuracy discussion below), you
can add 13 NOP instructions at the end.
Note that in our example R18 is zero, because 62,499 fits into two registers and we can
skip that, but without removing the code (because that would change code execution
times and with a shorter loop would go beyond the 16 bit limit).
With those instruments we can construct whatever type of counting loops (three, four,
five, etc. bytes) to fit to any time (minutes, hours, days and years) to arrive at an exact
timing.

1.2 Timer as clock counter
Get rid of the boring counting by the use of built-in timers. Those timers can count up
(and sometimes down, too) and they are very reliable and predictable. They are in any
case doing their jobs even if numerous interrupts or other tasks happen in between.
Unfortunately timers and their prescalers work in binary mode, so they like the two's more
than the ten's. If our clock rate is at 1 MHz (which is more the ten's domain) a timer
which divides by 64 yields 15,625. The next higher binary divider, 128, already has a fre 
quency which is not an integer value but has a fraction following. Not to speak about di
vider rates of 256, 2,048 or 16,384 or even higher (for which all timers have a prescaler).

1.2.1 16-bit timer with a crystal
As a clock device needs a crystal anyway to be exact enough, we can select xtal frequen
cies that better fit to binaries, e.g. 2.048 MHz, 2.097152 MHz, 2.4576 Mhz, 3.072 MHz,
3.2768 MHz or 4.194304 MHz.
This here are the divider
rates when using possible
prescaler values in 8- and
16-bit timers. We see that
the 8-bit timer has rather
low divider rates even with a
prescaler of 1,024. To come
down to one second our xtal would have to be either of the 32,768 Hz type or hand-craft
ed to our selected frequency. And: the controller would act but would act like a lame duck
at those low frequencies.
More promising is a 16-bit timer with a prescaler of 64. Xtals with a frequency of
4.194,304 MHz are sold on each street corner or electronics store and cost around
25 Cents. With that the seconds timer is already complete, just do the following:
1. attach the Xtal and two ceramic capacitors of 18 pF
to GND to an AVR that has an internal Xtal oscillator,
2. change the fuses of the AVR to use an external xtal
with medium or high frequency,
3. copy the interrupt vector table of the controller with
all RETI except for the Reset vector (RJMP Start) and
the overflow interrupt for TC1 (RJMP TC1OvfIsr),
4. write an overflow interrupt service routine "TC1Ov
fIsr" for the timer with the two instructions SET (set
the T flag in the status register) and RETI (Return
from Interrupt),
5. in the main program "Start:" init the stack pointer,
6. then set the 16-bit timer to normal operation and the prescaler to 64,
7. enable TC1 overflow interrupts (TOIE1) and the interrupts generally with SEI, and
8. ask in a loop if the seconds flag T is set, if yes clear it with CLT and do what you'll
want to do in each second.
That is all you need for a xtal controlled clock. Anything else that you want to do with that
seconds pulse, e.g. advancing time and date counters, displaying on an LCD or on a 7segment display, can last as long as necessary: the timer works correct if that does not
last longer than two seconds. Which is rather long, even with a lame-duck type of 32.768kHz-oscillator.
• Those who find a 4.1 MHz clock too fast and want to save battery current, or
• who do not find such an Xtal on their street corner,
• who need the 16-bit timer for other more valuable purposes, e.g. to play a melody
with it when the clock reaches 07:30 in the morning,
can find other solutions for that.

1.2.2 8-bit timer with a register divider
In any case the selected xtal frequency di
vided by the prescaler and by the timer
used has to be a pure integer value. If the
result of the division is not one but an in
teger value: use an 8-bit (smaller or equal
256) or a 16-bit register (256 to 65,536)

to divide the resulting frequency down to one.
With the above named xtals an 8-bit timer has the following register divider values.
These are the crystals that
can be used with an 8-bit
timer,
the
different
prescalers and a register
divider. Red marked regis
ter dividers are 16-bit,
green marked ones are 8 bit. Only at 3.072 MHz a 16-bit divider is necessary.
In Assembler dividing within the overflow interrupt service routine goes like this:
; Calculation constants
.equ clock = 2097152 ; Xtal frequency
.equ prescaler = 1024
.equ timertop = 256
.equ second_divider = clock / timertop / prescaler ; = 8
;
; Registers
.def rSreg = R15 ; Save status
.def rSeconddivider = R17 ; Counter down
.def rFlag = R18 ; Flag register
.equ bSek = 0 ; Second flag in flag register
;
Tc0OvflwInt:
in rSreg,SREG ; Save SREG
dec rSeconddivider ; Divide by 8
brne TC0OvflwIntRet ; Not yet zero
ldi rSecondendivider,second_divider ; Restart divider
sbr rFlag,1<<bSek ; Set seconds flag: one second is over
Tc0OvflwIntRet:
out SREG,rSreg ; Restore SREG
reti

1.2.3 Timer in CTC mode
The method does not fit well if you are addicted to a certain clock frequency due to other
reasons. Only at 4.0 MHz and an 8-bit timer dividing by 256 the integer 15.625 results.
This can be divided in a 16-bit register, such as R25:R24, by use of the instruction SBIW
R24,1. When the timer reaches zero, it is restarted with 15,625 and the second is over.
This does not work on many other frequencies. But for those there is another method
available.
Many nice clock frequencies such as 1 or 2 MHz do not work with those methods. In those
cases the timer has to be made more ten-friendly by clearing it when he reaches a count
of 100 or 250. This can be done in the so-called CTC mode (Clear Timer on Compare). If
the counter reaches the value that is stored in its Compare Port Register COMPA, the next
count will restart the timer at zero (in 16-bit timers the Input Capture Port Register ICR
can also be used for the comparison purpose). The one extra clock cycle means that the
compare value has to be one smaller than the desired division rate. If you need the same
timer also for other purposes, e.g. as a pulse width generator, the CTC mode setting also
alters its resolution.
With 1 MHz clock the prescaler can be set to 64 (which means 15,625 kHz on the timer in
put), can set the compare value to 124 (dividing with CTC by 125) and dividing with a
register divider by 125 to yield one second. Many other frequencies can be counted down
like that to an exact second.

Provided that we have
• the counter running,
• its interrupts enabled (TOIE0/TOIE1 with an overflow int resp. OCIE0A/OCIE1A with
a compare match A int),
• the main interrupt flag I in SREG is set,
• the interrupt vector is properly set and branches to the service routine, and
• the stack is initiated and works properly.

2 Date and time formats
Dates and times can be handled in four different usual formats. Fortunately all those for
mats can be converted into each other.
Instead of dividing the day into two halves (AM and PM) we use here the European nota
tion for hours (0 to 23).

2.1 Time in ASCII format
This is the simplest format in respect
to displaying the time on an LCD:
each digit in one byte, encoded as
ASCII character. ASCII has been de
signed for some US teletype ma
chines, so it has some interesting
control codes between characters 0 and 31, such as BEL for wringing a bell on your tele
type - ahem, computer. Or the character number 13, which moves the write head to the
left side to the margin (carriage return). Or number 10, that moves the printing paper of
the teletype printer by one line up (linefeed). As those low level codes of ASCII were al
ready occupied for this kind of pseudo characters the numbers start with code number 48
for zero and reach until 57 for nine. The rest of the ASCII codes is not used for time en
coding, only the ':' character makes some sense here.
Writing the numbers as '0' to '9' means that the ASCII representation is meant. Even as 
semblers understand that:
ldi R16,'0' ; Load ASCII 0
ldi R17,'7' ; Load ASCII 7
ldi R18,':' ; Load : character as separator

If we want to write eight characters to SRAM we reserve space in SRAM for it:
.dseg
TimeAscii:
.bytes 8

and write the following code:
.cseg
ldi ZH,HIGH(TimeAscii) ; Point to SRAM address
ldi ZL,LOW(TimeAscii)
ldi R16,'0' ; Load ASCII 0
st Z+,R16 ; Store in SRAM and increase address
ldi R16,'1'
st Z+,R16
ldi R16,':'

st Z+,R16
ldi R16,'2'
st Z+,R16
ldi R16,'3'
st Z+,R16
ldi R16,':'
st Z+,R16
ldi R16,'4'
st Z+,R16
ldi R16,'5'
st Z,R16

A different formulation for that would be:
ldi YH,High(TimeAscii) ; MSB of the SRAM address to YH
ldi YL,Low(TimeAscii) ; LSB to YL
ldi R16,'0' ; ASCII-Null in R16
st Y,R16 ; To Hour-Tens
inc R16
std Y+1,R16 ; To Hour-Ones
inc R16
std Y+3,R16 ; To Minutes-Tens
inc R16
std Y+4,R16 ; To Minute-Ones
inc R16
std Y+6,R16 ; To Second-Tens
inc R16
std Y+7,R16 ; To Second-Ones
ldi R16,':' ; Separator to R16
std Y+2,R16 ; To the first separator location
std Y+5,R16 ; To the second separator location

STD (and when reading: LDD) does not change Y but temporarily adds the displacement
behind + and writes the byte in R16 to this location. This works with Y and Z, but not with
X.
To increase the second-ones by one second. If this leads to the ASCII character behind '9',
it has to restart with '0' and has to increase the second-tens. With the constant address in
Y that goes like that:
ldi YH,High(TimeAscii) ; MSB of the SRAM address to YH
ldi YL,Low(Zeit) ; LSB to YL
ldd R16,Y+7 ; Read second-ones to R16
inc R16 ; Increase second-ones by one
std Y+7,R16 ; Write increased second-ones
cpi R16,'9'+1 ; Compare with ASCII code for next char behind nine
brcs Done ; If carry set no overflow to next higher digit
ldi rmp,'0' ; Restart second-ones
std Y+7,R16 ; Write ASCII-zero to second-ones
ldd R16,Y+6 ; Read second-tens
inc R16 ; Increase second-tens
std Y+6,R16 ; and write back to second-tens
cpi R16,'6' ; Second-tens at six?
brcs Done ; No, ready
; ... Minutes and hours similarly
Fertig:
; ... Done with the clock increase

When increasing the hour, those two criteria come into play:
• if the hour-ones are larger than '9', and

• if the hour-ones are four AND the hour-tens are '2'.
By using the relative addressing with STD and LDD time increasing gets rather simple.

2.2 Time in BCD format
With this second method the ones and tens of
seconds, minutes and hours are not stored as
ASCII characters but as binary encoded deci
mal digits (BCD). Those bytes range from bi
nary
0
(0b00000000)
to
binary
9
(0b00001001) for the ones, from 0 to 5 for the second- and minute-tens and from 0 to 2
for the hour-tens.
The comparison if the ones have exceeded the nine is now done with the instruction CPI
R16,10 instead of CPI R16,'9'+1. Restart of the ones is done with CLR R16 instead of LDI
R16,'0'. Anything else remains the same, but the ':' makes no sense any more and is in 
serted when displaying the time but has no own SRAM location.
When displaying the BCD codes on the LCD one simply has to add 48 to the BCD. Because
there is no ADDI instruction on the AVR, there are three different ways around for that:
1. We write the 48 to another register (e.g. LDI R17,48) and add this register to the
BCD in R16: ADD R16,R17.
2. Set the bits bits 4 and 5 in the BCD with either ORI R16,0x30 or with SBR
R16,0x30 or with SBR R16,(1<<4)|(1<<5).
3. Subtract -48 from the BCD with SUBI R16,-'0'. Subtracting a negative number is
the same as adding the positive number.
All three methods have the same result. Only the first methods is different because it re 
quires an additional register.
When displaying on the LCD do not forget to insert the separator on the correct location,
otherwise the time would look a bit strange.

2.3 Time in packed BCD format
Because a BCD needs only four bits,
you can pack two of those into one
byte - and save some memory and
registers. The ones are fine in the
lower four bits (0 to 3), the tens fit
into the upper four bits (4 to 7). The
package of four bits are called lower and upper "nibble". The complete set of time infor
mation now fits into three bytes. The format is called "packed BCD".
If we want to increase such a second, minute or hour we can use INC R16, too. But it is
more complicated then to detect whether the lower nibble exceeded 9: first we would
have to copy the register (MOV R17,R16), then we have to clear the upper nibble (ANDI
R17,0x0F) and then we can compare this with 10 (CPI R17,10) to decide if we would have
to add (0x10 - 10) to R16, by that clearing the lower nibble and increasing the upper nib
ble.
The more simple solution for that uses a special hardware feature that each CPU, including
the AVR CPUs, has: the half-overflow flag H in the status register. This flag bit signals if
during adding an overflow from the lower to the upper nibble occurred. To use this feature
we add 7 to the lower nibble: if the lower nibble was nine before adding, a half overflow
would occur setting the H flag. The upper nibble would be increased and would already be
fine. In case that the lower nibble was not nine H is clear and we would have to subtract 6
from the result. The conditional branching instructions BRHC and BRHS can be used.

The source code would be:
; Increase lower nibble of R16
ldi R17,7 ; Add 7 to packed BCD
add R16,R17 ; in R16
brhs DoNotSubtract6
ldi R17,6 ; Subtract 6 from packed
sub R16,R17 ; from R16
DoNotSubtract6:
; Done

To save one register (R17) we can use SUBI R16,-7 instead, but the H-bit now is also re
versed: it is cleared when a half overflow occurred and set if not. So BRHS will have to be
changed to BRHC.
Even though it is a little bit more complicated to increase the lower nibble it is much sim 
pler to check if seconds or minutes exceed 59 or the hours exceed 23: just compare the
packed BCD with the next higher limit.
cpi R16,0x60 ; Compare seconds with 60
brne done
clr R16 ; Restart seconds
; ...
cpi R17,0x60 ; Compare minutes with 60
brne done
clr R17 ; Restart minutes
; ...
cpi R18,0x24 ; Compare hours with 24
brne done
clr R18 ; Restart hours
; ...

Instead of two registers to be compared now we have only one to be compared if the
hours reached 24. And instead of two registers to be cleared there is only one. A clear ad
vantage of packed BCD over ASCII and BCD.
The whole packed BCD second increase as assembler source code:
ldi ZH,High(sTimePbcd) ; Z points to hours in packed BCD format
ldi ZL,Low(sTimePbcd)
ldd R16,Z+2 ; Read the seconds
subi R16,-7 ; Add seven
brhc ChkSec ; H clear, tens increased, check seconds for 60
subi R16,6 ; H set, subtract six
ChkSec:
std Z+2,R16 ; Write seconds
cpi R16,0x60 ; 60 seconds?
brcs Done ; No, completed
; One minute is over
clr R16 ; Restart seconds
std Z+2,R16 ; Write seconds to SRAM
ldd R16,Z+1 ; Read minutes
subi R16,-7 ; Add seven
brhc ChkMin ; H clear, tens increased, check minutes for 60
subi R16,6 ; H set, subtract six
ChkMin:
std Z+1,R16 ; Write minutes to SRAM
cpi R16,0x60 ; 60 minutes reached?
brcs Done ; No, completed
; One hour is over
clr R16 ; Restart minutes
std Z+1,R16 ; And write to SRAM

ld R16,Z ; Read hours
subi R16,-7 ; Add seven
st Z,R16 ; And write to SRAM
brhc ChkHour ; H clear, tens increased, check hours for 24
subi R16,6 ; H set, subtract six
st Z,R16 ; and write to SRAM
ChkHour:
cpi R16,0x24 ; 24 hours over?
brcs Done ; Smaller than 24
clr R16 ; Restart hours
st Z,R16 ; Write to SRAM
; Increase date here
Done: ; Increase done

This is it. Those who do not believe that it works can simulate the source code by setting
the SRAM location sTimePbcd to some desired values, such as 0x23:0x59:0x59 and run
through the code.
To display the time formatted as packed BCDs on an LCD the upper nibble has to convert
ed to an ASCII character, then the lower nibble. In case of the hours that goes like that:
ld R16,Z ; Z points to hour in SRAM, read hours
swap R16 ; Exchange nibbles: make upper to lower nibble
and R16,0x0F ; Isolate lower nibble
subi R16,-'0' ; Add ASCII zero
rcall LcdChar ; R16 to LCD
ld R16,Z ; Read hours again
andi R16,0x0F ; Isolate lower nibble
subi R16,-'0' ; Add ASCII zero
rcall LcdChar ; R16 to LCD

The further display of the separator, the minutes, another separator and the seconds is all
the same. If you use the instruction LD R16,Z+ for the second copy from SRAM instruc
tion, you can call the above hour display three times and you are done.
That is all and your time processing is complete in packed BCD.

2.4 Time in binary format
Finally the simplest of all time formats: sec
onds, minutes and hours in binary format.
As a maximum of 59 has to be handled all
fits into one byte. Increasing by one second
is done with INC R16, detection of the com
plete minutes and hours is with CPI R16,60 resp. CPI R16,24. The following example of
seconds increase assumes that the time is in three registers.
.def rHour = R4 ; Hours register
.def rMin = R5 ; Minutes register
.def rSec = R6 ; Seconds register
IncSec:
ldi R16,60 ; Detect end of seconds and minutes
inc rSec
cp rSec,R16 ; Seconds smaller than minute end?
brcs Done ; No
clr rSec ; Restart seconds
inc rMin ; Increase minutes
cp rMin,R16 ; Minutes smaller than hour end?
brcs Done ; No
clr rMin ; Restart minuts
inc rHour ; Increase hours

ldi R16,24 ; Hours per day
cp rHour,R16 ; Hours smaller than day end?
brcs Done ; No
clr rHour ; Restart hours
Done:
; Time increase done

With those 14 simple instructions for a 24 hour clock it is a simple thing to program a
clock. No reason at all to include a powerful datetime library and blow up those 14 simple
instructions by at least 100-fold.
The display of the binary encoded time information now requires a binary-to-decimalto-ASCII conversion. The following routine converts the binary in R16 to two ASCII digits
and displays those on the LCD.
Bin2Dec2:
clr R0 ; Count tens in R0
Bin2Dec2a:
inc R0 ; Increase counter
subi R16,10 ; Subtract 10
brcc Bin2Dec2a ; No carry, repeat counting and subtract
subi R16,-10-48 ; Undo last subtraction (add 10) and
; convert to ASCII (add 48)
push R16 ; Is needed later on, pushed to stack
ldi R16,'0'-1 ; Counter minus one plus ASCII zero
add R16,R0 ; Add counter value
rcall LcdChar ; R16 as character to LCD
pop R16 ; Restore second digit from stack
rjmp LcdChar ; and write to LCD

The routine LcdChar, usually in the LCD include code, displays the character in R16 on the
current position on the LCD and advances the cursor to the next position on the LCD. Due
to the use of RCALL and PUSH/POP the stack has to be working.
For testing the routine you can point Z to the SRAM location SRAM_START and instead of
the include routine LcdChar: you use ST Z+,R16 and RET to write the converted ASCII
characters to SRAM.
The complete display of the time then goes like this:
Display:
mov R16,rStd ; Hours to R16
rcall Bin2Dec2 ; Call conversion and display routine
ldi R16,':' ; Separator
rcall LcdChar ; to LCD
mov R16,rMin ; Minutes to R16
rcall Bin2Dec2 ; Call conversion/display
ldi R16,':' ; Another separator
rcall LcdChar ; to LCD
mov R16,rSec ; Seconds to R16
rjmp Bin2Dec2 ; Jump to conversion/display

With these 20 instructions, of which 10 are for the display of binaries on the LCD, the
whole stuff is not too sophisticated.

2.5 The best format
... is of course the binary, but the other three formats have also their pro's. So choose
what you want, anything goes.

3 Time and date
After having handled times we can handle dates as well. It works like times but has some
extra rules making it a little bit more complicated:
• Pope Gregor the XIIIth has, in the year 1582, issued the bulletin "Inter gravissi
mas", which decided that in contrary to seconds, minutes, hours and years days
and years do not start with zero but with a one,
• the fact that the speed of the earth in rouding the sun is not a simple number but
has some fractional parts (365.242188 days), that make it necessary to define
some extra time to the 365 days. The Egyptians rounded this up a bit (to 365.25
days) and worked with that when they defined their tax paying algorithm. Pope
Gregor wanted to be more exact, but did not e.g. define an extra "Earth rumble
day" with a special duration of 5 hours, 48 minutes, 45 seconds, 43 milliseconds,
199 microseconds and a few nanoseconds. Instead of this simple solution pope Gre
gor went into a much more sophisticated algorithm that makes the life of calendar
designers and assembler programmers more complicated,
• he decided that the months have not an equal number of days, e.g. ten months
from 0 to 9, with 36 days each (day 0 to 35), but defined 12 months instead, even
though 365 / 12 = 30.416666 (periodic) and started with 1 instead,
• the decision to choose 12 as the month's basis leaded to further complicated conse
quences:
• the number of days per month was not a constant but varied,
• this number of days did not follow simple rules, e.g. such as 30/31 alternat
ing with the December length varying, but the alternating rule was reversed
between July and August and February's length was largely varied,
• the February's length was defined as being 28, but
• not every four years if the year can be divided by four without a frac
tional part, then it is 29 days long,
• but not every 100 years, when the year can be divided by 100 without
a fractional part, then it is 28 days long,
• but not every 400 years, when the year can be divided by 400 without
a fractional part, then it is 29 days long,
• he further decided that one week has seven weekdays and not 10, an ancient
"magic" number rather than a decimal's society preferred selection, but
• their numbering remained undefined, still causing confusion: some start with the
Monday, others with the Sunday.
When writing dates on the LCD, and when storing those in SRAM, further illogical decisions
complicate the life of the assembler programmers. While the time follows the simple rule
that the further left is the more significant number (hours:minutes:seconds) dates are
confused by other rules:
• The format convention Days:Months:Years reverses priority.
• The format convention Months/Days/Years simply confuses any priority rule.
• The format convention Years:Months:Days would be similar to the time format, but
is rarely used.
But whatever you want to use: place them in the same row in SRAM as you want to dis
play them, either forwards or backwards, so your pointer can increase or decrease after
having outputted it. Or use displacement pointers.

This all makes programmers con
fused. To determine the number of
days of a month we need this algo
rithm. Looks complicated but isn't
that complicated in assembler:
;
; Subroutine calculates the
; number of days of a month
;
Month in rMonth, Year in
;
rYear, result in rDom
;
DaysOfMonth:
ldi rDom,31 ; 31 days
cpi rMonth,2 ; February?
brcs DaysOfMonthRet ; January
brne DaysOfMonth1 ; Not
February
; February
ldi rDom,28 ; No leap year
mov R16,rYear ; Leap year?
andi R16,0x03 ; Last two bits
; of year?
brne DaysOfMonthRet ; Last two
bits not 00
ldi rDom,29 ; Leap year
rjmp DaysOfMonthRet
DaysOfMonth1:
mov R16,rMonth ; Copy month
cpi R16,8
brcs DaysOfMonth2 ; March to July
dec R16 ; Monthes larger or equal August, reverse
DaysOfMonth2:
ldi rDom,31 ; Months with 31 days
andi rmp,0x01 ; Uneven?
brne DaysOfMonthRet
ldi rDom,30
DaysOfMonthRet:
ret ; Done

Simulation of the DOM
routine with the mon
thes 1 to 12 for a leap
year (first line) and a
non-leap year (second
line) demonstrates that the days for the months (+01 to +0C), as written to the SRAM,
are correct.
Simulation was, of course, performed with avr_sim.
With these tools we can start to design a complete time and date flow diagram for a
date/time clock. Rectangles are calculations, rhombuses are display operations and
squares rotated by 45 degrees are decisions or conditional branches.
In the algorithm only those display elements are updated that are changed during an in
crease of that second. As a change of the year requires an update to all displayed ele
ments on the LCD the update algorithm runs from bottom to the top while the time and
date algorithm runs from top to bottom.

The first decision is that the date and time on
the LCD is controlled by a different routine
during a date and time input. Note that in that
case the increase in time does not work but
waits for a completion of the input procedure.
Here the complete increase time/date routine
in assembler:
;
; Increase time in seconds
;
IncSec:
ldi ZH,High(DateTimeBuffer)
ldi ZL,Low(DateTimeBuffer)
ldd R16,Z+6 ; Seconds
inc R16
std Z+6,R16
cpi R16,60
brcs DisplaySec
clr R16
std Z+6,R16
ldd R16,Z+5 ; Minutes
inc R16
std Z+5,R16
cpi R16,60
brcs DisplayMin
clr R16
std Z+5,R16
ldd R16,Z+4 ; Hours
inc R16
std Z+4,R16
cpi R16,24
brcs DisplayHours
clr R16
std Z+4,R16
ld R16,Z ; Weekdays
inc R16
st Z,R16
cpi R16,7
brcs IncDay
clr R16
st Z,R16
IncDay:
rcall DaysOfMonth ; Get days of that
; month in R16
inc R16
mov rData,R16
ldd R16,Z+1
inc R16
std Z+1,R16
cp R16,R0
brcs DisplayWeekdays
ldi R16,1
std Z+1,R16
ldd R16,Z+2 ; Monthes
inc R16
std Z+2,R16
cpi R16,13
brcs DisplayMonthes
ldi R16,1
std Z+2,R16
ldd R16,Z+3 ; Years

inc R16
cpi R16,100
std Z+3,R16
brcs DisplayYears
clr R16
std Z+3,R16
DisplayYears:
; ...
DisplayMonthes:
; ...
DisplayWeekdays:
; ...
DisplayHours:
; ...
DisplayMinutes:
; ...
DisplaySeconds:
; ...
ret

To simulate the work of IncSec: here a few tests. First, the year change on the 31nd of
December, 2017, is shown. The line from address
• 0x0070
format,
• 0x0080
• 0x0090
• 0x00B0

shows the original data for the Sunday at that date, for 23:23:59, in binary
shows the date/time increased by one second, as binary,
displays the original date/time in ASCII, and
on shows the increased date/time in ASCII.

The change in the
year works cor
rect.
Simulation
was
again, of course,
performed
with
avr_sim.

This simulates the
seconds increase
for February 28,
2019, which was
not a leap year.
Also correct.

And this is the seconds increase on February 28, 2020, which is a leap year. The routine
obviously works correct.

I wish you success in the self-making of date/time routines in assembler.

A DCF77 AM Superhet Receiver
1 What is DCF77?
DCF77 is a transmitter in the
VLW band with a frequency of
77.5 kHz
at
50 kW,
see
Wikipedia for more details. It is
located at Mainflingen in midwest Germany (near Hanau and
Frankfurt) and transmits the "of
ficial" time in Germany by
• reducing its amplitude
down to 15% at the start
of each second for a dura
tion of either 100 or
200 milliseconds, by that
encoding zeros and ones
bits,
• a missing 59th bit that
signals the end of a
minute and the start of
the next amplitude reduction as the begin of the next minute,
• encoding time and date information as well as several parity bits in those 59 bits.
With that information a clock can be synchronized to display exact time and date.

2 Why a home-brewed receiver for DCF77?
Besides that it is fun to build an own receiver for those Very-Long-Wave signals there are
some rational reasons to do that:
1. Contrary to commercially available DCF77 receiver modules this receiver has a very
high amplification of the signal and so can receive the signal over a very large dis 
tance, beyond the 2,000 km limit that DCF77 has been designed for. Two amplifica
tion regulators reduce the amplification level in the near-field so that it works well
in my 30 km distance to DCF77 (with very small amplification) as well as in the farfield (at a very high amplification level). Even enclosed within an iron cookie box
the AM receiver still works fine.
2. Several LC resonators and a ceramic filter increase the immunity to noise. Noise in
this frequency range comes from switching regulators (e.g. power supply devices,
switched lamps, etc.) that can have very large amplitudes and can tamp any receiv
er that has less selectivity. E. g. my energy-saving lamp in my shag or the power
supply of my laptop produces so much noise that any commercially available DCF
receiver module in less than 2 m distance is overwhelmed and produces nothing.
3. Just because you can do it by yourself: receiving VLW waves via a ferrite rod anten

na, decoupling of this already very selective LC resonator with an FET, preamplify 
ing this signal in a regulated HF amplifier, mixing the HF with a (77.5+455) =
532.5 kHz oscillator, selecting the 455 kHz mixer result with a High-Performance IF
filter, feeding this into a high-gain regulated amplifier, rectification of the resulting
IF and to filter the amplitude in an RC filter to detect the duration of the second's
amplitude reduction can be done within a single chip and some external compo
nents.
But why use such an old IC TCA440 instead of more later ones?
• Because more modern IC combine AM and FM in one chip, but the FM part is use 
less here. The same with audio amplifiers on board: useless for a DCF77 receiver.
• Because more modern ICs only offer a very small part of the overall amplification of
a TCA440, they are designed to cover only very large signal strengths with that low
amplification factors.
• Other commercially available DCF77 receivers are of the direct amplification type
and filtering resonators (of the ceramic or crystal type) for 77.5 kHz are rarely
available, if at all. By mixing the signal High-Performance IF filters can be used to
increase the selectivity. And the TCA440 has anything on board that is necessary
for that.
• To increase the selectivity of more modern AM receiver ICs a large number of exter 
nal components would be necessary. The TCA440 has anything on board for that.

2 Schematic of the receiver
On the next page is the electrical scheme of the receiver.
The VLW signals are received with a 10 cm ferrite rod with enameled copper wire. The ca
pacitors are to be selected for the measured impedance of the rod (see below) and a trim
capacitor of 90 pF allows for adjusting the antenna resonator for a maximum signal.
The N channel FET BF245 (any other N channel type can be used) decouples the resonator
and generates a symmetrical (differential) signal for the HF input of the TCA440, which is
coupled to the input with two capacitors.
The HF amplifier of the TCA440 is connected with pin 16, where an LC filter with a Germa
nium diode and a 15 k resistor and a 100 µF electrolytic capacitor produces a regulator
voltage for the gain of the HF amplifier. This regulator voltage controls the HF amplifica 
tion in the near-field. The LC filter is self-made on a Neosid coil body 05 9539 00 (9.5 µH).
The LC resonator on pins 4, 5 and 6 of the TCA440 generates an oscillator signal of
532.5 kHz to be fed to the mixer. This LC coil is also self-made and produces a stable os
cillator signal.
The output of the mixer on pin 15 is fed into a commercial SFT006H IF filter consisting of
a combination of two LC resonators and a 455 kHz ceramic resonator. The filtered signal
enters the IF amplifier on pin 12 of the TCA440.
IF amplifier output on pin 7 of the TCA440 is filtered with a commercially available
455 kHz resonator and rectified with a germanium diode. A RC filter network filters the
fast seconds pulses, to be fed to the positive input of the CA3140 FET opamp, and aver
ages those very slowly to be fed to the negative input pin of CA3140. The slow average
voltage is also fed to the IF regulator voltage input on pin 9 of the TCA440 and to the dis
play driver input. The display driver output voltage drives a 288 µV instrument and can be
read on an AD converter input pin of the controller.

The output of the opamp CA3140 drives a transistorized-driven LED and provides the sec
onds pulses for the controller readout and DCF77 signal analysis. The output signal follows
the amplitude of DCF77: high amplitude = high, low amplitude = low. A good controller
software should be able to detect the signal no matter if it is high- or low-active (like the
one described here).
All external components, including the power supply of 5 V, are tied to a 6-pin connector.

Simulation of the voltages
These are the voltages that occur on the capacitors of the RC network when a logical zero
and one occur in the DCF77 signal by decreasing the amplitude to 15%.

The voltage on the first capacitor, after rectification of the IF, decreases and increases very
fast. Its small capacity of 1 µF has no relevant influence and just smoothes down the
455&nbps;kHz peaks from the rectifier diode.
The voltage on the fast integrator with R3 (2k2) and C2 (1µF) decreases relatively fast
with the reduced amplitude. It just buffers the signal during the time when the rectifier
does not load the capacitor.
The voltage on the slow integrator with R4 (39k) and C3 (220µF) remains uninfluenced by
the second's amplitude reduction (very small decrease). It takes many seconds until the
voltage reaches its average value.
The digital output signal on the CA3140 is very stable, but very short spikes can occur
from the 455 kHz rectified signal. Those can easily be filtered by the controller's software.

3 Mounting
3.1 Mounting the antenna
The VLW receiver an
tenna is build on a
ferrite rod, where a
0.25 mm copper wire
is wounded on. Litz
wire is not necessary
due to the low fre
quency.
On both ends the coil is fixed with tape. The antenna is mounted in two plastic holders
onto the PCB.

To determine the inductance of the anten
na coil a capacitor of 10 nF is added to
form a resonator and a signal generator
(can be a digital one, such as described
here) feeds its signal over a small capaci
tor such as 10 pF into that resonator. Ei
ther the signal on the resonator is viewed
with an oscilloscope or the shown rectifier
produces a DC voltage from that, which can be measured by a usual voltage meter. By
that the resonance can be identified.
From the resonance frequency fres the inductance of the coil can be calculated:
L(H) = 1 / 4 / Π2 / f2 / C.
The inductance shall be approximately 3.5 mH.
From the measured inductance the capacitor C, which is necessary for a resonance fre
quency of 77.5 kHz, can be calculated:
C(F) = 1 / 4 / Π2 / 775002 / L

3.2 Mounting the LC filters
The LC filters are mounted as follows.

3.2.1 HF preamplifier circuit
Primary coil 91 windings, secondary 45 windings, 0.1 mm enameled copper wire on
Neosid coil body 05 9539 00 (9.5 µH).

3.2.2 Oscillator circuit
Primary coil 30 plus 76 windings, secondary 13 windings, 0.1 mm enameled copper wire
on the same Neosid coil body.

3.2.3 IF filter
This filter is a commercially available TOKO SFT006H.

3.2.4 IF output filter
The output filter is a commercially available filter 7MC-A2549HM.

3.3 Mounting the PCB
The receiver can be build on a
small PCB and wired with
0.25 mm solder-able enameled
copper wire. The TOKO filter
does not fit into the 2.54 mm
pattern and larger holes have to
be made to fit that into.

3.4 Adjustment
When adjusting the receiver's
circuits the ferrite rod should be directed to Mainflingen and this direction is not to be

changed during adjustment (because it has a strong influence on signal strength's).
Note that adjusting is easy because none of the adjustments affects other adjustments as
well. So bringing all to a maximum is easy. But also note that the temperature also has a
strong influence on the result, so do not be disappointed if an evening adjustment yields
other result than an afternoon adjustment.
If you have an oscilloscope start with it connected to the source of the BF245. If not, con 
nect a voltmeter to pin 5 of the 6-pin connector. You can also use a field strength instru
ment, if you have one attached.
First adjust the trim capacitor parallel to the antenna coil to a maximum signal strength. If
you are very far away from Mainflingen you might not be able to see the signal yet on
your oscilloscope. In that case use the voltage measurement or the signal strength indica
tor instead. If that does not work either come back to this step later on, when the rest of
the filters has been adjusted.
If you own a frequency counter, such as the one described here, you can first adjust the
oscillator frequency to 532.5 kHz on pin 6 of the TCA440. If not, adjust to maximum.
Adjusting the other resonators goes all around and in direction of the maximum.

4 Operation
The signal strength is varying mainly due to the sensitivity of the ferrite rod antenna to di 
rectional changes. When changing direction it might takes rather long until the signal level
stabilizes due to the slow averaging. The LED can be used to identify a successful direction
change: a flickering signal on that shows unsuccessful changes. But in any case leave
enough time for averaging.
Good luck with receiving DCF77.

1 The software structure
The source code for the assembler is parted into the following files:
1. dcf77_clock_m16_v5_en.asm: is the main program. It inits the hardware and pro
cesses interrupts, performs DCF77 signal processing, the clock, reacts on keys and
controls the AD conversion,
2. lcd.inc: holds all routines for control and operation of the LCD: initializing, position 
ing, output of control commands and character writing, conversion and displaying
binaries in decimal and hex, generating special characters on the LCD,
3. music_code.inc: contains all routines for generating and encoding of music that is
played on wake-up, definition of music notes, encoding of notes, their duration and
of pauses, stored pieces of music, conversion of stored music pieces to a series of
tone frequencies and their duration,
4. debug_code_en.inc: all routines, that are needed for processing and displaying de
bug options, this code is assembled only if debug options are enabled, for the final
version all dbg switches have to be set to zero.

2 The debug switches
As an assistance during setting up the hardware several debug switches were built into
the software that allow for testing the hardware components of the device. Those options
are activated by setting those switches in the part "Debug switches" of the software to one
and re-assembling. Note that some of the options start solely that specific debugging pro
gram part and do not perform other debugging parts or normal operations any more. As
the debug output appears on the same line of the LCD it makes no sense to set more than
one debug option to one.
The following switches are implemented and are listed in their hierarchical order:

1. dbgLcd: This option switches the LED
on, sets its backlight to 50% and dis
plays the opening message in lines 1
and 2, while line 3 shows that debug
ging is active. Line 4 should be emp
ty. After having done this nothing
else happens. As all subsequent de
bug options are depending on the LCD to work correct, dbgLcd will have to be set
to zero to test other debug options.
2. dbgAdc: This option switches the LCD
and its backlight on, measures the
selected ADC channel (0, 1 or 2) and
displays the result in hex on line 4.
No further actions are performed. dg
bAdc must be set to 0 to activate fur
ther debug options or normal opera
tion.
3. dbgKey: This option switches the LCD
and its backlight on and displays the
key pins PB0, PB1 and PB2 in binary
format. No other actions are per
formed. Set dgbKey to zero to select
other debug options or normal opera
tion.
4. dbgSpk: This option tests the attached speaker by issuing a permanent tone on the
speaker output pin. No further actions are performed.
5. DCF debug options: When those options are selected the normal operation is per
formed, but no menus appear on line 4.
• dbgDcfDur: displays the dura
tion of incoming signals on the
DCF input pin displays those in
Milliseconds in decimal format
in line 4. Active-High and Ac
tive-Low signals are displayed
separately. Ok and Error sig
nalization for the DCF signal in
line 3 also works correct (here: s1 for a too short signal).
• dbgDcfSig: displays errors in
the received DCF signal in text
form on line 4 (the text dis
played stems from the table
DcfReports: in the source code
of the main asm file). Short
versions of the error messages
in line 3 works correct.
• dbgDcfBits: displays the num
ber of received DCF bits in hex
and the last eight bits in bina
ry form on line 4.

• dbgDcfRcvd: displays the re
ceived DCF information day,
month, year, hour and minute
in hex format, including those
that were only received par
tially (undecoded parts are
displayed as 0xFF),
6. dbgMusic: displays in line 4 the out
put progress of music tones. In each
second are displayed
• 'a' and the address in SRAM,
that is played next,
• 'c' and the CTC value of timer
1 in hex (that corresponds to
clock (3,276,800) / 2 / frequency of the tone - 1),
• 'd' and the counter value that determines the duration ofe the played tone or
pause (depending from the frequency),
all as 16 bit hex.
7. dbgPlayStat: displays on line 4 the
current status of timer TC1 in each
second:
• the
interrupt
enable
bit
OCIE1A (i = off, I = on),
• the OC1A bits (T = Toggle, C
= Clear),
Additionally 'St' and the current stack pointer value in SPH:SPL is displayed in hex
format.
All debug switches have to be set to zero to achieve normal operation conditions and reassembled. The debug code is not assembled in the final version.

3 Setting software properties
The software for normal operation is controlled with with two parameter sets:
1. LCD related properties, and
2. other properties.

3.1 LCD properties
The LCD is connected to the ATmega16 as follows:
• The 8 bit data bus is connected to Port C. By default port operation is unidirectional
(the R/W control input of the LCD is permanently connected with GND). The neces 
sary wait cycles are generated with time loops.
• The LCD input pin E (Enable) is on portpin PB3. Depending from the controller clock
frequency this is activated for one microsecond (by default four nop cycles) and
then deactivated again.
• The LCD input LCD-RS (Register Select) is on PB4. Its polarity allows the selection
of control (Low) or data (High) is written to the LCD.
• The anode of the backlight LEDs of the LCD is connected to the operating voltage,
the cathode via a resistor of 220Ω on the OC2 output pin of the controller. Timer 2
controls the brightness via PWM mode.

Version 5 of the software works with the LCD routines of lcd.inc, which in detail is de
scribed here. All properties that lcd.inc works with are defined in the part
;
; ================================================
;
L C D
C O N F I G U R A T I O N
; ================================================
;

of the source code.
The following properties are configured.

3.1.1 Controller clock
The controller's clock frequency is configurable with the constant clock. By default the fol
lowing clock rate is defined:
.equ Clock = 3276800 ; X-tal clock

Possible and useful, and without any other changes in the source code, xtals with 2.048 or
2.097152 or 2.4576 MHz can be used and adapted with that constant.
If an xtal with 4.194304 MHz shall be used the calculated constant cSecDiv in the section
"Fixed and derived constants" has to be set to zero (which is 256), otherwise an error
message would occur.
Other xtals cannot be used without more complicated changes to the source code (con
cerning mainly the TC0 operation and the related interrupt service routine, while lcd.inc
operates correct with any possible frequency).

3.1.2 LCD size
The software only works correct if a 4-by-20 LCD is attached.
; LCD size:
.equ LcdLines = 4 ; Number of lines (1, 2, 4)
.equ LcdCols = 20 ; Number of characters per line (8..24)

Adaption to any other size requires major changes in the software.

3.1.3 LCD ports and portpins
The following constants concern the LCD ports for data and control and the used portpins
and are configured as the schematic shows.
Communication mode has been set as LcdWait = 1, which inserts wait cycles following
each control and data write. If you prefer the busy option, tie the R/W pin of the LCD to a
Port B portpin (any pin used for ISP would be ok), define the port and portpin constants
and change to LcdWait = 0 to configure R/W and busy operation mode of the LCD.

3.1.4 Conversion options
The constant LcdDecimal has to be defined because decimal conversion is used by several
instances of the software. As the light sensor's measurements are displayed in hex, and in
any case if you use debug options, the constant LcdHex also has to be defined. The values
to which those two constants are set do not matter.

3.2 Other properties to be adjusted
Please find the other propertied that can be adjusted in the main source code under the
section
; ================================================
;
C O N S T A N T S
T O
A D J U S T
; ================================================
;

3.2.1 DCF77 time constants
These constants, that start with cDcf and end with Time, in the source code are set as fol
lows:
; DCF signal durations
.equ cDcfIgnoreShortTime = 20 ; Ignore short pulses less than 20 ms
.equ cDcfMinTime = 50 ; Minimum signal duration time
.equ cDcfMaxZeroTime = 150 ; Maximum zero time
.equ cDcfMaxOneTime = 250 ; Maximum one time
.equ cDcfMinInactiveTime = 700 ; Minimum inactive time
.equ cDcfMaxInactiveTime = 1000 ; Maximum inactive time
.equ cDcfMin59Time = 1700 ; Minimum 59th second time
.equ cDcfMax59Time = 2000 ; Maximum 59th second time
.equ cDcfTimeOutTime = 2500 ; Time out
;

The meanings of those constants are described in the section DCF77 signal analysis. By
changes to these constants, differing properties of the DCF receiver can be adjusted.
Please note that the polarity of the DCF input pin (Active-High or Active-Low) is of no im
portance, because the signal analysis is insensitive for this and reacts correct with both
configurations.
When changing to those constants are made make sure that no overlapping can occur.

3.2.2 LCD backlight
The backlight of the LCD is controlled via
the timer TC2. TC2 is operated with a
prescaler of 64 as a PWM (f PWM =
3.276.800 / 64 / 256 = 200 Hz). Higher
values in the Compare port register in
crease the brightness of the backlight
LEDs.

The collector voltage of the photo tran
sistor (or diode) with a 100k resistor to
plus is measured for its voltage each
three seconds (adjustable with cBackPe
riodsTime). The AD converter of channel
ADC1 sums up 64 single values und inverts the MSB of the sum (the higher the ambience
brightness the smaller the collector voltage). This value is multiplied with the term (255 Minimum brightness) and the minimum brightness is added to the MSB of the result (mini
mum brightness adjustable with the constant cBackMin) and the result is written to the
compare port register OC2 of TC2.
From this the brightness of the LED backlight is depending from the ambient brightness

like shown in the diagram.
The constants
; LCD backlight brightness, 80 to 200
.equ cBackPeriodsTime = 3000 ; ms for renewal of backlight setting
.equ cBackMin = 10 ; Minimum brightness of the backlight
;

configure the backlight renewal sequence timing and the minimum of the backlight. Adjust
the minimum so that the LCD can be read in absolute darkness, if your LCD has different
backlight current properties. The cBackPeriodsTime is a matter of taste and avoids flicker
ing of the LCD.

3.2.3 S-Meter display
S-Meter display only makes sense if your DCF77 receiver provides such a signal and if the
signal strength voltage is attached to the analog input pin ADC2. If not, the display shows
S0.
The renewal property can be adjusted with the constant
;
; DCF77 signal strength display update
.equ cSigStrTime = 2000 ; Update LCD every milliseconds
;

If no signal is provided, set this constant to a very long time (maximum around 24,000,
depending from the clock frequency).
Which external voltage leads to which displayed S stage is defined in the table
SignalTable: in the source code. The table is by default adjusted to the described DCF77
receiver with the TCA440. If your receiver provides different voltages it can be useful to
change to debug mode dbgAdc with channel ADC2 to measure those voltages and to re
calculate the table values according to those readings. Note that the last entry in the table
is for a displayed value of S9+.

3.2.4 Debouncing of the keys
Debouncing of the keys is adjusted with the constant
;
; Key processing: delay on key cleared
.equ cKeyCntTime = 30 ; key must be inactive for 30 ms
;

A minimum of 20 ms and a maximum of 100 ms makes sense.

3.2.5 Alarm repetitions
The number of alarm repetitions is a matter of taste and can be adjusted with the con
stant
;
; Alarm repetitions
.equ cAlarmRepet = 3 ; Number of alarm repetions
;

3.2.6 Melody at restart
This property is only useful if you correct a currently stored or if you write a new melody.
The constant
;
; Default music play at startup
.equ cDefaultMelody = 8 ; Melody to be played at startup
;

allows you to select this so that you do not have to wait randomly to check if your melody
is correct.

4 Time and date
A comprehensive general description of time and
date in AVRs with assembler is here. The following is
based on that more general description and de
scribes the implementation in the DCF77 alarm clock
with the ATmega16.
Timer TC0 generates the seconds clock signal:
1. The prescaler of 64 divides the controller
clock to yield a timer tick of 51.2 kHz.
2. For 256 ticks 5 ms are required (200 Hz),
then a TC0 overflow interrupt occurs.
3. Within the overflow interrupt service routine
the register rSecDiv is down-counted from
200 to zero. If that reaches zero, the flag
bSec in the flag register rFlag is set, which
outside of the service routine increases time
and date and triggers other reactions as well.
Date and time are located in SRAM from address
sDateTime: on in binary format. The storage se
quence follows the display sequence:
0.
1.
2.
3.
4.
5.
6.

Weekday: 0 (Monday) to 6 (Sunday)
Day: 1 to 31
Month: 1 to 12
Year: 0 to 99
Hours: 0 to 23
Minutes: 0 to 59
Seconds: 0 to 59

The alarm time (hour, minute) is following in SRAM.
The routine IncSec: increases the time by one sec
ond and, if necessary, increases minutes, hours,
weekday, day, month and year. This also works with
out the DCF77 signal analysis, when the clock works
as an unsynchronized stand-alone device. Only
those components of time and date are renewed on
the LCD that have changed within the last second's
increase.
If the manual input of date and time is activated
(yellow key), the output of date and time is sup

pressed. Other than displayed, the increase of seconds (and synchronization with DCF77)
also works during manual input but just is not displayed (to be able to skip input and to
return back).
The second's increase is straight forward and as defined in the more general description in
the date/time attachment.

5 DCF77 signal analysis and synchronization
The implementation of the DCF77 signal analysis in this alarm clock here is rather sophis 
ticated, therefore it is described in any detail in this chapter.

5.1 DCF77 signal input
Recognition of those signals is performed via the INT0 external interrupt. Both edges
(Low==>High, High==>Low) lead to an interrupt.
In the interrupt service routine for INT0
1. the current state of the counter rDcfCntH:rDcfCntL is read, which is increased by
the timer TC0 each 5 ms,
2. checks if the counter is smaller than defined in the constant cDcfIgnoreShort, if that
is the case, the pulse is ignored and the time counter runs on,
3. the count in the register pair rDcfH:rDcfL is copyed,
4. the flag bDcf in the flag register rFlag is set, and
5. the time counter rDcfCntH:rDcfCntL is cleared.
Further processing of those recognized DCF77 signals follows outside the interrupt service
routine and within the routine DcfAct.

No matter which polarity the DCF77 signal from the receiver has the signal durations are
as follows:
1.
2.
3.
4.

a Zero bit is 100 ms,
a One bit is 200 ms,
a missing second (minute signal) is between 1,800 and 1.900 ms long,
the time between the end of a bit and the begin of the next bit is between 800 (One
bit) and 900 ms (One bit) long.

According to the selected timing when increasing the counter (5 ms) the 100 ms of a Zero
are associated with a count of 20, etc.etc.

Because the recognition of a Zero and One bit in the receiver's RC needs a certain time,
the above times have to include a tolerance. This tolerance has been selected in a wide
range, e. g. a Zero bit can be between 50 and 149 ms long, the count can be between 10
and 29 to avoid signal errors.

5.2 DCF77 signal checks
From that the accepted signal durations and the errors in the duration can be derived. Sig
nals with blue are correct and accepted, signals in red are errors.

5.3 DCF77 signal processing
That leads to the flow diagram of the DCF77
signal analysis. N are the counter values
(based on 5 ms).
Detected signal errors are assigned to error
reporting codes between s1 and s4, to be
displayed in line 3 of the LCD. If the DCF77
is inactive for longer than 2.5 s, the "S5
Signal missing" is displayed.
Correct signal durations for a Zero or One
bit shift a respective bit into the SRAM buf
fer sDcfBits: (by shifting all 64 bits from left
to right, with the 58th bit left adjusted in
the highest byte). The 59 collected bits per
minute are analyzed when the missing sec
ond pulse is recognized.
The analysis first checks if
1. no signal errors (s1 to s5) occurred,
and
2. 59 bits have been received.
In the second case the number of received
bits is displayed in line 3 of the LCD. In both
cases further analysis of the received bits is
not continued.

5.4 DCF77 information conversion
5.4.1 DCF77 minutes and hours
If the 59 signals are fine, the first bit of the minute (DCF bit 1 = bit 5 in byte 0 of the
SRAM buffer in sDcfBits:) is checked: it has to be zero, otherwise error "E0" is reported
and further analysis is skipped.

Then the four
bits of minute
ones
are
checked. Tho
se are located
in the bits 2 to
5 at the SRAM
address sDcf
Bits+3. If tho
se are larger
than nine, er
ror "MO" is
thrown
and
further execu
tion is skip
ped.
Then the minute tens (three bits) are isolated from bit 6 and 7 of byte 3 and bit 0 in byte
4, by using two LSL and ROL instructions, followed by an ANDI 0x07. If the minute tens
exceed five error "MT" is reported and further execution is skipped. If not, the tens are
multiplied by ten and added to the minute ones.
Then the parity of the minute bits (bits 2 to 7 in byte 3 plus bits 0 and 1 in byte 4) is cal
culated by counting the ones in those bits. If the sum is odd (parity is not zero) the "PM"
error is reported and further analysis is stopped. If the sum of ones is even, the received
minutes are stored.
Similarly the hours are checked for
1. the ones exceeding 9, and
2. the tens exceeding 2, and
3. the number of ones including the parity bit being odd.
If none of these conditions are detected the hour byte is stored and the date information
is analyzed.

5.4.2 DCF77 date
The date informa
tions are distrib
uted over three
bytes in the DCF
bits. Error checking
is similar, as shown
here. The parity bit
in that case covers
all 23 bits of the
date
information.
The weekday is en
coded as 1 for Mon
day to 7 for Sun
day.
If all error checking is absolved and all information is collected, the date and time is syn 
chronized with DCF77. The decoded date and time is copied to the buffer DateTime: (if
manual input is not active) and "ok<" is displayed in line 3 of the LCD. The seconds
prescaler is set to 200 and the seconds are cleared. The displayed date and time on the
LCD is updated.

6 Keys
The three keys are read within
the interrupt service routine of
timer TC2 each 5 ms. If one of
the keys is pressed the flag
bKey in the flag register rFlag is
set and the debouncing register
rKeyCnt is set to 30 ms (can be
adjusted with the constant cK
eyCntTime). Further key events
are accepted if this register is at zero.
If key input is deactivated and if alarm is off (sKeyMode = 0) then pressing the
• red key activates the alarm function (sKeyMode = 1),
• yellow key activates manual date and time (sKeyMode = 2), and the
• white key activates adjusting the alarm time (sKeyMode = 3).
If the alarm is on (displayed on line 2 of the LCD) the
• red key deactivates the alarm,
• yellow key increases the alarm time by 5 minutes, and
• white key increases the alarm time by 10 minutes.
Those increases are set only temporarily, by deactivating the alarm the original alarm time
is restored.
In both manual adjustment modes (date/time, alarm) the currently active input position
blinks. Even though only one character blinks the ones and tens are adjusted in one step.
In those modes the
• red key moves the input cursor left by one location, if the input position is the most
left position (date/time: weekday, alarm time: alarm hour) the input is skipped and
the original value is restored,
• yellow key restores the original (previous) value and moves one position to the
right,
• white key moves the input position one right, if the input position is the last one
(date/time: seconds, alarm time: minutes) the changed data is copied to the re
spective SRAM buffer, if the alarm time is finally adjusted the alarm is set on.
Changes at the activated input position are made by moving the potentiometer position
left (decreasing values) or right (increasing values).
Page top Structure Debug Properties Time/date DCF77 Keys Music

7 Music
Tone generation works with the 16 bit timer TC1. The prescaler is set to 1. The tone fre 
quency is determined with the compare port register A. If the TC1 count reaches the com 
pare value
• the timer restarts (CTC mode),
• either toggles the output pin OC1A (tone on) or clears OC1A (tone off), and
• triggers the Compare-A-interrupt (if tone generation is active).
If tone generation is activated the compare interrupt service routine decreases a 16 bit
counter rDurH:rDurL. This counter counts the number of cycles and determines the dura
tion of the tone or pause. If the number of cycles reaches zero, the next CTC value is read

from the SRAM buffer. If that value is 0xFFFF the speaker output is cleared on compare
match (COM1A1 to clear), otherwise the output is set to toggle (COM1A1 to toggle). The
CTC value is then written to the compare A port register. Following is the reading of the
counter value in rDurH:rDurL from SRAM. If this value is zero, the melody ends and the
timer TC1 interrupt is disabled.
If the alarm is active (sKeyMode = 1) each minute change triggers comparison of the
alarm time with the current time. If equal, one of the stored melodies in the flash memory
is selected and converted to CTC and counter values in the SRAM buffer (routine Music
Convert:). The number of repetitions in sAlarmRepeat: is set to the adjusted constant
cAlarmRepeat:. In each following minute the same melody is restarted (routine Music
Play:) until either the red key decativates the alarm (sKeyMode = 0) or sAlarmRepeat
reaches zero.
The melodies prior to their conversion are encoded in flash memory tables as follows.
• Musical notes are defined with names starting with 'n' and assigning a tone (C, D,
E, F, G, A and H) and its octave (German notation: m, none, 1 to 7) with a con
stant, e.g.
1. nCm = 9 with 65.4 Hz,
2. nc = 16 with 130.8 Hz,
3. nc1 = 23 with 261.6 Hz,
4. nc2 = 30 with 523.25 Hz,
5. nc3 = 37 with 1,046.5 Hz,
6. nc4 = 44 with 2,093 Hz,
7. nc5 = 51 with 4,186 Hz,
8. nc6 = 58 with 8,372 Hz, und
9. nc7 = 65 with 16,744 Hz.,
Additionally a note p is defined as a pause. See the complete note definitions in the
source code include file under NotesNameTable:. Those 70 constants ease writing
melody tables.
• In the melody table each note and pause constant has to be followed by a duration
constant. The lower nibble of the byte says how often the duration has to be divid
ed by two:
0. no division, duration one second,
1. divide by two, duration half second,
2. divide by four, duration 0.25 seconds,
3. divide by eight, duration 0.125 seconds,
4. divide by sixteen, duration 0.0625 seconds.
Note that high division rates with low tones might not yield exact times due to
rounding.
The upper nibble determines how often to add the divided cycle count. By that it is
possible to encode durations of 3/4 or 5/8.
The constants d1, d2, d4, d8 and d16 define two-divided durations, d38 and d316
encode three octa and three sixteenth durations.
• The melodies are encoded in the flash, the table MelodyTab: points to the beginning
of those melodies (address multiplied by two).
• For each musical note the frequency is assigned as a constant (see source code in
the include file under NoteFrequencyTable:).
• Each note frequency is converted to a table that consists of
1. a 16 bit CTC value, and
2. a 16 bit counter value for the duration in cycles per second.
See this table in the source code under NotesTimerTable:. With this table the
melody consisting of note constants can be converted to CTC compare and duration
words.

1 The main program dcf77_clock_m16_v5_en.asm
Link to this code in asm format
;
; ***********************************************
; * ATmega16 clock with LCD and DCF77-Synchron- *
; * nization, Version 5.0 as of August 2018
*
; * (C)2018 by G.Schmidt, avr-asm-tutorial.net *
; ***********************************************
;
; Include file for target controller type
.NOLIST
.INCLUDE "m16def.inc" ; Header file for ATmega16
.LIST
;
; ===================================
;
S W I T C H E S
; ===================================
;
; Display language 0 = German, 1 = English
.equ LangEN = 1 ; Select language
;
; ===================================
; D E B U G G E R
S W I T C H E S
; ===================================
;
; All debug switches zero for final version
;

; 1. Debug the LCD
.equ dbgLcd = 0 ; Debug the LCD
;
; 2. Debug the ADC
.equ dbgAdc = 0 ; Debug the ADC
.equ debugAdcChannel = 0 ; Select channel 0, 1 or 2
;
; 3. Debug the keys
.equ dbgKey = 0 ; Debug the keys attached
.equ dbgKeyState = 0 ; Debug the key's state on line 3
;
; 4. Debug the speaker
.equ dbgSpk = 0 ; Debug the speaker attached
;
; 5. Debug the DCF77 signal
.equ dbgDcfDur = 0 ; Debug the DCF77 signal input duration
.equ dbgDcfSig = 0 ; Debug DCF77 signal errors
.equ dbgDcfBits = 0 ; Debug incoming DCF77 bits
.equ dbgDcfRcvd = 0 ; Debug received DCF77 data
;
; 6. Debug music play
.equ dbgMusic = 0 ; Debug the play process of music
.equ dbgPlayStat = 0 ; Debug the status of TC1
;
;
; ============================================
;
H A R D W A R E
I N F O R M A T I O N
; ============================================
;
;
ATmega16__________
;
/
|
;
Sw-W o--|PB0
ADC0|--o Vpot
;
Sw-Y o--|PB1
ADC1|--o Vopto
;
Sw-R o--|PB2
PA2|--o DCF-signal strength
;
LCD-E o--|PB3
PA3|--o NC
;
LCD-RS o--|PB4
PA4|--o NC
; ISP6-MOSI o--|MOSI
PA5|--o NC
; ISP6-MISO o--|MISO
PA6|--o NC
;
ISP6-SCK o--|SCK
PA7|--o NC
;
RESET o--|RST
AREF|--o AREF
;
VCC o--|VCC
GND|--o GND
;
GND o--|GND
AVCC|--o AVCC
;
XTAL2 o--|XTAL2
PC7|--o LCD-D7
;
XTAL1 o--|XTAL1
PC6|--o LCD-D6
;
NC o--|PD0
PC5|--o LCD-D5
;
NC o--|PD1
PC4|--o LCD-D4
; INT0/DCF77 o--|INT0
PC3|--o LCD-D3
;
NC o--|PD3
PC2|--o LCD-D2
;
NC o--|PD4
PC1|--o LCD-D1
;
LSP o--|OC1A
PC0|--o LCD-D0
;
NC o--|PD6
OC2|--o LCD-LED-K
;
|___________|
;
; ============================================
;
P O R T S
A N D
P I N S
; ============================================
;
; Key ports
.equ pKeyO = PORTB ; Switch port output
.equ pKeyD = DDRB ; Switch direction
.equ pKeyI = PINB ; Switch input
.equ bKeyWO = PORTB0 ; White key output pin
.equ bKeyWD = DDB0 ; White key direction pin
.equ bKeyWI = PINB0 ; White key input pin

.equ bKeyYO = PORTB1 ; Yellow key output pin
.equ bKeyYD = DDB1 ; Yellow key direction pin
.equ bKeyYI = PINB1 ; Yellow key input pin
.equ bKeyRO = PORTB2 ; Red key output pin
.equ bKeyRD = DDB2 ; Red key direction pin
.equ bKeyRI = PINB2 ; Red key input pin
.equ cKeyMO = (1<<bKeyWO)|(1<<bKeyYO)|(1<<bKeyRO) ; Switch mask key outputs
.equ cKeyMI =(1<<bKeyWI)|(1<<bKeyYI)|(1<<bKeyRI) ; Input pin mask
;
; LCD ports
;
(see lcd configuration section)
;
; DCF77 ports
.EQU pDcfO = PORTD ; DCF77 Receiver Port Output
.EQU pDcfD = DDRD ; DCF77 Receiver Direction Port
.EQU pDcfI = PIND ; DCF77 Receiver Input Port
.EQU bDcfD = DDD2 ; DCF77 Reciver portpin Direction
.EQU bDcfI = PIND2 ; DCF77 Receiver portpin Input
;
; Speaker port
.equ pSpkO = PORTD ; Speaker output port
.equ pSpkD = DDRD ; Speaker direction port
.equ bSpkO = PORTD5 ; Speaker portpin output
.equ bSpkD = DDD5 ; Speaker portpin direction
;
; LED port
.EQU pLedO = PORTD ; LED cathode output port
.EQU pLedD = DDRD ; LED cathode direction port
.EQU bLedO = PORTD7 ; LED cathode output portpin
.EQU bLedD = DDD7 ; LED cathode direction portpin
;
; ADC port
.EQU pAdcO = PORTA ; ADC output port
.EQU pAdcD = DDRA ; ADC direction port
.EQU bAdcP = 0 ; ADC poti portbit
.EQU bAdcO = 1 ; ADC brightness portbit
.equ bAdcS = 2 ; DCF77 receiver signal strength
;
; ================================================
;
L C D
C O N F I G U R A T I O N
; ================================================
;
; Standard parameter set of properties/definitions
.equ Clock = 3276800 ; X-tal clock
; LCD size:
.equ LcdLines = 4 ; Number of lines (1, 2, 4)
.equ LcdCols = 20 ; Number of characters per line (8..24)
; LCD bus interface
.equ LcdBits = 8 ; Bus size (4 or 8)
; If 4 bit bus:
;.equ Lcd4High = 1 ; Bus nibble (1=Upper, 0=Lower)
.equ LcdWait = 1 ; Access mode (0 with busy, 1 with delay loops)
; LCD data ports
.equ pLcdDO = PORTC ; Data output port
.equ pLcdDD = DDRC ; Data direction port
; LCD control ports und pins
.equ pLcdCEO = PORTB ; Control E output port
.equ bLcdCEO = PORTB3 ; Control E output portpin
.equ pLcdCED = DDRB ; Control E direction port
.equ bLcdCED = DDB3 ; Control E direction portpin
.equ pLcdCRSO = PORTB ; Control RS output port
.equ bLcdCRSO = PORTB4 ; Control RS output portpin
.equ pLcdCRSD = DDRB ; Control RS direction port
.equ bLcdCRSD = DDB4 ; Control RS direction portpin

; If LcdWait = 0:
;.equ pLcdDI = PINA ; Data input port
;.equ pLcdCRWO = PORTB ; Control RW output port
;.equ bLcdCRWO = PORTB2 ; Control RW output portpin
;.equ pLcdCRWD = DDRB ; Control RW direction port
;.equ bLcdCRWD = DDB2 ; Control RW direction portpin
; If you need binary to decimal conversion:
.equ LcdDecimal = 1 ; If defined: include those routines
; If you need binary to hexadecimal conversion:
.equ LcdHex = 1 ; If defined: include those routines
; If simulation in the SRAM is desired:
;.equ avr_sim = 1 ; 1=Simulate, 0 or undefined=Do not simulate
;
; ================================================
;
C O N S T A N T S
T O
A D J U S T
; ================================================
;
; Controller clock X-tal frequency
;
Defined under LCD configuration:
;
.equ Clock = 3276800 ; X-tal clock
;
; DCF signal durations
.equ cDcfIgnoreShortTime = 20 ; Ignore short pulses less than 20 ms
.equ cDcfMinTime = 50 ; Minimum signal duration time
.equ cDcfMaxZeroTime = 150 ; Maximum zero time
.equ cDcfMaxOneTime = 250 ; Maximum one time
.equ cDcfMinInactiveTime = 700 ; Minimum inactive time
.equ cDcfMaxInactiveTime = 1000 ; Maximum inactive time
.equ cDcfMin59Time = 1700 ; Minimum 59th second time
.equ cDcfMax59Time = 2000 ; Maximum 59th second time
.equ cDcfTimeOutTime = 2500 ; Time out
;
; LCD backlight brightness, 80 to 200
.equ cBackPeriodsTime = 3000 ; ms for renewal of backlight setting
.equ cBackMin = 10 ; Minimum brightness of the backlight
;
; DCF77 signal strength display update
.equ cSigStrTime = 2000 ; Update LCD every milliseconds
;
; Key processing: delay on key cleared
.equ cKeyCntTime = 30 ; key must be inactive for 30 ms
;
; Alarm repetitions
.equ cAlarmRepet = 3 ; Number of alarm repetions
;
; Default music play at startup
.equ cDefaultMelody = 8 ; Melody to be played at startup
;
; ===============================================
; F I X E D + D E R I V E D
C O N S T A N T S
; ===============================================
;
; Constants for TC0 operation
.equ cTc0Presc = 64 ; TC0 prescaler
.equ cTc0Divider = 256 ; TC0 8 bit divider
.equ cSecDiv = clock / cTc0Presc / cTc0Divider ; Divider until second over
.if cSecDiv>255
.error "cSecDiv out of range!"
.endif
;
; Constants for TC1 operation (audio generation)
.equ cTc1Presc = 1 ; Default prescaler
;
; Constants for TC2 operation (backlight PWM)

.equ cTc2Presc = 64 ; TC2 LCD backlight PWM
.equ cTc2Cnt = Clock/cTc2Presc/256 ; 200 Interrupts per second
.equ cBackMult = (255-cBackMin) ; Multiplier for opto sensor
;
; DCF signal durations
.equ cPulse=(cTc0Presc*cTc0Divider*1000+clock/2)/clock ; Pulse in ms
.equ cDcfIgnoreShort = cDcfIgnoreShortTime/cPulse ; Ignore shorter pulses
.equ cDcfMin = cDcfMinTime/cPulse ; Minimum pulse count
.equ cDcfMaxZero = cDcfMaxZeroTime/cPulse ; Maximum zero pulse
.equ cDcfMaxOne = cDcfMaxOneTime/cPulse ; Maximum one pulse
.equ cDcfMinInactive = cDcfMinInactiveTime/cPulse ; Minimum inactive time
.equ cDcfMaxInactive = cDcfMaxInactiveTime/cPulse ; Maximum inactive time
.equ cDcfMin59 = cDcfMin59Time/cPulse ; Minimum 59th second pulse
.equ cDcfMax59 = cDcfMax59Time/cPulse ; Maximum 59th second pulse
.equ cDcfTimeOut = cDcfTimeOutTime/cPulse ; Time out pulse
;
; Key debouncing
.equ cKeyCnt = cKeyCntTime/cPulse ; Pulses over which key has to be inactive
;
; LCD backlight update
.equ cAdcPulse = 1000*128*13*64*3/clock ; Duration of ADC rounds in ms
.equ cBackPeriods = cBackPeriodsTime/cAdcPulse ; Number of periods to update
.if cBackPeriods > 255
.error "cBackPeriodsTime too long!"
.endif
;
; DCF77 signal strength display update
.equ cSigStr = cSigStrTime/cAdcPulse ; Number of periods for signal strength
update
;
; Debugging active
.set fDbg=(dbgLcd>0)||(dbgAdc>0)||(dbgKey>0)||(dbgKeyState>0)
.set fDbg=fDbg||(dbgSpk>0)||(dbgDcfDur>0)||(dbgDcfSig>0)
.set fDbg=fDbg||(dbgDcfBits>0)||(dbgDcfRcvd>0)||(dbgMusic>0)
.set fDbg=fDbg||(dbgPlayStat>0)
;
; ================================================
;
T I M I N G
A N D
T I M E R S
; ================================================
;
; TC0: 8-Bit-Timer for measuring
;
a) Seconds pulse advancing the clock
;
b) DCF pulse length measurements
;
;
Crystal osc ==> TC0-Presc ==> TC0-Tick ==>
;
f 3276800 Hz
64
51,200 Hz
;
t 305.1758 ns
19.53125 us
;
TC0-Overflow (256)
;
f
200 Hz
;
t
5 ms
;
a) Seconds pulse ==> downcount second divider,
;
if zero restart and set flag
;
b) DCF77: Increase DCF77 counter,
;
if excxeeding maximum count set flag
;
; TC1: 16-Bit-Timer for tone generation
; Crystal ==> TC1-Presc ==> TC1-CTC ==> 25 Hz to
; 3276800
1
0..65535
1.6384 MHz
;
CTC-Compare A int:
;
Decrease R25:R24, if zero:
;
Read next ICR value, if MSB and LSB = 0:
;
end of music, switch to speaker off
;
if MSB = 0xF4: turn speaker off for pause
;
if not: turn speaker on for tone

;
If not zero:
;
read duration word and transfer to R25:R24
;
; TC2: 8-Bit-Timer as PWM for LCD background light
;
and key processing
;
Crystal ==> TC2-Presc ==> TC2-OvFlw ==> OvFlw-Int
;
3276800 ==>
64
==>
256
200 Hz
;
Read key input, if
;
no key pressed: if rKeyCnt>0
;
downcount rKeyCnt, if zero clear flag rKeyAct
;
any one key pressed:
;
if bKeyAct flag = clear set flag rKeyAct
;
= set reload rKeyCnt
;
; ADC: Measuring three analogue input channels
;
Crystal ==> ADC-Presc ==> ADC-1m ==> ADC-64m ==> 3 Ch
;
3276800
128
13
64
3
;
= 10.256 cs/s or 97 ms per round
;
; ============================================
;
R E G I S T E R
D E F I N I T I O N S
; ============================================
;
; R0, R1 used for calculations and diverse purposes
; free: R2, R3
.def rMelody = R4 ; Melody select, set by key events
.def rSigStr = R5 ; Signal strength update counter
.def rBackCnt = R6 ; LCD backlight update counter
.def rKeyCnt = R7 ; Key pressed counter
.def rDcfBitCnt = R8 ; Bit counter DCF
.def rDcfErr = R9 ; DCF77 signal error counter
.def rAdc = R10 ; ADC MSB result
.def rAdcL = R11 ; ADC sum LSB
.def rAdcH = R12 ; dto., MSB
.def rDcfL = R13 ; Last pulse/pause length DCF, LSB
.def rDcfH = R14 ; dto., MSB
.def rSreg = R15 ; Saving SREG in Interrupts
.def rmp = R16 ; multi purpose register
.def rimp = R17 ; multi purpose, interrupts
.def rFlag = R18 ; flags
.equ bSec = 0 ; Second over
.equ bDcf = 1 ; Active DCF77 pulse flag
.equ bDcfTO = 2 ; DCF77 signal time out
.equ bAdc = 3 ; ADC 64 measurements complete
.equ bKey = 4 ; A key has been pressed
.equ bKeyAct = 5 ; Key input is active
.equ bMinute = 6 ; Minute changed, check alarm
.def rSecDiv = R19 ; down counter seconds
.def rDcfCntL = R20 ; DCF77 pulse duration counter LSB
.def rDcfCntH = R21 ; dto., MSB
.def rAdcC = R22 ; ADC counter
; free: R23
.def rDurL = R24 ; Duration of tones, LSB
.def rDurH = R25 ; dto., MSB
; used: R27:R26 = X for various purposes
; used: R29:R28 = Y for music play
; used: R31:R30 = Z for various purposes
;
; ============================================
;
S R A M
D E F I N I T I O N S
; ============================================
;
.DSEG
.ORG SRAM_START

; Real date and time for display
sDateTime:
sWeekday:
.byte 1
sDay:
.byte 1
sMonth:
.byte 1
sYear:
.byte 1
sHour:
.byte 1
sMin:
.byte 1
sSec:
.byte 1
sDateTimeEnd:
sAlrmHr:
.byte 1
sAlrmMin:
.byte 1
;
; DCF77 bits
sDcfBits:
.byte 8
;
; DCF77 received parts
sDcfRcv:
.byte 1 ; Weekday 0..6
.byte 1 ; Day 1..31
.byte 1 ; Month 1..12
.byte 1 ; Year 0..99
.byte 1 ; Hour 0..23
.byte 1 ; Minute 0..59
.byte 1 ; Seconds 0..59
;
; Process key events
sKeyMode:
.byte 1 ; 0=off/normal
;
1=alarm activated
;
2=input date/time
;
3=input alarm time
;
sKeyPos:
.byte 1 ; Mode 1 Mode 2 Mode 3
;
min=4
min=0
min=7
;
max=6
max=3
max=8
;
sKeyCnt:
.byte 1 ; ADC period counter for key actions
;
sAlarm:
.byte 2 ; Stored alarm time
;
sInputData:
.byte 9 ; Date (weekday, day, month, year),
; time (hour, min, sec),
; alarm time (hour, min)
;
sAlarmRepeat:
.byte 1 ; Repeat alarm every minute
;
sMelody:
.byte 500 ; Buffer for melody to play

sMelodyEnd:
;
; ==============================================
;
R E S E T
A N D
I N T
V E K T O R S
; ==============================================
;
.CSEG
.ORG $0000
jmp Main ; Reset vector
jmp Int0Isr ; INT0 Vector
reti ; INT1 Vector
nop
reti ; TC2COMP Vector
nop
jmp Tc2OvfIsr ; TC2OVF Vector
reti ; TC1CAPT Vector
nop
jmp Tc1CmpAIsr ; TC1 COMPA Vector
reti ; TC1COMPB Vector
nop
reti ; TC1OVF Vector
nop
jmp Tc0OvfIsr ; TC0OVF Vector
reti ; SPI-STC Vector
nop
reti ; USART-RXC Vector
nop
reti ; USART-UDRE Vector
nop
reti ; USART-TXC Vector
nop
jmp AdcIsr ; ADC Vector
reti ; EE_RDY Vector
nop
reti ; ANA_COMP Vector
nop
reti ; TWI Vector
nop
reti ; INT2 Vector
nop
reti ; TC0_COMP Vector
nop
reti ; SPM_RDY Vector
nop
;
; ==========================================
;
I N T E R R U P T
S E R V I C E
; ==========================================
;
; DCF77 receiver pulse input
Int0Isr:
in rSreg,SREG ; Save SREG
mov rimp,rDcfCntL ; Suppress zero counts
cpi rimp,cDcfIgnoreShort ; Shorter than minimum pulses?
brcs Int0IsrRet ; Yes, ignore
; Valid DCF signal detected
mov rDcfL,rDcfCntL ; Copy counter LSB
mov rDcfH,rDcfCntH ; dto., MSB
sbr rFlag,1<<bDcf ; Set flag
clr rDcfCntL ; Clear LSB counter
clr rDcfCntH ; dto., MSB
Int0IsrRet:
out SREG,rSreg ; Restore SREG
reti

;
; TC0 overflow int, DCF77 pulse width timer
Tc0OvfIsr:
in rSreg,SREG ; Save Sreg
dec rSecDiv ; Decrease seconds divider
brne Tc0OvfIsr1 ; Not zero
ldi rSecDiv,cSecDiv ; Restart counter
sbr rFlag,1<<bSec ; Set seconds flag
Tc0OvfIsr1:
inc rDcfCntL ; Increase DCF pulse counter
brne Tc0OvfIsr2 ; No carry to MSB
inc rDcfCntH ; Carry to MSB
Tc0OvfIsr2:
cpi rDcfCntH,High(cDcfTimeOut) ; MSB at time out?
brcs Tc0OvfIsr3 ; No
cpi rDcfCntL,Low(cDcfTimeOut) ; LSB at time out?
brcs Tc0OvfIsr3 ; No
sbr rFlag,1<<bDcfTO ; Set time-out-flag
clr rDcfCntL ; Restart DCF counter
clr rDcfCntH
Tc0OvfIsr3:
out SREG,rSreg ; Restore SREG
reti
;
; TC1 compare match A interrupt, playing music
Tc1CmpAIsr:
in rSreg,SREG ; Save SREG
sbiw rDurL,1 ; Downcount duration counter
brne Tc1CmpAIsrRet
ld rimp,Y+ ; Read MSB first
out OCR1AH,rimp ; To MSB compare A
cpi rimp,0xFF ; MSB speak off?
ld rimp,Y+ ; Read LSB next
out OCR1AL,rimp ; To LSB compare A
brne Tc1CmpAIsr1 ; Sprecher not off
cpi rimp,0xFF ; LSB speaker off?
brne Tc1CmpAIsr1 ; Do not turn speaker off
ldi rimp,(1<<COM1A1) ; Speaker off
rjmp Tc1CmpAIsr2
Tc1CmpAIsr1:
ldi rimp,(1<<COM1A0) ; OCR1A to toggle
Tc1CmpAIsr2:
out TCCR1A,rimp ; Set OCR1A toggle/clear
ld rDurH,Y+ ; Read MSB duration
ld rDurL,Y+ ; Read LSB duration
mov rimp,rDurH ; Test for 0x0000
or rimp,rDurL
brne Tc1CmpAIsrRet ; Continue music lay
; Music play off
ldi rimp,1<<COM1A1 ; OCR1A to clear
out TCCR1A,rimp
ldi rimp,(1<<TOIE0)|(1<<TOIE2) ; Disable TC1 Int
out TIMSK,rimp
ldi rimp,High(clock / 2000 - 1) ; CTC to 1 kHz
out OCR1AH,rimp
ldi rimp,Low(clock / 2000 - 1)
out OCR1AL,rimp
Tc1CmpAIsrRet:
out SREG,rSreg ; Stelle SREG wieder her
reti
;
; TC2 overflow int, Backligth PWM, key rocessing
Tc2OvfIsr:
in rSreg,SREG ; Save SREG

in rimp,pKeyI ; Read key input
andi rimp,cKeyMI ; Isolate keys in input port
cpi rimp,0x07 ; All keys clear?
brne Tc2OvfIsrKey ; No, process key
tst rKeyCnt ; Key count clear?
breq Tc2OvfIsrRet ; Yes, ignore
dec rKeyCnt ; Decrease key counter
brne Tc2OvfIsrRet ; Not at zero
cbr rFlag,1<<bKeyAct ; Clear key active flag
rjmp Tc2OvfIsrRet
Tc2OvfIsrKey:
sbrs rFlag,bKeyAct ; Flag bKeyAct already set?
sbr rFlag,(1<<bKeyAct)|(1<<bKey) ; Set key pressed flag
ldi rimp,cKeyCnt ; Set key count to default delay
mov rKeyCnt,rimp
in rMelody,TCNT1L ; Read counter 1 as random generator
Tc2OvfIsrRet:
out SREG,rSreg ; Restore SREG
reti
;
; ADC complete interrupt service routine
AdcIsr:
in rSreg,SREG ; Save SREG
in rimp,ADCL ; Read LSB result
add rAdcL,rimp ; Add to sum
in rimp,ADCH ; Read MSB result
adc rAdcH,rimp ; Add to sum with carry
dec rAdcC ; Decrease counter
breq AdcIsr1 ; Counter is zero
ldi rimp,(1<<ADEN)|(1<<ADSC)|(1<<ADIE)|(1<<ADPS2)|(1<<ADPS1)|(ADPS0)
out ADCSRA,rimp ; Start next conversion
out SREG,rSreg ; Restore SREG
reti
AdcIsr1:
mov rAdc,rAdcH ; Copy MSB
clr rAdcL ; Clear sum
clr rAdcH
sbr rFlag,1<<bAdc ; Set ADC flag
ldi rAdcC,64 ; Restart counter
out SREG,rSreg ; Restore SREG
reti
;
; ============================================
;
M A I N P R O G R A M
I N I T
; ============================================
;
Main:
; Init Stack
ldi rmp, HIGH(RAMEND) ; Init MSB stack
out SPH,rmp
ldi rmp, LOW(RAMEND) ; Init LSB stack
out SPL,rmp
; Init the key input pins
cbi pKeyD,bKeyWD ; White key as input
sbi pKeyO,bKeyWO ; White key pull up on
cbi pKeyD,bKeyYD ; Yellow key as input
sbi pKeyO,bKeyYO ; Yellow key pull up on
cbi pKeyD,bKeyRD ; Red key as input
sbi pKeyO,bKeyRO ; Red key pull up on
; Init TC2 as LCD Backlight PWM
sbi pLedD,bLedD ; set portbit as output
ldi rmp,0x80 ; Half light
out OCR2,rmp
ldi rmp,(1<<WGM21)|(1<<WGM20)|(1<<COM21)|(1<<COM20)|(1<<CS22) ; Fast PWM,

presc=64
out TCCR2,rmp
; Previous version: Init the LCD-Ports and LCD
;cbi pLcdCO,bLcdE
;sbi pLcdCD,bLcdE ; Enable Output Ausgang
;sbi pLcdCD,bLcdRs ; RS output pin as output
;cbi pLcdCO,bLcdRs ; Clear RS
;sbi pLcdCD,bLcdRw ; R/W output pin as output
;cbi pLcdCO,bLcdRw ; R/W to low
;ldi rmp,0xFF ; LCD data port as output
;out pLcdDD,rmp
call LcdInit ; Init the LCD
ldi ZH,High(2*CodeChars) ; Special characters
ldi ZL,Low(2*CodeChars)
call LcdSpec
ldi ZH,0
ldi ZL,0
call LcdPos
ldi ZH,High(2*LcdTxtInit1)
ldi ZL,Low(2*LcdTxtInit1)
call LcdText
; If any debugging enabled
.if fDbg
call DebugActive ; Outut message on LCD
.endif
; If debugging the LCD: stop here
.if dbgLcd == 1
LoopLcd:
rjmp LoopLcd ; Stop operation
.endif
; Delay for LCD operation mask
ldi rmp,100
Delay:
call LcdWait50ms
dec rmp
brne Delay
; If debugging the ADC: perform measurements
.if dbgAdc == 1
jmp DebugAdc ; Perform measurements
.endif
; If debugging the keys: display key inputs
.if dbgKey == 1
jmp DebugKey ; Display key state
.endif
; If debugging the speaker: play tone
.if dbgSpk == 1
jmp DebugSpeaker ; Play tone
.endif
; If debugging DCF77 signals: display DCF input signals
.if dbgDcfDur == 1 ; Debug DCF77 signal duration
call DebugDcfDur ; Display signals
.endif
; Start normal screen frame
ldi ZH,0 ; Home position of the LCD
ldi ZL,0
call LcdPos
ldi ZH,High(2*LcdTxtFrame) ; Point to text frame
ldi ZL,Low(2*LcdTxtFrame)
call LcdText ; LCD to output mask
call LcdNoBlink ; No blinking
; Start AD conversion
clr rAdcL ; Clear ADC sum
clr rAdcH
ldi rAdcC,64 ; Perform 64 measurements

ldi rmp,1<<REFS0 ; Start with channel 0
out ADMUX,rmp
ldi rmp,(1<<ADEN)|(1<<ADSC)|(1<<ADIE)|(1<<ADPS2)|(1<<ADPS1)|(ADPS0)
out ADCSRA,rmp ; Start first conversion
; Init the SRAM content
rcall InitDT
lds rmp,sDateTime+7 ; Copy alarm time to sAlarm
sts sAlarm,rmp
lds rmp,sDateTime+8
sts sAlarm+1,rmp
rcall SetAlarmOff ; Clear alarm
rcall DateTimeLcd
rcall DcfClearBuffer ; Init the DCF receiver buffer
; Init flags and software counters
clr rFlag ; Clear all flags
clr rDcfErr ; Clear DCF77 error counter
clr rDcfBitCnt ; Clear DCF77 bit counter
ldi rmp,cBackPeriods ; Backlight update period
mov rBackCnt,rmp
clr rmp ; Init key input mode
sts sKeyMode,rmp ; Key input mode off
sts sAlarmRepeat,rmp ; Alarm repeat counter
rcall DisplMenue ; Init menue line
; Init TC0 as time counter DCF77 signal duration timer
ldi rSecDiv,cSecDiv ; Seconds down counter
ldi rmp,(1<<CS01)|(1<<CS00) ; prescaler 64
out TCCR0,rmp
; Init TC1 as music player, CTC mode
ldi rmp,High(clock/2000-1) ; Init compare A
out OCR1AH,rmp
ldi rmp,Low(clock/2000-1)
out OCR1AL,rmp
ldi rmp,(1<<COM1A1) ; Compare match clear
out TCCR1A,rmp
ldi rmp,(1<<WGM12)|(1<<CS10) ; CTC-A, prescaler=1
out TCCR1B,rmp
sbi pSpkD,bSpkD ; OCR1A as output
; Init Music output
ldi rmp,cDefaultMelody ; Melody to be played at startup
rcall MusicConvert ; Play melody
; Enable timer interrupts
ldi rmp,(1<<TOIE2)|(1<<OCIE1A)|(1<<TOIE0) ; Timer Ints TC0 and TC2
out TIMSK,rmp
; Sleep mode, external interrupts
ldi rmp,(1<<SE)|(1<<ISC00) ; Enable sleep and INT0
out MCUCR,rmp
; External INT0 enable
ldi rmp,1<<INT0
out GICR,rmp
sei ; Enable interrupts
;
; ==============================
;
P R O G R A M
L O O P
; ==============================
;
Loop:
sleep ; Go to sleep
nop ; Dummy for wake-up
sbrc rFlag,bDcf ; DCF77 active pulse?
rcall DcfAct ; Active pulse DCF77
sbrc rFlag,bDcfTO ; DCF77 time-out?
rcall DcfTimeOut ; Time out of the DCF signal
sbrc rFlag,bSec ; second is over?
rcall IncSec ; increase seconds

sbrc rFlag,bMinute ; Check minute for alarm?
rcall CheckMinute ; Yes check minute
sbrc rFlag,bKey ; Has a key been pressed?
rcall KeyPress ; Process the pressed key
sbrc rFlag,bAdc ; ADC complete?
rcall AdcCmplt ; ADC measurements complete
; If key state to be debugged
.if dbgKeyState == 1
call KeyState ; Display key state
.endif
rjmp loop ; Back to Loop
;
; ====================================
; P R O C E S S
D C F P U L S E S
; ====================================
;
; Active pulse DCF77
DcfAct:
cbr rFlag,1<<bDcf ; clear flag
; If debugging DCF signal duration
.if dbgDcfDur == 1
call DebugDcfDurDispl ; Display DCF77 signal duration
.endif
ldi rmp,Low(cDcfMin) ; Check minimum
sub rmp,rDcfL
ldi rmp,High(cDcfMin)
sbc rmp,rDcfH
ldi rmp,1 ; Error number 1
brcc GotoDcfErr
DcfAct1:
ldi rmp,Low(cDcfMaxOne) ; Check One-Bit
sub rmp,rDcfL
ldi rmp,High(cDcfMaxOne)
sbc rmp,rDcfH
brcc Dcf01Ok ; Low- or High-Bit ok
ldi rmp,Low(cDcfMinInactive) ; Check shorter than inactive?
sub rmp,rDcfL
ldi rmp,High(cDcfMinInactive)
sbc rmp,rDcfH
ldi rmp,2 ; Error number 2
brcc GotoDcfErr
DcfAct2:
ldi rmp,Low(cDcfMaxInactive) ; Check shorter than inactive?
sub rmp,rDcfL
ldi rmp,High(cDcfMaxInactive)
sbc rmp,rDcfH
brcc DcfOk ; Duration ok
ldi rmp,Low(cDcfMin59) ; Check minimum 59th bit?
sub rmp,rDcfL
ldi rmp,High(cDcfMin59)
sbc rmp,rDcfH
ldi rmp,3 ; Error number 3
brcc GotoDcfErr
ldi rmp,Low(cDcfMax59) ; Check maximum 59th bit?
sub rmp,rDcfL
ldi rmp,High(cDcfMax59)
sbc rmp,rDcfH
brcc Dcf59Ok ; Missing bit ok
ldi rmp,4 ; Error # 4
rjmp GotoDcfErr
DcfOk: ; Signal duration ok
ret ; Time ok
GotoDcfErr:
mov rDcfErr,rmp ; Store error number

rjmp DcfReport ; Report error
;
; The DCF77 receiver detected a correct bit
;
Shift this into the SRAM buffer
Dcf01Ok:
ldi rmp,Low(cDcfMaxZero) ; Zero or one?
sub rmp,rDcfL
ldi rmp,High(cDcfMaxZero)
sbc rmp,rDcfH
; Carry set if one
ldi ZH,High(sDcfBits+8) ; Point Z to bit buffer
ldi ZL,Low(sDcfBits+8)
ld rmp,-Z ; Shift into byte 7
ror rmp
st Z,rmp
ld rmp,-Z ; Shift into byte 6
ror rmp
st Z,rmp
ld rmp,-Z ; Shift into byte 5
ror rmp
st Z,rmp
ld rmp,-Z ; Shift into byte 4
ror rmp
st Z,rmp
ld rmp,-Z ; Shift into byte 3
ror rmp
st Z,rmp
ld rmp,-Z ; Shift into byte 2
ror rmp
st Z,rmp
ld rmp,-Z ; Shift into byte 1
ror rmp
st Z,rmp
ld rmp,-Z ; Shift into byte 0
ror rmp
st Z,rmp
inc rDcfBitCnt ; Count received bits
; If debugging DCF bits is on
.if dbgDcfBits == 1
call DebugDcfBitEcho ; Echo DCF bits on LCD
.endif
ret
;
; The DCF77 receiver detected the missing 59th second
;
Check if all bits are correct and set date and time
Dcf59Ok:
; Check all signal durations
tst rDcfErr ; All signals durations ok?
breq Dcf59Ok1 ; Yes
; Signaldauer nicht ok
clr rDcfBitCnt ; Restart bit counter
clr rDcfErr ; Restart error counter
ret
Dcf59Ok1:
; Have 59 bits been received correct?
mov rmp,rDcfBitCnt
cpi rmp,59 ; Only 58 or less bits received?
breq Dcf59Ok2 ; Correct number of bits
; Error detected
ldi rmp,6 ; Display error message number of bits
mov rDcfErr,rmp ; Error # 6
rcall DcfReport ; Display errors
clr rDcfBitCnt ; Restart bit counter
clr rDcfErr ; Restart error register
ret

Dcf59Ok2:
; Start bit analysis
clr rDcfBitCnt ; Restart bit count
ldi rmp,7 ; Start with error #7
mov rDcfErr,rmp
; Check bit 1 = 0
lds rmp,sDcfBits ; Read least significant byte
sbrc rmp,5 ; DCF bit 1 is bit 5
rjmp DcfReport ; Error #7
; Check Minute ones
inc rDcfErr ; Next error #
lds ZL,sDcfBits+3 ; Read minute ones
lsr ZL ; Shift right
lsr ZL ; Another shift right
andi ZL,0x0F ; Isolate ones
cpi ZL,10 ; Larger than 9?
brcs Dcf59Ok4 ; No, fine
rjmp DcfReport ; Error report
Dcf59Ok4:
; Check Minute tens
inc rDcfErr ; Next error #
lds ZH,sDcfBits+4 ; Read Minute 40s
andi ZH,0x01 ; Isolate last Bit
lds rmp,sDcfBits+3 ; Read Minute 20s and 10s
lsl rmp ; Left shift 20s
rol ZH ; and roll into ZH
lsl rmp ; Left shift 10s
rol ZH ; and roll into ZH
andi ZH,0x07 ; Isolate 40/20/10 minutes
cpi ZH,6 ; Larger than 59?
brcs Dcf59Ok5 ; No
rjmp DcfReport ; Error reporting
Dcf59Ok5:
rcall BitCombine ; Combine tens and ones
sts sDcfRcv+5,ZL ; Save minutes in buffer
; Check parity bit of minutes
inc rDcfErr ; Next error #
lds ZL,sDcfBits+3 ; Read Minute ones
lds ZH,sDcfBits+4 ; Read Minute tens
lsr ZH ; DShift together
ror ZL
lsr ZH
ror ZL
rcall CheckParityZL ; Check minutes parity
breq Dcf59Ok6 ; Fine
rjmp DcfReport ; Error reporting
Dcf59Ok6:
; Get Hours ones
inc rDcfErr ; Next error #
lds ZL,sDcfBits+4 ; Read hour ones
lsr ZL
lsr ZL
andi ZL,0x0F ; Isolate hour ones
cpi ZL,10 ; Larger than 9
brcs Dcf59Ok7 ; No
rjmp DcfReport ; Error reporting
Dcf59Ok7:
; Get Hours tens
inc rDcfErr ; Next error #
lds ZH,sDcfBits+4 ; Read 10-hours
lsl ZH ; Shift left
rol ZH ; and roll in from the right
rol ZH
andi ZH,0x03 ; Isolate 20/10s

cpi ZH,0x03 ; Larger than 20
brcs Dcf59Ok8 ; Fine
rjmp DcfReport ; Error reporting
Dcf59Ok8:
rcall BitCombine ; Combine tens and ones
sts sDcfRcv+4,ZL ; Save hours in buffer
; Check parity hours
inc rDcfErr ; Next error #
lds ZL,sDcfBits+4 ; Read hours ones
lds ZH,sDcfBits+5 ; and tens again
lsr ZH ; Combine in ZL
ror ZL
lsr ZL
rcall CheckParityZL ; Check hours parity
breq Dcf59Ok9 ; Fine
rjmp DcfReport ; Error reporting
Dcf59Ok9:
; Get day ones
inc rDcfErr ; Next error #
lds ZL,sDcfBits+5 ; Read day ones
lsr ZL ; Shift right
andi ZL,0x0F ; Isolate day ones
cpi ZL,10 ; Larger than 9?
brcs Dcf59Ok10 ; No, fine
rjmp DcfReport ; Error reporting
Dcf59Ok10:
; Get day tens
inc rDcfErr ; Next error #
lds ZH,sDcfBits+5 ; Read day tens
swap ZH ; Upper to lower nibble
lsr ZH ; One bit right
andi ZH,0x03 ; Isolate bit 0 and 1
rcall BitCombine ; Combine ZH and ZL in ZL
sts sDcfRcv+1,ZL ; Save days in buffer
tst ZL ; day = 0?
brne Dcf59Ok11 ; No, fine
rjmp DcfReport ; Error reporting
Dcf59Ok11:
; Get Weekday
inc rDcfErr ; Next error #
lds ZL,sDcfBits+5 ; Read weekday, part 1
lds ZH,sDcfBits+6 ; Read weekday, part 2
lsl ZL ; Combine in ZH
rol ZH
andi ZH,0x07 ; Isolate weekday
brne Dcf59Ok12 ; Not zero, fine
rjmp DcfReport ; Error reporting
Dcf59Ok12:
dec ZH ; Decrease weekday (0 to 6)
sts sDcfRcv,ZH ; Save weekday in buffer
; Get month
inc rDcfErr ; Next error #
lds rmp,sDcfBits+6 ; Read month
andi rmp,0x7C ; Isolate month tens and ones
lsr rmp ; Shift left
lsr rmp
sbrc rmp,4 ; Month tens?
subi rmp,-10-16 ; Yes, add ten and clear bit 4
sts sDcfRcv+2,rmp ; Store month in buffer
tst rmp ; Month = 0?
brne Dcf59Ok13 ; No, fine
rjmp DcfReport ; Error reporting
Dcf59Ok13:
; Get year ones

inc rDcfErr ; Next error #
lds ZL,sDcfBits+6 ; Read year ones, part 1
lds ZH,sDcfBits+7 ; Read year ones, part 2
lsl ZL
rol ZH
mov ZL,ZH
andi ZL,0x0F ; Isolate year ones
cpi ZL,10 ; Larger than 9?
brcs Dcf59Ok14 ; No, fine
rjmp DcfReport ; Error reorting
Dcf59Ok14:
; Get year tens
inc rDcfErr ; Next error #
lds ZH,sDcfBits+7 ; Read year tens
lsr ZH
lsr ZH
lsr ZH
andi ZH,0x0F ; Isolate year tens
cpi ZH,10 ; Larger than 9?
brcs Dcf59Ok15 ; No, fine
rjmp DcfReport ; Error reporting
Dcf59Ok15:
rcall BitCombine ; Combine years ones and tens
sts sDcfRcv+3,ZL ; Store in buffer
; Check date parity
inc rDcfErr ; Next error #
lds ZL,sDcfBits+5 ; Read date byte 1
andi ZL,0xFE ; Clear last bit
rcall CheckParityZL ; Start checking parity
lds ZL,sDcfBits+6 ; Read date byte 2
rcall CheckParityZL1 ; Continue checking parity
lds ZL,sDcfBits+7 ; Read date byte 3
rcall CheckParityZL1 ; Continue checking parity
breq Dcf59Ok16 ; Date parity fine
rjmp DcfReport ; Error reporting
Dcf59Ok16:
lds rmp,sKeyMode ; Key input mode active?
cpi rmp,2 ; Mode 2 active?
breq Dcf59Ok18 ; Yes, skip setting date/time
clr rmp ; Clear seconds in buffer
sts sDcfRcv+6,rmp
ldi XH,High(sDateTime) ; Point X to date/time buffer
ldi XL,Low(sDateTime)
ldi ZH,High(sDcfRcv) ; Point Z to received buffer
ldi ZL,Low(sDcfRcv)
ldi rmp,7 ; Copy seven bytes
Dcf59Ok17:
ld R0,Z+ ; Read received
st X+,R0 ; Copy to date/time
dec rmp ; Count down
brne Dcf59Ok17 ; Continue copying
; Restart at seconds = 0
; Delay by signal and byte analysis: 486 clock cycles
; Equals 486/64 Timer 0 ticks
; Equals 486/64/256 Timer 0 overflows
; = 0.03 Delay by analysis
; = Below correction necessity for cSecDiv
ldi rSecDiv,cSecDiv ; Restart seconds down counter
ldi XH,High(sDateTime+3) ; Point X to Year
ldi XL,Low(sDateTime+3)
rcall DateTimeLcd ; Renew date and time on display
Dcf59Ok18:
sbr rFlag,1<<bMinute ; Set minute change flag
clr rDcfErr ; Error # = 0

rjmp DcfReport
;
; Combines ones bits in ZL with tens bits in ZH
;
Result in ZL
BitCombine:
mov rmp,ZH ; Copy tens
add ZH,ZH ; Multiply by 2
add ZH,ZH ; Multiply by 4
add ZH,rmp ; Add one to 5
add ZH,ZH ; Multiply by 10
add ZL,ZH ; Add tens to ones
ret
;
; Check even parity of ZL, Z set if even
;
Counts the number of ones
;
Returns Z set if number is even
CheckParityZL: ; Start parity check
clr rmp
CheckParityZL1: ; Continue parity check
tst ZL ; No ones any more?
breq CheckParityZL2 ; Yes
lsr ZL ; Shift to carry
brcc CheckParityZL1 ; Bit is zero
inc rmp ; Count bit
rjmp CheckParityZL1 ; Continue
CheckParityZL2:
andi rmp,0x01 ; Last bit of counter
ret
;
; Clears the receiver buffer in SRAM
DcfClearBuffer:
ldi XH,High(sDcfRcv) ; To start of buffer
ldi XL,Low(sDcfRcv)
ldi ZL,0xFF ; Fill with FF
ldi rmp,6 ; Six bytes
DcfClearBuffer1:
st X+,ZL
dec rmp
brne DcfClearBuffer1
ret
;
; DCF signal has timed out
;
Detected in TC0 interrupt service routine
;
Display error #5
DcfTimeOut:
cbr rFlag,1<<bDcfTO ; Clear that flag
ldi rmp,5 ; Error # 5
mov rDcfErr,rmp
rjmp DcfReport ; Display
;
; Display DCF report
;
Error # in rDcfErr
DcfReport:
ldi ZH,2 ; To short report position on LCD
ldi ZL,cDcfPos ; Position of the report
call LcdPos ; Position the LCD
mov rmp,rDcfErr ; rDcfErr to text table
cpi rmp,6 ; Number of received bits false?
brne DcfReport1 ; No, other error #
mov rmp,rDcfBitCnt ; Copy number of bits received
call LcdDec2 ; Write in decimal to LCD
rjmp DcfReport3 ; Check additional reorting
; Error number not 6
DcfReport1:

cpi rmp,cDcfErrMax ; Check inside error numbers
brcs DcfReport2 ; Fine
ldi rmp,cDcfErrMax-1 ; Set to last error number
DcfReport2:
; Multiply error number by 22
clr R0 ; R0 is Carry adder
lsl rmp ; Multiply by 2
lsl rmp ; Multiply by 4
mov XL,rmp ; Copy to X = 4-fold
clr XH
lsl XL ; Multiply by 8
rol XH
lsl XL ; Multiply by 16
rol XH
add XL,rmp ; Add 4 to yield 20-fold
adc XH,R0 ; Add zero and carry
lsr rmp ; Division by 2 to yield 2-fold
add XL,rmp ; Add to get 22-fold
adc XH,R0 ; Add zero and carry
; Add to table start
ldi ZH,High(2*DcfReports) ; Point Z to text table
ldi ZL,Low(2*DcfReports)
add ZL,XL ; Add table displacement
adc ZH,XH
; Display short report
lpm rmp,Z+ ; Read first character
call LcdChar ; and display
lpm rmp,Z+ ; Read second character
rcall LcdChar ; and display
DcfReport3:
; If the long version shall be displayed
.if dbgDcfSig == 1
call DebugDcfSig2 ; Display long version
.endif
; If received signals shall be dislayed
.if dbgDcfRcvd == 1
call DebugDcfRcvd ; Display received signals
rcall DcfClearBuffer ; Set the DCF buffer to FF
.endif
rjmp ActivatePos ; If in input mode set cursor
;
; All DCF reports
;
char 1+2: short abbreviation
;
char 3..18: Long version for debug
DcfReports:
; If english version selected
.if LangEN == 1
.db "okAll DCF bits ok. ",0xFE,0xFE ; 0 DCF no errors
.db "s1Signal too short! ",0xFE,0xFE ; 1
.db "s2Signal in between!",0xFE,0xFE ; 2
.db "s3Signal too long! ",0xFE,0xFE ; 3
.db "s4Signal minute!
",0xFE,0xFE ; 4
.db "s5No signal!
",0xFE,0xFE ; 5
.db "nnnn"
LcdErr6:
.db " bits!
",0xFE,0xFE ; 6
.db "E0Bit 1 not zero!
",0xFE,0xFE ; 7
.db "MOMinute ones!
",0xFE,0xFE ; 8
.db "MTMinute tens!
",0xFE,0xFE ; 9
.db "PMMinute parity!
",0xFE,0xFE ; 10
.db "HOHours ones!
",0xFE,0xFE ; 11
.db "HTHours tens!
",0xFE,0xFE ; 12
.db "PHHours parity!
",0xFE,0xFE ; 13
.db "D1Day ones!
",0xFE,0xFE ; 14

.db "D0Day zero!
",0xFE,0xFE ; 15
.db "W0Weekday zero!
",0xFE,0xFE ; 16
.db "M0Month zero!
",0xFE,0xFE ; 17
.db "YOYear ones!
",0xFE,0xFE ; 18
.db "YTYear tens!
",0xFE,0xFE ; 19
.db "PDParity date!
",0xFE,0xFE ; 20
.db "ueUnknown error!
",0xFE,0xFE ; 21
.else
; If german version selected
.db "okAlle DCF Bits ok. ",0xFE,0xFE ; 0 DCF ohne Fehler
.db "s1Signal zu kurz!
",0xFE,0xFE ; 1
.db "s2Signal dazwischen!",0xFE,0xFE ; 2
.db "s3Signal zu lang!
",0xFE,0xFE ; 3
.db "s4Signal Minute!
",0xFE,0xFE ; 4
.db "s5Kein Signal!
",0xFE,0xFE ; 5
.db "nnnn"
LcdErr6:
.db " Bits!
",0xFE,0xFE ; 6
.db "E0Bit 1 nicht Null! ",0xFE,0xFE ; 7
.db "MEMinuten Einer!
",0xFE,0xFE ; 8
.db "MZMinuten Zehner!
",0xFE,0xFE ; 9
.db "PMMinuten Paritaet! ",0xFE,0xFE ; 10
.db "SEStunden Einer!
",0xFE,0xFE ; 11
.db "SZStunden Zehner!
",0xFE,0xFE ; 12
.db "PSStunden Paritaet! ",0xFE,0xFE ; 13
.db "TETag Einer!
",0xFE,0xFE ; 14
.db "T0Tag ist Null!
",0xFE,0xFE ; 15
.db "W0Wochentag Null!
",0xFE,0xFE ; 16
.db "M0Monat gleich Null!",0xFE,0xFE ; 17
.db "JEJahr Einer!
",0xFE,0xFE ; 18
.db "JZJahr Zehner!
",0xFE,0xFE ; 19
.db "PDParitaet Datum!
",0xFE,0xFE ; 20
.db "ufUnbek. Fehler!
",0xFE,0xFE ; 21
.endif
.equ cDcfErrMax = 22 ; Last error
;
; =====================================
; S E C O N D S P R O C E S S I N G
; =====================================
;
IncSec:
cbr rFlag,1<<bSec ; Clear flag
; If music play debugging is selected
.if dbgMusic == 1
call DebugMusic ; Display status
.endif
; If TC1 play stat debugging is selected
.if dbgPlayStat == 1
call DebugPlayStat
.endif
lds rmp,sKeyMode ; Key mode input on date/time?
cpi rmp,2
brne IncSec1 ; Input mode not active
ret ; Input active, do not change time
IncSec1:
ldi ZH,High(sDateTime) ; Point Z to sDateTime
ldi ZL,Low(sDateTime)
ldd rmp,Z+6
inc rmp ; Increase seconds
std Z+6,rmp ; and store
cpi rmp,60 ; 60 seconds reached?
brcc IncMin
rjmp SecLcd ; Display seconds
; Increase minutes

IncMin:
clr rmp ; Seconds to zero
std Z+6,rmp
ldd rmp,Z+5 ; Read minutes
inc rmp ; Increase minutes
sbr rFlag,1<<bMinute ; Set minute flag
std Z+5,rmp ; and save
cpi rmp,60 ; 60 minutes over?
brcs MinLcd ; No, display minutes and seconds
; Increase hours
clr rmp ; Restart minutes
std Z+5,rmp ; and save in date/time buffer
ldd rmp,Z+4 ; Read hours
inc rmp ; Increase hours
std Z+4,rmp ; Seichern Stunden
cpi rmp,24 ; 24 Stunden erreicht?
brcs HourLcd ; Hours to LCD
clr rmp ; Clear hours
std Z+4,rmp ; in sDateTime
; Increase weekday
ld rmp,Z ; Read weekday
inc rmp ; Increase
cpi rmp,7 ; Weekday = 7?
brcs IncDay ; No
ldi rmp,0 ; Restart weekdays
; Increase day
IncDay: ; Next day
st Z,rmp ; Store weekday
ldd rmp,Z+1 ; Read day
inc rmp ; Increase day
std Z+1,rmp ; Save day
rcall GetDaysOfMonth
cp rmp,XH ; Compare with days of month
brcs WeekDayLcd ; Display weekday
; Increase Month
ldi rmp,1 ; Day = 1
std Z+1,rmp ; Day = 1
ldd rmp,Z+2 ; Read month
inc rmp ; Increase month
std Z+2,rmp ; Store month
cpi rmp,13 ; Month = 13?
brcs MonthLcd ; No, display on LCD
; Increase year
ldi rmp,1 ; Month = 1
std Z+2,rmp ; Store month
ldd rmp,Z+3 ; Read year
inc rmp ; Increase year
std Z+3,rmp ; Store year
cpi rmp,100 ; Year = 100?
brcs YearLcd ; Dislay
ldi rmp,0 ; Year = 0
std Z+3,rmp ; Store year
; Date and time to LCD
; Format: Tu-01.05.18-23:59:59
; Position 01234567890123456789
; Z points to position in SRAM
DateTimeLcd:
ldi ZH,High(sDateTime) ; Point to sDateTime
ldi ZL,Low(sDateTime)
YearLcd:
ldd rmp,Z+3 ; Z points to buffer start
ldi XL,9 ; Column year on LCD
rcall NumberLcd ; Display decimal
MonthLcd:

ldd rmp,Z+2 ; Read month
ldi XL,6 ; Column month on LCD
rcall NumberLcd ; Display decimal
WeekDayLcd:
rcall DisplWeekday ; Display weekday
DayLcd:
ldd rmp,Z+1 ; Read day
ldi XL,3 ; Column day on LCD
rcall NumberLcd ; Display decimal
HourLcd:
ldd rmp,Z+4 ; Read hour
ldi XL,12 ; Column hour on LCD
rcall NumberLcd ; Display decimal
MinLcd:
ldd rmp,Z+5 ; Read minute
ldi XL,15 ; Column minute on LCD
rcall NumberLcd ; Display decimal
SecLcd:
ldd rmp,Z+6 ; Read seconds
ldi XL,18 ; Column seconds on LCD
rcall NumberLcd ; Display decimal
rjmp ActivatePos ; Activate input cursor
;
; Display the number in rmp in
;
decimal format on the LCD
;
on line 1
NumberLcd:
push rmp ; Save number
lds rmp,sKeyMode ; Read key mode
cpi rmp,2 ; In normal or alarm mode?
brcs NumberLcd0 ; Yes
pop rmp ; Do not update LCD
ret
NumberLcd0: ; Write to LCD
push ZH ; Save Z
push ZL
ldi ZH,0 ; Line 1
mov ZL,XL ; Column from XL
rcall LcdPos
pop ZL ; Restore Z
pop ZH
ldi rmp,0x2F ; ASCII tens counter
mov R0,rmp ; in R0
pop rmp ; Restore number
NumberLcd1:
inc R0 ; Increase ASCII
subi rmp,10 ; Subtract 10
brcc NumberLcd1 ; No carry, repeat
push rmp ; Save rest of number
mov rmp,R0 ; R0 to rmp
call LcdChar ; Write ASCII in rmp to LCD
pop rmp ; Restore rest of number
subi rmp,-58 ; Add 10 and ASCII-0
rjmp LcdChar ; Display second digit
;
; Days of current month
;
Returns days of current month + 1
; 1:31, 2:28/29, 3:31, 4:30, 5:31, 6:30
; 7:30, 8:31, 9:30, 10:31, 11:30, 12:31
GetDaysOfMonth:
ldd XL,Z+2 ; Load Month
cpi XL,2 ; February?
brne GetDaysOfMonth2 ; No
ldd XL,Z+3 ; February, load year

andi XL,0x03 ; Leap year?
brne GetDaysOfMonth1 ; No
ldi XH,30 ; Leap year, 29 days
ret
GetDaysOfMonth1:
ldi XH,29 ; No leap year, 28 days
ret
GetDaysOfMonth2:
cpi XL,7 ; Month less than July?
brcs GetDaysOfMonth3 ; Yes
dec XL ; Reverse row
GetDaysOfMonth3:
lsr XL ; Month uneven?
brcc GetDaysOfMonth4 ; No
ldi XH,32 ; 31 days
ret
GetDaysOfMonth4:
ldi XH,31 ; 30 days
ret
;
; Display the weekday 0 to 6 on LCD
;
Text is in WeekDayTable
DisplWeekday:
lds rmp,sKeyMode ; Read key mode
cpi rmp,2 ; In date/time input mode?
brne DisplWeekday0
ret
DisplWeekday0:
push ZH ; Save Z
push ZL
ld rmp,Z ; Read weekday
push rmp
ldi ZH,0 ; Line 1
ldi ZL,0 ; Column 1
rcall LcdPos
ldi ZH,High(2*WeekdayTable) ; Point to table
ldi ZL,Low(2*WeekdayTable)
pop rmp ; Restore weekday
lsl rmp ; Multiply by 2
add ZL,rmp ; Add to weekday table
ldi rmp,0 ; Carry adder
adc ZH,rmp ; Add carry
lpm rmp,Z+ ; Read first character
rcall LcdChar ; and display
lpm rmp,Z ; Read second character
pop ZL ; Restore Z
pop ZH
rjmp LcdChar ; Display second character
;
; Table with the weekday abbreviations
WeekdayTable:
.if LangEN == 1
; If English use that
.db "MoTuWdThFrSaSu"
.else
; If German use that
.db "MoDiMiDoFrSaSo"
.endif
;
; Init date and time
InitDT:
ldi XH,High(sDateTime)
ldi XL,Low(sDateTime)
ldi ZH,High(2*InitDTTable)

ldi ZL,Low(2*InitDTTable)
InitDT1:
lpm rmp,Z+
st X+,rmp
cpi ZH,High(2*InitDTTableEnd-1)
brne InitDT1
cpi ZL,Low(2*InitDTTableEnd-1)
brne InitDT1
rcall DateTimeLcd
ldi rmp,8 ; Init alarm time
sts sAlrmHr,rmp
ldi rmp,20
sts sAlrmMin,rmp
rjmp DisplAlarm
;
; Start table for date/time
; Date/Time = Mo-30.04.18-23:59:59
; Alarm time =
00:03
InitDTtable:
.db 0,30,4,18,23,59,59,0,3,0
InitDTtableEnd:
;
; =======================================
; C H E C K
A L A R M
R E A C H E D
; =======================================
;
; Check each minute if alarm condition reached
CheckMinute:
cbr rFlag,1<<bMinute ; Clear flag
lds rmp,sKeyMode ; Check key mode
cpi rmp,1 ; Alarm active?
brne CheckMinuteRet ; Alarm is not active
lds rmp,sAlarmRepeat ; Alarm to be repeated?
tst rmp ; Alarm repetition still active?
brne CheckMinute1 ; Yes
lds rmp,sDateTime+8 ; Read minutes alarm
lds R0,sDateTime+5 ; Read minutes time
cp rmp,R0 ; Compare
brne CheckMinuteRet ; Not equal
lds rmp,sDateTime+7 ; Read hours alarm
lds R0,sDateTime+4 ; Read hours time
brne CheckMinuteRet ; Not equal
; Start alarm
ldi rmp,cAlarmRepet ; Load repetition
sts sAlarmRepeat,rmp
rcall IncAlarm ; Alarm to next minute
mov rmp,rMelody ; Read random melody
rjmp MusicConvert ; Play this melody
CheckMinute1: ; Alarm active
; Alarm still active, restart melody
rcall MusicPlay ; Restart current melody
lds rmp,sAlarmRepeat ; Read repetitions left
dec rmp ; Decrease repeats
sts sAlarmRepeat,rmp ; and store
breq CheckMinute2 ; Zero, update display
rjmp IncAlarm ; Increase alarm minute
CheckMinute2:
; Switch alarm off
rjmp SetAlarmOff
CheckMinuteRet:
ret
;
; Increase the alarm minute
IncAlarm:

ldi ZH,High(sDateTime) ; Point to sDateTime
ldi ZL,Low(sDateTime)
ldd rmp,Z+8 ; Read alarm minute
inc rmp ; Increase
std Z+8,rmp ; Store
cpi rmp,60 ; Next hour?
brcs IncAlarmRet ; No
clr rmp ; Clear minute
std Z+8,rmp
ldd rmp,Z+7 ; Read hour
inc rmp ; Increase hour
std Z+7,rmp ; Store
cpi rmp,24 ; Next day?
brcs IncAlarmRet ; No
clr rmp ; Clear hour
std Z+7,rmp
IncAlarmRet:
rjmp DisplAlarm
;
; ===================================
; P R O C E S S K E Y E V E N T S
; ===================================
;
; A key has been pressed
;
KeyPress:
cbr rFlag,1<<bKey ; Clear flag
lds rmp,sKeyMode ; Read input mode
tst rmp ; Zero?
brne KeyPress1 ; No
sbis pKeyI,bKeyRI ; Red key
rjmp KeyPressRed ; Perform red key
sbis pKeyI,bKeyYI ; Yellow key
rjmp KeyPressYellow ; Perform yellow key
sbis pKeyI,bKeyWI ; White key
rjmp KeyPressWhite ; Perform white key
ret
KeyPress1: ; Mode not zero
cpi rmp,2 ; Mode = 2?
brcs KeyPressMode1 ; No, 0 or 1
breq KeyPressMode2 ; Yes
rjmp KeyPressMode3 ; No, mode = 3
ret
;
KeyPressMode1:
sbis pKeyI,bKeyRI ; Red key
rjmp KeyPress1Red ; Yes, perform red in mode 1
sbis pKeyI,bKeyYI ; Yellow key?
rjmp KeyPress1Yellow ; Yes, perform yellow in mode 1
sbis pKeyI,bKeyWI ; White key?
rjmp KeyPress1White ; Yes, perform white in mode 1
ret
;
KeyPressMode2:
sbis pKeyI,bKeyRI ; Red key
rjmp KeyPress2Red ; Perform red in mode 2
sbis pKeyI,bKeyYI ; Yellow key?
rjmp KeyPress2Yellow ; Perform yellow in mode 2
sbis pKeyI,bKeyWI ; White key?
rjmp KeyPress2White ; Perform white in mode 2
ret
;
KeyPressMode3:
sbis pKeyI,bKeyRI ; Red key

rjmp
sbis
rjmp
sbis
rjmp
ret

KeyPress3Red ; Perform red in mode 3
pKeyI,bKeyYI ; Yellow key
KeyPress3Yellow ; Perform yellow in mode 3
pKeyI,bKeyWI ; White key
KeyPress3White ; Perform white in mode 3

;
; Keys in mode 0
;
; Red key pressed
KeyPressRed:
lds rmp,sKeyMode ; Read current mode
cpi rmp,1 ; Mode = 1?
brcs SetAlarmOn ; No, mode = 0
breq SetAlarmOff ; Yes, mode = 1
rjmp DisplAlarm ; Mode 2 or 3
SetAlarmOff:
ldi rmp,(1<<COM1A1) ; Speaker output clear
out TCCR1A,rmp
ldi rmp,(1<<TOIE0)|(1<<TOIE2) ; Interrupt off
out TIMSK,rmp
ldi rmp,High(clock/2000-1) ; CTC TC1 to 1000 Hz
out OCR1AH,rmp
ldi rmp,Low(clock/2000-1)
out OCR1AL,rmp
ldi rmp,0 ; Switch mode to normal
sts sKeyMode,rmp ; Switch alarm off
sts sAlarmRepeat,rmp ; Alarm repetition clear
lds rmp,sAlarm ; Restart alarm time
sts sDateTime+7,rmp
lds rmp,sAlarm+1
sts sDateTime+8,rmp
rjmp DisplAlarm
SetAlarmOn:
ldi rmp,1 ; Set mode = 1
sts sKeyMode,rmp
ldi rmp,0 ; Clear repeat
sts sAlarmRepeat,rmp
rjmp DisplAlarm
;
; The yellow key has been pressed
;
Start date/time input
KeyPressYellow:
ldi rmp,2 ; Mode 2
sts sKeyMode,rmp
clr rmp ; Input position to weekday
sts sKeyPos,rmp
ldi ZH,High(sDateTime) ; Point Z to weekday
ldi ZL,Low(sDateTime)
ldi XH,High(sInputData) ; Point X to input
ldi XL,Low(sInputData)
ldi rmp,7 ; Copy seven bytes
KeyPressYellow1:
ld R0,Z+ ; Read date/time
st X+,R0 ; Write to input buffer
dec rmp ; Count down
brne KeyPressYellow1 ; Continue read/write
rcall LcdBlink ; Blink cursor on
rjmp ActivatePos
;
; The white key is pressed
;
Start with alarm time input
KeyPressWhite:
ldi rmp,3 ; Mode = 3

sts sKeyMode,rmp
ldi rmp,7 ; Position in input buffer
sts sKeyPos,rmp
lds rmp,sDateTime+7 ; Alarm time hour
sts sInputData+7,rmp
lds rmp,sDateTime+8
sts sInputData+8,rmp
rcall DisplAlarm ; Display alarm time
rcall DisplMenue ; Display menu
rcall ActivatePos ; Activate cursor
ldi ZH,1 ; Cursor to alarm hour
ldi ZL,12
rcall LcdPos
rjmp LcdBlink ; Blinking on
;
; Keys in mode 1
;
; Red key mode 1
KeyPress1Red:
; Set alarm off
rjmp SetAlarmOff
;
; Display the current alarm state and time
DisplAlarm:
ldi ZH,1 ; Alarm state position
ldi ZL,cAlarmPos
rcall LcdPos
ldi ZH,High(2*AlarmText0) ; Text for alarm off
ldi ZL,Low(2*AlarmText0)
lds rmp,sKeyMode ; Is the alarm active?
cpi rmp,1
brcs DisplAlarm1 ; No, display this text
ldi ZH,High(2*AlarmText1) ; Text for alarm on
ldi ZL,Low(2*AlarmText1)
breq DisplAlarm1 ; Yes display this text
ldi ZH,High(2*AlarmText2) ; Text for alarm suspended
ldi ZL,Low(2*AlarmText2)
DisplAlarm1:
call LcdText ; Display text from Z
ldi rmp,' ' ; Display blank
rcall LcdChar
lds rmp,sDateTime+7 ; Alarm time hour
rcall LcdDec2 ; Display decimal
ldi rmp,':' ; Display colon
rcall LcdChar
lds rmp,sDateTime+8 ; Alarm time minute
rcall LcdDec2 ; Display decimal
ldi rmp,' ' ; Display blank
rcall LcdChar
lds rmp,sAlarmRepeat ; Repetitions
tst rmp ; Zero?
breq DisplAlarm3 ; Yes display two blanks
dec rmp ; Decrease by one
cpi rmp,100 ; Larger than 99?
brcs DisplAlarm2
ldi rmp,99 ; Load maximum
DisplAlarm2:
rjmp LcdDec2 ; Display decimal
DisplAlarm3:
ldi rmp,' ' ; Display two blanks
rcall LcdChar
ldi rmp,' '
rcall LcdChar
rjmp DisplMenue

;
; Alarm active/inactive text
.if LangEN == 1
; English version
AlarmText0:
.db "off",0xFE
AlarmText1:
.db " on",0xFE
AlarmText2:
.db " ..",0xFE
.else
; German version
AlarmText0:
.db "aus",0xFE
AlarmText1:
.db " an",0xFE
AlarmText2:
.db " ..",0xFE
.endif
;
; Yellow key mode 1
KeyPress1Yellow:
; Increase alarm time by 5 minutes
lds rmp,sDateTime+8 ; Read minutes
subi rmp,-5 ; Add five minutes
SetAlarmTime:
sts sDateTime+8,rmp ; Store
cpi rmp,60 ; smaller than 60?
brcs SetAlarmTimeFinal
subi rmp,60 ; Subtract 60
sts sDateTime+8,rmp
lds rmp,sDateTime+7 ; Increase hour
inc rmp
cpi rmp,24 ; End of day?
brcs SetAlarmTimeFinal
subi rmp,24 ; Subtract 24
sts sDateTime+8,rmp
SetAlarmTimeFinal:
rjmp DisplAlarm ; Display alarm time
;
; White key mode 1
KeyPress1White:
; Increase alarm time by 10 minutes
lds rmp,sDateTime+8 ; Read minutes
subi rmp,-10 ; Add ten minutes
rjmp SetAlarmTime ; Set alarm time
;
; Keys in mode 2
;
; Red key mode 2
KeyPress2Red:
; Move left
lds rmp,sKeyPos ; Read input position
tst rmp ; At zero?
breq KeyPress2Red1 ; Yes, skip input
dec rmp ; One position left
sts sKeyPos,rmp ; and store
rjmp ActivatePos ; and activate
KeyPress2Red1:
; Skip input, normal mode
lds rmp,sAlarm ; Restore original alarm time
sts sDateTime+7,rmp
lds rmp,sAlarm+1
sts sDateTime+8,rmp

rcall SetAlarmOff ; Alarm off
rcall DisplMenue ; Display menu
rjmp LcdNoBlink ; Blinking off

;
; Yellow key mode 2
KeyPress2Yellow:
; Restore original value
lds rmp,sKeyPos ; Read input position
ldi XH,High(sDateTime) ; Point X to Date/Time
ldi XL,Low(sDateTime)
ldi ZH,High(sInputData) ; Z to input buffer
ldi ZL,Low(sInputData)
clr R0 ; Zero-Adder for carry adder
add XL,rmp ; Add to date/time pointer
adc XH,R0 ; Add carry
add ZL,rmp ; Add to input pointer
adc ZH,R0 ; Add carry
ld rmp,X ; Read from date/time buffer
st Z,rmp ; Store to input buffer
rcall DisplInp ; Display current input data
rjmp KeyPress2White ; And move to next position
;
; Display input data at current position
DisplInp:
rcall ActivatePos ; Active position
lds rmp,sKeyPos ; Read input position
tst rmp ; Weekday?
brne DisplInp1
; Display weekday
ldi ZH,High(2*WeekdayTable)
ldi ZL,Low(2*WeekdayTable)
lds rmp,sInputData ; Read data
lsl rmp ; Multiply by 2
add ZL,rmp ; Add displacement
ldi rmp,0
adc ZH,rmp ; and carry
lpm rmp,Z+ ; Read first character from table
rcall LcdChar
lpm rmp,Z ; Read second character from table
rjmp LcdChar
DisplInp1:
ldi ZH,High(sInputData) ; Point to input buffer
ldi ZL,Low(sInputData)
add ZL,rmp
ldi rmp,0
adc ZH,rmp
ld rmp,Z ; Read data
rjmp LcdDec2
;
; White key mode 2
KeyPress2White:
; Next position
lds rmp,sKeyPos ; Read position
cpi rmp,6 ; Last position?
brcc KeyPress2White1 ; Yes
inc rmp ; Next position
sts sKeyPos,rmp ; Save new position
rjmp ActivatePos ; Activate position
KeyPress2White1:
ldi XH,High(sInputData) ; Point to input
ldi XL,Low(sInputData)
ldi ZH,High(sDateTime) ; Point to date/time
ldi ZL,Low(sDateTime)
ldi rmp,7 ; Seven bytes to transfer

KeyPress2White2:
ld R0,X+ ; Read byte from input buffer
st Z+,R0 ; Write byte to date/time buffer
dec rmp ; Count down
brne KeyPress2White2 ; Continue copying
ldi rmp,0 ; Alarm off
sts sKeyMode,rmp
rcall LcdNoBlink
rcall DateTimeLcd ; Display date/time
rjmp DisplMenue ; Display menu
;
; Keys in mode 3
;
; Red key mode 3
KeyPress3Red:
; One left
lds rmp,sKeyPos ; Read input position
cpi rmp,7 ; Leftmost position?
breq KeyPress3Red1 ; Yes, skip input
dec rmp ; One position left
sts sKeyPos,rmp ; Store
rjmp ActivatePos ; and activate
KeyPress3Red1:
lds rmp,sAlarm ; Skip, restore original
sts sDateTime+7,rmp
lds rmp,sAlarm+1
sts sDateTime+8,rmp
rcall SetAlarmOff ; Alarm off
rcall DisplMenue ; Display menu
rjmp LcdNoBlink ; Do not blink any more
;
; Yellow key mode 3
KeyPress3Yellow:
; Restore previous
lds rmp,sKeyPos ; Read position
cpi rmp,8 ; Minute position?
brcc KeyPress3Yellow1 ; No, hour
lds rmp,sDateTime+8 ; Get old minute
sts sInputData+8,rmp
sts sAlarm+1,rmp
rcall DisplAlarmInp
rjmp KeyPress3White
KeyPress3Yellow1:
lds rmp,sDateTime+7 ; Restore old alarm hour
sts sInputData+7,rmp
sts sAlarm,rmp
lds rmp,sKeyPos
inc rmp ; Next position
sts sKeyPos,rmp
rjmp DisplAlarmInp
;
; Display alarm input
DisplAlarmInp:
ldi ZH,1 ; Alarm position hour
ldi ZL,12
rcall LcdPos
lds rmp,sInputData+8
rcall LcdDec2
ldi ZH,1 ; Alarm position minute
ldi ZL,15
rcall LcdPos
lds rmp,sInputData+9
rcall LcdDec2
rjmp ActivatePos

;
; White key mode 3
KeyPress3White:
; Next position
lds rmp,sKeyPos ; Read position
cpi rmp,8 ; Last position?
breq KeyPress3White1 ; Copy input
inc rmp ; No, increase
sts sKeyPos,rmp ; and write
rjmp ActivatePos ; and activate
KeyPress3White1:
lds rmp,sInputData+7 ; Read alarm time hours
sts sAlarm,rmp ; Store new alarm hour
sts sDateTime+7,rmp ; and in date/time buffer
lds rmp,sInputData+8 ; Read alarm time minutes
sts sAlarm+1,rmp ; Store new alarm minute
sts sDateTime+8,rmp ; and in date/time
rcall SetAlarmOn ; Activate alarm
rcall DisplMenue ; Display menu
rjmp LcdNoBlink ; Blinking off
;
; Display menu for selected mode
;
DisplMenue:
.if ! fDbg ; Display in non-debug mode only
ldi ZH,3 ; Menue in line 3
ldi ZL,0
rcall LcdPos
ldi ZH,High(2*Menuetext)
ldi ZL,Low(2*Menuetext)
clr R1
lds R0,sKeyMode
ldi rmp,22
DisplMenue1:
tst R0
breq DisplMenue2
add ZL,rmp
adc ZH,R1
dec R0
rjmp DisplMenue1
DisplMenue2:
lds rmp,sKeyMode ; Read mode
cpi rmp,2
brcs DisplMenue3
breq Displ2Menue
lds rmp,sKeyPos ; Mode = 3
cpi rmp,7
ldi ZH,High(2*MenueFirst)
ldi ZL,Low(2*MenueFirst)
breq DisplMenue3
ldi ZH,High(2*MenueLast)
ldi ZL,Low(2*MenueLast)
rjmp DisplMenue3
Displ2Menue:
lds rmp,sKeyPos ; Mode = 2
tst rmp
brne Displ2Menue1
ldi ZH,High(2*MenueFirst)
ldi ZL,Low(2*MenueFirst)
rjmp DisplMenue3
Displ2Menue1:
cpi rmp,6
brne DisplMenue3
ldi ZH,High(2*MenueLast)

ldi ZL,Low(2*MenueLast)
DisplMenue3:
rjmp LcdText
.else
ret
.endif
;
; Activate the current position
; of the LCD for blinking
ActivatePos:
lds rmp,sKeyPos
cpi rmp,7
brcs ActivatePos1
ldi ZH,1
ldi ZL,12
breq ActivatePosSet
ldi ZL,15
rjmp ActivatePosSet
ActivatePos1:
ldi ZH,0
cpi rmp,1
ldi ZL,0
brcs ActivatePosSet
ldi ZL,3
breq ActivatePosSet
ldi ZL,6
cpi rmp,3
brcs ActivatePosSet
ldi ZL,9
breq ActivatePosSet
ldi ZL,12
cpi rmp,5
brcs ActivatePosSet
ldi ZL,15
breq ActivatePosSet
ldi ZL,18
ActivatePosSet:
rjmp LcdPos
;
Menuetext:
.if LangEN == 1
.db "Alarm Time Alarmtime",0xFE,0xFE
.db "NoAlarm +5min +10min",0xFE,0xFE
.db "Prev
Default Next",0xFE,0xFE
.db "Prev
Default Next",0xFE,0xFE
MenueFirst:
.db "Skip
Default Next",0xFE,0xFE
MenueLast:
.db "Prev
Default
Set",0xFE,0xFE
.else
.db "Wecken Zeit Weckzeit",0xFE,0xFE
.db "Weckaus +5min +10min",0xFE,0xFE
.db "Links
Alt
Rechts",0xFE,0xFE
.db "Links
Alt
Rechts",0xFE,0xFE
MenueFirst:
.db "Abbr.Default Naechst",0xFE,0xFE
MenueLast:
.db "Links
Alt
Setzen",0xFE,0xFE
.endif
;
; ======================================
; P R O C E S S
A D C R E S U L T S
; ======================================
;

; ADC has summed up 64 values in the current channel
;
MSB of the summed results is in rAdc
;
AdcCmplt:
cbr rFlag,1<<bAdc ; Clear flag
in rmp,ADMUX ; Read channel
andi rmp,0x1F ; Isolate channel bits
push rmp ; Save channel
inc rmp ; Next channel
cpi rmp,0x03 ; Channel 3?
brcs AdcCmplt1 ; No, smaller
clr rmp
AdcCmplt1:
ori rmp,(1<<REFS0) ; Set REFS0 bit
out ADMUX,rmp
ldi rAdcC,64 ; 64 measurements
ldi rmp,(1<<ADEN)|(1<<ADSC)|(1<<ADIE)|(1<<ADPS2)|(1<<ADPS1)|(ADPS0)
out ADCSRA,rmp ; Start next conversion
pop rmp ; Old channel
cpi rmp,1 ; Channel = 1?
brcs AdcSet0 ; No, channel = 0
brne AdcSet2 ; No, channel = 2
rjmp AdcSet1 ; Yes, channel = 1
;
; Channel 2 complete
;
Display signal strength
AdcSet2:
; Downcount signal strength counter
dec rSigStr
breq AdcSet2A ; Reached zero count
AdcSet2Ret:
ret ; Done
AdcSet2A:
ldi rmp,cSigStr ; Restart counter
mov rSigStr,rmp
lds rmp,sKeyMode ; Key mode input?
cpi rmp,2 ; Key mode input?
brcc AdcSet2Ret ; Yes, no display
; Channel = 2
ldi ZH,2 ; Line 3 for display
ldi ZL,7
rcall LcdPos
mov rmp,rAdc ; Read MSB ADC sum
cpi rmp,30 ; Larger or equal 30?
ldi rmp,' ' ; Write blank
brcs SignalNoPlus ; larger than 30
ldi rmp,'+' ; Display plus
SignalNoPlus:
rcall LcdChar
ldi ZH,2 ; Position S value
ldi ZL,6
rcall LcdPos
ldi ZH,High(2*SignalTable) ; Get signal strength from table
ldi ZL,Low(2*SignalTable)
clr R0 ; R0 is counter
SignalStrength:
inc R0 ; Inc counter
lpm rmp,Z+ ; Read table value
cp rAdc,rmp ; Compare with table value
brcc SignalStrength ; rAdc larger or equal
ldi rmp,'0'-1 ; Add ASCII-0 minus 1
add rmp,R0
rjmp LcdChar ; Display character
;

; Table to display signal strength
SignalTable:
.db 3,6,9,12,15,18,21,24,27,255
;
; Channel 0: Potentiometer result
;
If in input mode: update byte at current position
AdcSet0:
; Check input mode active
lds rmp,sKeyMode
cpi rmp,2
brcc AdcSet0_1 ; In input mode
AdcSet0Ret:
ret ; Done
AdcSet0_1:
lds rmp,sKeyCnt ; Read key counter
tst rmp ; Key counter at zero?
breq AdcSet0a ; At zero, go on
dec rmp ; Decrease key counter
sts sKeyCnt,rmp ; and store
rjmp AdcSet0Ret ; No further action
AdcSet0a:
ldi rmp,5 ; Restart 5 cycles delay
sts sKeyCnt,rmp
rcall GetMinMax ; Get the minimum (R0) and maximum (R1)
mov rmp,R1 ; Maximum - Minimum
sub rmp,R0 ; Subtract minimum
mov XL,rAdc ; Multiply rAdc by difference
clr XH
clr ZL ; Z is result
mov ZH,R0 ; Add minimum to MSB
AdcSet0b:
tst rmp ; Check zero reached?
breq AdcSet0d ; Yes, completed multiplication
lsr rmp ; Shift lowest bit to carry
brcc AdcSet0c
add ZL,XL ; Add the multiplier
adc ZH,XH
AdcSet0c:
lsl XL ; Shift multiplier left
rol XH
rjmp AdcSet0b ; Continue multiplying
AdcSet0d:
ldi XH,High(sInputData) ; point to input buffer
ldi XL,Low(sInputData)
lds rmp,sKeyPos
add XL,rmp ; Add position
ldi rmp,0
adc XH,rmp ; with carry
st X,ZH ; Write Multiplication result
rcall DisplInp
rjmp ActivatePos ; Activate again
;
; Get minimum and maximum for position
;
minimum in R0, maximum +1 in R1
GetMinMax:
ldi ZH,High(2*MinMaxTab) ; Z to table
ldi ZL,Low(2*MinMaxTab)
clr R0 ; is carry adder
lds rmp,sKeyPos ; Key position
lsl rmp ; Multiply by 2
add ZL,rmp
adc ZH,R0 ; Add carry
lpm R0,Z+ ; Read minimum
lpm R1,Z ; Read maximum

ret
;
; Maxima table by position
MinMaxTab:
;
wd day mon year hour min sec alhr almin
.db 0,7,1,32,1,13,0,100,0,24,0,60,0,60,0,24,0,60
;
; ADC channel 1 complete
;
Display light sensor in hex
AdcSet1:
; Check if delay period over
dec rBackCnt ; Decrease counter
brne AdcSet1Ret ; Not zero
ldi rmp,cBackPeriods ; Backlight update period
mov rBackCnt,rmp ; Restart counter
ldi ZH,2 ; Display sensor result on LCD
ldi ZL,18
rcall LcdPos
mov rmp,rAdc
rcall LcdHex2
AdcSet1a:
tst rAdc ; Result zero?
brne Light0 ; No
inc rAdc ; Set to 1
Light0:
neg rAdc ; Revert
; Multiply light sensor result
ldi rmp,cBackMin ; Add minimum
mov R1,rmp ; To MSB in result in R1:R0
clr R0 ; Clear LSB
ldi ZL,cBackMult ; Load multiplicator
clr ZH ; Clear MSB
Light1: ; Multiply with constant
lsr rAdc ; Shift bit to carry
brcc Light2 ; Bit = zero
add R0,ZL ; Add to result
adc R1,ZH
brcc Light2 ; No carry
clr R1 ; Set result to FF
dec R1
Light2:
lsl ZL ; Multiply by 2
rol ZH
tst rAdc ; Check still ones?
brne Light1 ; Yes, continue multiplication
out OCR2,R1 ; Update backlight PWM
AdcSet1Ret:
ret
;
; Music code include routines
.include "music_code.inc"
;
; LCD include routines
.include "lcd.inc"
;
; LCD subroutines
;
; Blinking of the LCD on
;
LcdBlink:
ldi rmp,0x0F
rjmp LcdCtrl
;
; Blinking of the LCD off

;
LcdNoBlink:
ldi rmp,0x0C
rjmp LcdCtrl
;
.if fDbg
; If debugging active include routines
.include "debug_code_en.inc"
.endif
;
; Lcd text
;
LcdTxtInit1:
.if LangEN == 1
; English version
.db " DCF77_clock_m16 V5 ",0x0D,0xFF
.else
; German version
.db "DCF77_m16 V5 Atomuhr",0x0D,0xFF
.endif
.db "
",7,"2018 DG4FAC
",0x0D,0xFF
.db "
",0x0D,0xFF
.db "
",0xFE,0xFE
;
LcdTxtFrame:
.db " - . . - : : ",0x0D,0xFF
.if LangEN == 1
; English version
.db " Alarm: ___
:
",0x0D,0xFF
.db "DCF: S0 __ Light:
",0x0D,0xFF
.else
; German version
.db "Wecken: ___
:
",0x0D,0xFF
.db "DCF: S0 __ Licht:
",0x0D,0xFF
.endif
.db "
",0xFE,0xFE
;
; Special characters
CodeChars:
.db 64,0,21,10,21,10,21,10,21,0 ; C = 0, Multi1
.db 72,0,4,4,4,21,31,14,4,0 ; C = 1, Arrow down
.db 80,0,4,14,31,21,4,4,4,0 ; C = 2, Arrow up
.db 88,0,0,0,8,21,3,7,0,0 ; C = 3, Arrow right down
.db 96,0,0,0,2,21,24,28,0,0 ; C = 4, Arrow right down
.db 104,0,4,10,4,10,4,10,4,0 ; C = 5, Dummy
.db 112,0,10,21,10,21,10,21,10,0 ; C = 6, Multi2
.db 120,0,14,17,21,25,21,17,14,0 ; C = 7, Arrow left up
.db 0,0 ; End of Table
;
; Edit positions
.equ cAlarmPos = 8
.equ cDcfPos = 8
;
; End of source code
; Copyright
.db "(C)2018 by Gerhard Schmidt " ; human readable
.db "C(2)10 8ybG reahdrS hcimtd " ; wordwise
;

2 The LCD routines lcd.inc
Link to this code in asm format

*********************************
* LCD include routines
*
* (C)2018 avr-asm-tutorial.net *
*********************************
***********************************************
* L C D
I N T E R F A C E R O U T I N E S *
***********************************************
+-------+----------------+--------------+
|Routine|Function
|Parameters
|
+-------+----------------+--------------+
|LcdInit|Inits the LCD
|Ports, pins
|
|
|in the desired |
|
|
|mode
|
|
+-------+----------------+--------------+
|LcdText|Displays the
|Z=2*Table
|
|
|text in flash
| address
|
|
|memory
|0x0D: Next
|
|
|
|
line
|
|
|
|0xFF: Ignore |
|
|
|0xFE: Text end|
+-------+----------------+--------------+
|LcdSram|Display the text|Z=SRAM-Address|
|
|in SRAM
|R16: number of|
|
|
|
characters|
+-------+----------------+--------------+
|LcdChar|Display charac- |R16: Character|
|
|ter on LCD
|
|
+-------+----------------+--------------+
|LcdCtrl|Output control |R16: Control |
|
|byte to LCD
|
byte
|
+-------+----------------+--------------+
|LcdPos |Set position on |ZH: Line 0123 |
|
|the LCD
|ZL: Col 0..
|
+-------+----------------+--------------+
|LcdSpec|Generate special|Z: 2*Table
|
|
|characters
|
address
|
+-------+----------------+--------------+
|
S W I T C H
L C D D E C I M A L |
+-------+----------------+--------------+
|LcdDec2|Convert to two |R16: Binary
|
|
|decimal digits |
8 bit
|
+-------+----------------+--------------+
|LcdDec3|Convert to three|R16: Binary
|
|
|decimal digits, |
8 bit
|
|
|supp. leading 0s|
|
+-------+----------------+--------------+
|LcdDec5|Convert to five |Z: Binary
|
|
|decimal digits, |
16 bit
|
|
|supp. leading 0s|
|
+-------+----------------+--------------+
|
S W I T C H
L C D H E X
|
+-------+----------------+--------------+
|LcdHex2|Convert to two |R16: Binary
|
|
|digits in hex
|
8 bit
|
+-------+----------------+--------------+
|LcdHex4|Convert to four |Z: Binary
|

; |
|digits in hex
|
16 bit
|
; +-------+----------------+--------------+
;
; *************************************
; P A R A M E T E R - T E M P L A T E
; *************************************
;
; (Copy to your source code, remove ; and adjust
; parameters to fit your hardware)
;
; Standard parameter set of properties/definitions
;.equ clock = 1000000 ; Clock frequency of controller in Hz
; LCD size:
;.equ LcdLines = 1 ; Number of lines (1, 2, 4)
;.equ LcdCols = 8 ; Number of characters per line (8..24)
; LCD bus interface
;.equ LcdBits = 4 ; Bus size (4 or 8)
; If 4 bit bus:
;.equ Lcd4High = 1 ; Bus nibble (1=Upper, 0=Lower)
;.equ LcdWait = 0 ; Access mode (0 with busy, 1 with delay loops)
; LCD data ports
;.equ pLcdDO = PORTA ; Data output port
;.equ pLcdDD = DDRA ; Data direction port
; LCD control ports und pins
;.equ pLcdCEO = PORTB ; Control E output port
;.equ bLcdCEO = PORTB0 ; Control E output portpin
;.equ pLcdCED = DDRB ; Control E direction port
;.equ bLcdCED = DDB0 ; Control E direction portpin
;.equ pLcdCRSO = PORTB ; Control RS output port
;.equ bLcdCRSO = PORTB1 ; Control RS output portpin
;.equ pLcdCRSD = DDRB ; Control RS direction port
;.equ bLcdCRSD = DDB1 ; Control RS direction portpin
; If LcdWait = 0:
;.equ pLcdDI = PINA ; Data input port
;.equ pLcdCRWO = PORTB ; Control RW output port
;.equ bLcdCRWO = PORTB2 ; Control RW output portpin
;.equ pLcdCRWD = DDRB ; Control RW direction port
;.equ bLcdCRWD = DDB2 ; Control RW direction portpin
; If you need binary to decimal conversion:
;.equ LcdDecimal = 1 ; If defined: include those routines
; If you need binary to hexadecimal conversion:
;.equ LcdHex = 1 ; If defined: include those routines
; If simulation in the SRAM is desired:
;.equ avr_sim = 1 ; 1=Simulate, 0 or undefined=Do not simulate
;
; *****************************************
;
T E X T
T E M P L A T E S
; *****************************************
;
; Tables to copy for diverse sizes
;
; -------------------------; Single line LCD
;
8 chars per line
; Text_1_8:
;
.db "
",0xFE,0xFF
;
01234567
;
;
16 chars per line
; Text_1_16:
;
.db "
",0xFE,0xFF
;
0123456789012345
;
;
20 chars per line

; Text_1_20:
;
.db "
",0xFE,0xFF
;
01234567890123456789
;
;
24 chars per line
; Text_1_24:
;
.db "
",0xFE,0xFF
;
012345678901234567890123
;
; -------------------------; Two line LCD
;
16 chars per line
; Text_2_16:
;
.db "
",0x0D,0xFF
;
.db "
",0xFE,0xFF
;
0123456789012345
;
;
20 chars per line
; Text_2_20:
;
.db "
",0x0D,0xFF
;
.db "
",0xFE,0xFF
;
01234567890123456789
;
;
24 chars per line
; Text_2_24:
;
.db "
",0x0D,0xFF
;
.db "
",0xFE,0xFF
;
012345678901234567890123
;
; -------------------------; Four line LCD
;
16 chars per line
; Text_4_16:
;
.db "
",0x0D,0xFF
;
.db "
",0x0D,0xFF
;
.db "
",0x0D,0xFF
;
.db "
",0xFE,0xFF
;
0123456789012345
;
;
20 chars per line
; Text_4_20:
;
.db "
",0x0D,0xFF
;
.db "
",0x0D,0xFF
;
.db "
",0x0D,0xFF
;
.db "
",0xFE,0xFF
;
01234567890123456789
;
;
24 chars per line
; Text_4_24:
;
.db "
",0x0D,0xFF
;
.db "
",0x0D,0xFF
;
.db "
",0x0D,0xFF
;
.db "
",0xFE,0xFF
;
012345678901234567890123
;
; *******************************
; P A R A M E T E R
C H E C K
; *******************************
;
; Are all parameters correct?
;
; Size defined?
.ifndef LcdLines
.error "LCD line size (LcdLines) undefined!"

.else
.if (LcdLines!=1)&&(LcdLines!=2)&&(LcdLines!=4)
.error "LCD illegal line size (LcdLines)!"
.endif
.endif
.ifndef LcdCols
.error "LCD column size (LcdCols) undefined!"
else
.if (LcdCols<8)||(LcdCols>24)
.error "LCD illegal column size (LcdCols)!"
.endif
.endif
;
; Clock defined?
.ifndef clock
.error "Clock frequency (clock) undefined!"
.endif
;
; 4- or 8-bit interface selected?
.ifndef LcdBits
.error "LCD data bus bits (LcdBits) undefined!"
.else
.if (LcdBits != 4) && (LcdBits != 8)
.error "LCD data bus bits (LcdBits) not 4 or 8!"
.endif
.if LcdBits == 4
.ifndef Lcd4High
.error "LCD 4 bit data bus nibble (Lcd4High) undefined!"
.else
.if (Lcd4High != 0) && (Lcd4High != 1)
.error "LCD 4 bit data bus nibble (Lcd4High) not 0 or 1!"
.endif
.endif
.endif
.endif
;
; LCD data ports
.ifndef pLcdDO
.error "LCD data output port (pLcdDO) undefined!"
.endif
.ifndef pLcdDD
.error "LCD data direction port (pLcdDD) undefined!"
.endif
.if LcdWait == 0
.ifndef pLcdDI
.error "LCD data input port (pLcdDI) undefined!"
.endif
.endif
;
; LCD control ports und pins
.ifndef pLcdCEO
.error "LCD control E output port (pLcdCEO) undefined!"
.endif
.ifndef pLcdCED
.error "LCD control E direction port (pLcdCED) undefined!"
.endif
.ifndef bLcdCEO
.error "LCD control E output pin (bLcdCEO) undefined!"
.endif
.ifndef bLcdCED
.error "LCD control E direction pin (bLcdCED) undefined!"
.endif
.ifndef pLcdCRSO
.error "LCD control RS output port (pLcdCRSO) undefined!"

.endif
.ifndef pLcdCRSD
.error "LCD control RS direction port (pLcdCRSD) undefined!"
.endif
.ifndef bLcdCRSO
.error "LCD control RS output pin (bLcdCRSO) undefined!"
.endif
.ifndef bLcdCRSD
.error "LCD control RS direction pin (bLcdCRSD) undefined!"
.endif
.ifndef LcdWait
.error "LCD operating property (LcdWait) undefined!"
.else
.if LcdWait == 0
.ifndef pLcdCRWO
.error "LCD control RW output port (pLcdCRWO) undefined!"
.endif
.ifndef bLcdCRWO
.error "LCD control RW output pin (bLcdCRWO) undefined!"
.endif
.ifndef pLcdCRWD
.error "LCD control RW direction port (pLcdCRWD) undefined!"
.endif
.ifndef bLcdCRWD
.error "LCD control RW direction pin (bLcdCRWD) undefined!"
.endif
.endif
.endif
;
; *************************************
; S I M U L A T I O N
A V R _ S I M
; *************************************
;
; Check if simulation desired
.ifdef avr_sim
.equ simulation = avr_sim
.else
.equ simulation = 0
.endif
.if simulation == 1
.dseg
SimStart:
SimDisplayPos:
.byte 1
SimCtrlClear:
.byte 1
SimCtrlReset:
.byte 1
SimCtrlInputmode:
.byte 1
SimCtrlDisplay:
.byte 1
SimCtrlCursorShift:
.byte 1
SimCtrlFunctionset:
.byte 1
SimCtrlCharGenRamAdr:
.byte 1
SimCtrlDisplayRamAdr:
.byte 1
SimDataDisplay:
.byte LcdLines*LcdCols
SimEnd:
.cseg

.endif
;
; *********************************
;
L C D
R O U T I N E S
; *********************************
;
; LcdInit: Init LCD ports and pins
;
Wait cycle for LCD start-up
;
Function set
;
Clear LCD
LcdInit:
; Init the LCD control bits
cbi pLcdCEO,bLcdCEO ; E pin low
sbi pLcdCED,bLcdCED ; E pin output
cbi pLcdCRSO,bLcdCRSO ; RS pin low
sbi pLcdCRSD,bLcdCRSD ; RS pin output
.ifdef pLcdCRWO
.ifdef bLcdCRWO
cbi pLcdCRWO,bLcdCRWO ; RW pin low
.endif
.endif
.ifdef pLcdCRWD
.ifdef pLcdCRWD
sbi pLcdCRWD,bLcdCRWD ; RW pin output
.endif
.endif
; Init the LCD data bus ports
.if LcdBits == 8
clr R16 ; Data bus to low
.else
in R16,pLcdDO ; Read output bits data bus
.if Lcd4High == 1
andi R16,0x0F ; Clear upper nibble
.else
andi R16,0xF0 ; Clear lower nibble
.endif
.endif
out pLcdDO,R16 ; Data bus output clear
.if LcdBits == 8
ldi R16,0xFF ; Set all direction bits high
.else
in R16,pLcdDD ; Read direction bits data bus
.if Lcd4High == 1
ori R16,0xF0 ; Set upper nibble
.else
ori R16,0x0F ; Set lower nibble
.endif
.endif
out pLcdDD,R16 ; Set direction bits data bus
; LCD-Startphase
.if simulation == 0
rcall LcdWait50ms
.endif
; LCD to 8 bit data bus
ldi R16,0x30
rcall Lcd8Ctrl ; In 8 bit mode to LCD control
.if simulation == 0
rcall LcdWait5ms ; Wait for 5 ms
.endif
ldi R16,0x30
rcall Lcd8Ctrl ; In 8 bit mode to LCD control
.if simulation == 0
rcall LcdWait5ms ; Wait for 5 ms
.endif

ldi R16,0x30
rcall Lcd8Ctrl ; In 8 bit mode to LCD control
.if simulation == 0
rcall LcdWait5ms ; Wait for 5 ms
.endif
ldi R16,0x30
rcall Lcd8Ctrl ; In 8 bit mode to LCD control
.if simulation == 0
rcall LcdWait5ms ; Wait for 5 ms
.endif
; If 4 bit data bus interface: switch to 4 bit mode
.if LcdBits == 4
ldi R16,0x20 ; 4 bit interface
rcall Lcd8Ctrl ; In 8 bit mode to LCD control
.if simulation == 0
rcall LcdWait5ms
.endif
.endif
; Function set
.if LcdBits == 8
ldi R16,0x30 ; 8 bit data bus
.else
ldi R16,0x20 ; 4 bit data bus
.endif
.if LcdLines > 1
ori R16,0x08 ; LCDs with more than one line
.endif
rcall LcdCtrl
; Display mode
ldi R16,0x0C ; Display on, underline off, cursor blink off
rcall LcdCtrl
; LCD entry mode set
ldi R16,0x06 ; Cursor right, cursor move not display shift
rcall LcdCtrl
; Clear LCD
ldi R16,0x01 ; LCD clear
rjmp LcdCtrl

;
; LcdText
;
Displays the text in flash memory on the LCD
;
Z points to 2*Text table address
;
0x0D: Line feed and carriage return
;
0xFF: Ignore (fill character, ignored)
;
0xFE: End of the text
LcdText:
push R0 ; Save R0
ldi R16,LcdLines ; Maximum line number
mov R0,R16 ; to R0
LcdText1:
lpm R16,Z+ ; Read next char from flash
cpi R16,0xFE ; End of text?
breq LcdText3 ; Yes
brcc LcdText1 ; Ignore fill characters
cpi R16,0x0D ; Line feed and carriage return?
brne LcdText2 ; No
dec R0 ; Decrease line counter
breq LcdText1 ; If zero, continue
.if simulation == 1
; CR/LF in SRAM memory
push ZH ; Save Z
push ZL
push R16 ; Save character
ldi ZH,High(SimDataDisplay) ; Perform newline
ldi ZL,Low(SimDataDisplay)

ldi R16,LcdLines
LcdText1a:
adiw ZL,LcdCols
dec R16
cp R16,R0
brne LcdText1a
ldi R16,Low(SimDataDisplay)
sub ZL,R16
ldi R16,High(SimDataDisplay)
sbc ZH,R16
sts SimDisplayPos,ZL
pop R16 ; Restore character
pop ZL ; Restore Z
pop ZH
.endif
; CR/LF on LCD
push ZH ; Save Z
push ZL
ldi ZH,LcdLines ; Calculate line
sub ZH,R0
clr ZL ; To line start
rcall LcdPos ; Set LCD position
pop ZL ; Restore Z
pop ZH
rjmp LcdText1 ; Next character in flash
LcdText2:
rcall LcdChar ; Character in R16 to LCD
rjmp LcdText1 ; Next character
LcdText3:
pop R0 ; Restore R0
ret
;
; LcdSRam displays text in SRAM at the current LCD position
;
Z points to SRAM address
;
R16: Number of characters
LcdSRam:
push R16 ; Save R16
ld R16,Z+ ; Read character
rcall LcdChar ; Display on LCD
pop R16 ; Restore R16
dec R16 ; Downcount number of characters
brne LcdSRam ; Further characters to display
ret
;
; LcdChar displays character in R16 at the current position
;
R16: Character
LcdChar:
.if simulation == 1
; Simulation, write character to SRAM
push ZH ; Save Z
push ZL
push R16 ; Save character
ldi ZH,High(SimDataDisplay) ; SRAM position
ldi ZL,Low(SimDataDisplay)
lds R16,SimDisplayPos
inc R16
sts SimDisplayPos,R16
dec R16
add ZL,R16
ldi R16,0
adc ZH,R16
pop R16
st Z,R16 ; Write to SRAM
pop ZL

pop ZH
.endif
rjmp LcdData ; Write as data to LCD

;
; LcdCtrl writes control byte in R16 on the LCD
LcdCtrl:
.if simulation == 1
cpi R16,0x80 ; Display RAM address write?
brcs LcdCtrlSim1
sts SimCtrlDisplayRamAdr,R16
push ZH
push ZL
mov ZH,R16
andi R16,0x40
brne LcdCtrlSim0a
clr ZL ; Line 1 or 3
rjmp LcdCtrlSim0b
LcdCtrlSim0a:
ldi ZL,LcdCols ; Line 2 or 4
LcdCtrlSim0b:
mov R16,ZH
andi R16,0x3F
subi R16,LcdCols
brcc LcdCtrlSim0c ; Line 3 or 4
subi R16,-LcdCols
rjmp LcdCtrlSim0d
LcdCtrlSim0c:
sbrc ZH,6 ; Line 4?
subi R16,-LcdCols ; Add column length
LcdCtrlSim0d:
add ZL,R16
sts SimDisplayPos,ZL
pop ZL
pop ZH
rjmp LcdCtrlSim9
LcdCtrlSim1:
cpi R16,0x40 ; Character generator RAM address set?
brcs LcdCtrlSim2
sts SimCtrlCharGenRamAdr, R16
rjmp LcdCtrlSim9
LcdCtrlSim2:
cpi R16,0x20 ; Function set?
brcs LcdCtrlSim3
sts SimCtrlFunctionSet, R16
rjmp LcdCtrlSim9
LcdCtrlSim3:
cpi R16,0x10 ; Cursor shift?
brcs LcdCtrlSim4
sts SimCtrlCursorShift, R16
rjmp LcdCtrlSim9
LcdCtrlSim4:
cpi R16,0x08 ; Display control set?
brcs LcdCtrlSim5
sts SimCtrlDisplay, R16
rjmp LcdCtrlSim9
LcdCtrlSim5:
cpi R16,0x04 ; Display control?
brcs LcdCtrlSim6
sts SimCtrlInputmode, R16
rjmp LcdCtrlSim9
LcdCtrlSim6:
cpi R16,0x02 ; Reset?
brcs LcdCtrlSim7
sts SimCtrlReset,R16

rjmp LcdCtrlSim9
LcdCtrlSim7:
cpi R16,0x01 ; Clear?
brcs LcdCtrlSim8
sts SimCtrlClear,R16
; LCD clear display
push ZH
push ZL
ldi ZH,High(SimEnd)
ldi ZL,Low(SimEnd)
clr R16
LcdCtrl7a:
st -Z,R16 ; Fill with zeroes backwards
cpi ZL,Low(SimStart)
brne LcdCtrl7a
cpi ZH,High(SimStart)
brne LcdCtrl7a
ldi R16,0x01
pop ZL
pop ZH
rjmp LcdCtrlSim9
LcdCtrlSim8: ; 00 command
LcdCtrlSim9:
.endif
.if LcdWait == 0
rcall LcdBusy ; Wait for busy flag clear
.endif
cbi pLcdCRSO,bLcdCRSO ; RS bit low
.if LcdBits == 4
push ZL
push R16
in ZL,pLcdDO ; Read data output port
.if Lcd4High == 1
andi ZL,0x0F ; Clear upper nibble port
andi R16,0xF0 ; Clear lower nibble R16
.else
andi ZL,0xF0 ; Clear lower nibble port
swap R16 ; Upper to lower nibble
andi R16,0x0F ; Clear upper nibble R16
.endif
or R16,ZL ; Combine
out pLcdDO,R16 ; and output
rcall LcdPulseE ; Activate E
pop R16 ; R16 restore
push R16 ; and save again
in ZL,pLcdDO ; Read data output port
.if Lcd4High == 1
andi ZL,0x0F ; Preserve lower nibble of port
swap R16 ; Lower to upper nibble R16
andi R16,0xF0 ; Preserve upper nibble R16
.else
andi ZL,0xF0 ; Preserve upper nibble of port
andi R16,0x0F ; Preserve lower nibble R16
.endif
out pLcdDO,R16 ; To data bus
rcall LcdPulseE ; Activate E
pop R16 ; Restore R16
pop ZL ; Restore ZL
.else
out pLcdDO,R16 ; Input on data bus
rcall LcdPulseE ; Activate E
.endif
.if LcdWait == 1
andi R16,0xFC ; The six upper bits of control command

brne LcdCtrl1 ; are not zero, short delay
rjmp LcdWait1640us ; Wait 1,64 ms
LcdCtrl1:
rjmp LcdWait40us ; Wait 40 us
.else
ret
.endif

;
; LcdPos position the LCD cursor to the position in Z
;
ZH: Line (0 to number of lines - 1)
;
ZL: Column (0 to number of colums per line - 1)
LcdPos:
cpi ZH,1 ; Line = 1?
ldi R16,0x80 ; Address line 0
.if LcdLines < 2 ; LCD has only one line
rjmp LcdPos1
.endif
brcs LcdPos1 ; Line 1
ldi R16,0xC0 ; Line 2 address
.if LcdLines == 2
rjmp LcdPos1 ; Lcd has only two lines
.endif
breq LcdPos1 ; Line 2 selected
ldi R16,0x80+LcdCols ; Address line 3
cpi ZH,2 ; Line = 3
breq LcdPos1 ; Line 3
ldi R16,0xC0+LcdCols ; Address line 4
LcdPos1:
add R16,ZL ; Add column
rjmp LcdCtrl ; Send as control to LCD
;
; LcdSpec generates special characters on the LCD
;
Z points to 2*Table address
;
Table format:
;
1. byte: Address of character, 0b01zzz000,
;
0: End of the table
;
2. byte: Dummy character, ignored
;
3. to 10 th byte: Pixel data of the lines 1 to 8
LcdSpec:
push R0 ; R0 is counter
LcdSpec1:
lpm R16,Z+ ; Read address of character
tst R16 ; End of the table?
breq LcdSpec3 ; Yes
rcall LcdCtrl ; Write address
adiw ZL,1 ; Overread dummy
ldi R16,8 ; 8 byte per character
mov R0,R16 ; R0 is counter
LcdSpec2:
lpm R16,Z+ ; Read data byte from table
rcall LcdData ; Output as data byte
dec R0 ; Count down
brne LcdSpec2 ; Further data bytes
rjmp LcdSpec1 ; Next character
LcdSpec3:
pop R0 ; Restore R0
ldi ZH,0 ; Cursor to home position
ldi ZL,0
rjmp LcdPos
;
; ************************************
; D E C I M A L C O N V E R S I O N
; ************************************
;

; Routines for conversion and display decimals
;
.ifdef LcdDecimal
;
; LcdDec2 converts binary in R16 to a two digit decimal
;
without suppression of leading zeroes
LcdDec2:
mov ZL,R16 ; Copy number
ldi R16,'0'-1 ; Tens counter
cpi ZL,100 ; Larger that two digits?
brcs LcdDec2a ; No
ldi ZL,99 ; three digits, limit to max
LcdDec2a:
inc R16
subi ZL,10
brcc LcdDec2a
rcall LcdData ; Display tens
ldi R16,10+'0'
add R16,ZL
rcall LcdData ; Display ones
ret
;
; LcdDec3 converts binary in R16 to three digits decimal
;
blanks leading zeroes
LcdDec3:
ldi ZH,1 ; Flag leading zeroes
LcdDec3null: ; without changing leading zero flag
push R0 ; Save R0
mov ZL,R16 ; Binary in ZL
ldi R16,100 ; Hundreds
rcall LcdDec3a ; Convert to decimal
rcall LcdData ; Display digit
ldi R16,10 ; Tens
rcall LcdDec3a ; Convert to decimal
rcall LcdData ; Display digit
ldi R16,'0'
add R16,ZL ; Convert ones to ASCII
rcall LcdData ; Display digit
pop R0 ; Restore R0
ret
;
LcdDec3a: ; Calculate decimal digit
clr R0 ; Counter
dec R0 ; to - 1
LcdDec3b:
inc R0 ; Increase counter
sub ZL,R16 ; Subtract 100 or 10
brcc LcdDec3b ; No overflow, continue
add ZL,R16 ; Recover last subtraction
tst R0 ; Digit zero?
breq LcdDec3c ; Yes, check blanking
clr ZH ; No blanking any more
ldi R16,'0' ; ASCII-0
add R16,R0 ; Add counter
ret
LcdDec3c:
tst ZH ; Zero blanking?
breq LcdDec3d ; No
ldi R16,' ' ; Blank char
ret
LcdDec3d:
ldi R16,'0' ; Zero char
ret
;

; LcdDec5 converts binary Z to five decimal digits
;
blank leading zeroes
LcdDec5:
push R2 ; Save registers
push R1
push R0
clr R2 ; Leading zeroes
inc R2 ; Blanking on
mov R1,ZH ; Copy binary to R1:R0
mov R0,ZL
ldi ZH,High(10000) ; First decimal digit
ldi ZL,Low(10000)
rcall LcdDec5a ; Convert first digit
ldi ZH,High(1000) ; Second decimal digit
ldi ZL,Low(1000)
rcall LcdDec5a ; Convert second digit
ldi ZH,High(100) ; Third decimal digit
ldi ZL,Low(100)
rcall LcdDec5a ; Convert decimal digit
ldi ZH,High(10) ; Fourth decimal digit
ldi ZL,Low(10)
rcall LcdDec5a ; Convert decimal digit
ldi R16,'0' ; Convert fifth digit to ASCII
add R16,R0
rcall LcdChar ; Display digit
pop R0 ; Restore registers
pop R1
pop R2
ret
;
LcdDec5a: ; Convert to decimal digit
ldi R16,'0'-1 ; To ASCII-0 - 1
LcdDec5b:
inc R16 ; Increase counter
sub R0,ZL ; Subtract decimal digit LSB
sbc R1,ZH ; and MSB with Carry
brcc LcdDec5b ; Repeat until carry
add R0,ZL ; Take back last subtraction
adc R1,ZH
cpi R16,'0' ; Zero?
breq LcdDec5c ; Yes, check blanking
clr R2 ; End blanking
rjmp LcdChar ; Display char in R16
LcdDec5c:
tst R2 ; Blanking active?
breq LcdDec5d ; No
ldi R16,' ' ; Blank character
LcdDec5d:
rjmp LcdChar ; Display char in R16
;
.endif

;
; ****************************************
; H E X A D E C I M A L A U S G E B E N
; ****************************************
;
; Routines for conversion and displaying hexadecimals
;
.ifdef LcdHex
;
; LcdHex2 converts binary in R16 to hexadecimal
;
on the LCD
LcdHex2:
push R16 ; Is further needed

swap R16 ; Upper nibble to lower nibble
rcall LcdNibble ; Display lower nibble
pop R16 ; Restore binary
LcdNibble:
andi R16,0x0F ; Clear upper nibble
subi R16,-'0' ; Add ASCII-0
cpi R16,'9'+1 ; A to F?
brcs LcdNibble1 ; No
subi R16,-7 ; Add 7
LcdNibble1:
rjmp LcdChar ; Display char on LCD
;
; LcdHex4 converts binary in Z to hexadecimal
LcdHex4:
mov R16,ZH ; MSB to R16
rcall LcdHex2 ; Display byte hexadecimal
mov R16,ZL ; LSB to R16
rjmp LcdHex2 ; Display byte hexadecimal
;
.endif
;
; *******************************
; L C D
S U B R O U T I N E S
; *******************************
;
; Wait until LCD busy flag clear
;
needed only if wait mode is off
.if LcdWait == 0
LcdBusy:
push R16 ; Save R16
.if LcdBits == 8
clr R16 ; Data bus direction to input
.else
in R16,pLcdDD ; Read direction bits
.if Lcd4High == 1
andi R16,0x0F ; Clear upper nibble
.else
andi R16,0xF0 ; Clear lower nibble
.endif
.endif
out pLcdDD,R16 ; Write to direction port
.if LcdBits == 8
clr R16 ; All output pins to low
.else
in R16,pLcdDO ; Read output port
.if Lcd4High == 1
andi R16,0x0F ; Clear upper nibble
.else
andi R16,0xF0 ; Clear lower nibble
.endif
.endif
out pLcdDO,R16 ; Clear pull-Ups
cbi pLcdCRSO,bLcdCRSO ; RS pin to low
sbi pLcdCRWO,bLcdCRWO ; RW pin to high
LcdBusyWait:
rcall LcdIn ; Activate E, read data port, deactivate E
.if LcdBits == 4
rcall LcdPulseE ; Dummy for lower nibble
.endif
lsl R16 ; Busy flag to carry
brcs LcdBusyWait ; Flag = 1, wait on
cbi pLcdCRWO,bLcdCRWO ; RW pin to low
.if LcdBits == 8
ldi R16,0xFF ; Data to output

.else
in R16,pLcdDD ; Read direction data port
.if Lcd4High == 1
ori R16,0xF0 ; Upper nibble high
.else
ori R16,0x0F ; Lower nibble high
.endif
.endif
out pLcdDD,R16 ; Set direction port
pop R16 ; Restore R16
ret ; Done

;
; Read busy flag to R16
;
Needed only if wait mode off
LcdIn:
cbi pLcdCRSO,bLcdCRSO ; LCD RS pin to low
sbi pLcdCEO,bLcdCEO ; Set bit bLcdCEO on LCD control port
nop ; Wait at least one clock cycle
.if clock>1000000 ; Insert further NOPs for higher clocks
nop
.endif
.if clock>2000000
nop
.endif
.if clock>3000000
nop
.endif
.if clock>3000000
nop
.endif
.if clock>4000000
nop
.endif
.if clock>5000000
nop
.endif
.if clock>6000000
nop
.endif
.if clock>7000000
nop
.endif
.if clock>8000000
nop
.endif
.if clock>9000000
nop
.endif
.if clock>10000000
nop
.endif
.if clock>11000000
nop
.endif
.if clock>12000000
nop
.endif
.if clock>13000000
nop
.endif
.if clock>14000000
nop
.endif
.if clock>15000000

nop
.endif
.if clock>16000000
nop
.endif
.if clock>17000000
nop
.endif
.if clock>18000000
nop
.endif
.if clock>19000000
nop
.endif
in R16,pLcdDI ; Read data bus to R16
cbi pLcdCEO,bLcdCEO ; Clear bit bLcdCEO
ret
.endif
;
; 1 us pulse on LCD E pin
LcdPulseE:
sbi pLcdCEO,bLcdCEO ; Set bit bLcdCEO in LCD control port
.if clock>1000000 ; Add further NOPs for higher clocks
nop
.endif
.if clock>2000000
nop
.endif
.if clock>3000000
nop
.endif
.if clock>3000000
nop
.endif
.if clock>4000000
nop
.endif
.if clock>5000000
nop
.endif
.if clock>6000000
nop
.endif
.if clock>7000000
nop
.endif
.if clock>8000000
nop
.endif
.if clock>9000000
nop
.endif
.if clock>10000000
nop
.endif
.if clock>11000000
nop
.endif
.if clock>12000000
nop
.endif
.if clock>13000000
nop
.endif

.if clock>14000000
nop
.endif
.if clock>15000000
nop
.endif
.if clock>16000000
nop
.endif
.if clock>17000000
nop
.endif
.if clock>18000000
nop
.endif
.if clock>19000000
nop
.endif
cbi pLcdCEO,bLcdCEO ; Clear bit bLcdCEO
ret
;
; Write R16 in 8 bit mode to LCD control
Lcd8Ctrl:
.if simulation == 1
push R16
clr R16
sts SimDisplayPos,R16
pop R16
.endif
.if LcdBits == 4
push ZL ; Save ZL
in ZL,pLcdDO ; Read data output port
.if Lcd4High == 1
andi ZL,0x0F ; Clear upper nibble output
andi R16,0xF0 ; Clear lower nibble R16
.else
andi ZL,0xF0 ; Clear lower nibble output
swap R16 ; Swap upper and lower nibble
andi R16,0x0F ; Clear upper nibble R16
.endif
or R16,ZL ; Combine
pop ZL
.endif
out pLcdDO,R16 ; Data to LCD output port
cbi pLcdCRSO,bLcdCRSO ; RS bit low
rjmp LcdPulseE ; Activate E
;
; LcdData: Write data in R16 to LCD
;
R16: Character or data
LcdData:
.if LcdWait == 0
rcall LcdBusy ; Wait for busy flag of the LCD
.endif
sbi pLcdCRSO,bLcdCRSO ; Set RS bit
.if LcdBits == 4
push ZL
push R16
in ZL,pLcdDO ; Read data output port
.if Lcd4High == 1
andi ZL,0x0F ; Clear upper nibble port
andi R16,0xF0 ; Clear lower nibble R16
.else
andi ZL,0xF0 ; Clear lower nibble port
swap R16 ; Oberes und unteres Nibble vertauschen

andi R16,0x0F ; Clear upper nibble R16
.endif
or R16,ZL ; Combine
out pLcdDO,R16
rcall LcdPulseE ; Activate E
pop R16 ; Restore R16
push R16 ; and save again
in ZL,pLcdDO ; Read data output port
.if Lcd4High == 1
andi ZL,0x0F ; Preserve lower nibble of port
swap R16 ; Upper to lower nibble
andi R16,0xF0 ; Preserve upper nibble R16
.else
andi ZL,0xF0 ; Preserve upper nibble port
andi R16,0x0F ; Preserve lower nibble R16
.endif
or R16,ZL ; Combine
out pLcdDO,R16 ; To data bus
rcall LcdPulseE ; Activate E
pop R16 ; Restore R16
pop ZL ; Restore ZL
.else
out pLcdDO,R16 ; Byte to data bus
rcall LcdPulseE ; Activate E
.endif
.if LcdWait == 1
rjmp LcdWait40us ; Wait for 40 us
.endif
ret

;
; ********************************
;
W A I T
R O U T I N E S
; ********************************
;
; Wait for 50 ms
;
At clock frequencies above 3.1 MHz the maximum
;
counting capability of Z exceeds its limit,
;
therefore the 50 ms calls the 5 ms delay 10
;
times
LcdWait50ms:
push R16 ; Save R16
ldi R16,10 ; 10 times 5 ms
LcdWait50ms1:
rcall LcdWait5ms ; Call 5 ms delay
dec R16 ; Decrease counter
brne LcdWait50ms1 ; Repeat
pop R16 ; Restore R16
ret
;
LcdWait5ms:
; Wait for 5 ms, RCALL 3 clock cycles
.equ cLcdZ5ms = (5*clock/1000 - 18 + 2) / 4
push ZH ; Save ZH, + 2 cycles
push ZL ; Save ZL, + 2 cycles
ldi ZH,High(cLcdZ5ms) ; + 1 cycle
ldi ZL,Low(cLcdZ5ms) ; + 1 cycle
rjmp LcdWaitZ ; + 2 cycles
; Total: 11 clock cycles
;
.if LcdWait == 1 ;
; Wait routines necessary for wait mode
; Wait 1.64 ms
.equ cLcdZ1640us = (164*(clock/1000)/100 - 18 + 2) / 4
LcdWait1640us:

push ZH ; Save ZH
push ZL ; Save ZL
ldi ZH,High(cLcdZ1640us)
ldi ZL,Low(cLcdZ1640us)
rjmp LcdWaitZ
;
; Wait 40 us
.equ cLcdZ40us = (40*clock/1000000 - 18 + 2) / 4
LcdWait40us:
push ZH ; Save ZH
push ZL ; Save ZL
ldi ZH,High(cLcdZ40us)
ldi ZL,Low(cLcdZ40us)
rjmp LcdWaitZ
.endif

;
; Wait routine for Z cycles
LcdWaitZ: ; 11 clock cycles
sbiw ZL,1 ; Count down, +2 cycles
brne LcdWaitZ ; If not zero: go on, + 2 cycles; if zero: 1 cycle
pop ZL ; Restore ZL, +2 cycles
pop ZH ; Restore ZH, +2 cycles
ret ; Back, 4 cycles
; Total cycles: 11 + 4 * (Z - 1) + 3 + 8
;
; Number cycles= 11 + 4 * (Z - 1) + 3 + 8
;
11: RCALL, PUSH, PUSH, LDI, LDI, RJMP
;
4: 4 cycles per Z count (minus 1)
;
3: 3 cycles for last count
;
8: POP, RET
;
= 4 * Z + 11 - 4 + 3 + 8
;
= 4 * Z + 18
; Z = (Clock cycles - 18 + 2) / 4
;
+2: Rounding up during division by 4
;
; End of include file
;

3 The music music_code.inc
Link to this code in asm format
;
; ***********************************
; * Playing music with the ATmega16 *
; * (C)2018 by avr-asm-tutorial.net *
; ***********************************
;
; Convert a music piece to SRAM and start it
;
rmp points to a melody number (0..8)
MusicConvert:
cpi rmp,cMaxMelody+1 ; rmp inside number of melodies?
brcs MusicConvert1
subi rmp,cMaxMelody
brcc MusicConvert
subi rmp,-cMaxMelody ; Add number of melodies
MusicConvert1:
; Melody to play

ldi ZH,High(2*MelodyTab)
ldi ZL,Low(2*MelodyTab)
lsl rmp
add ZL,rmp
ldi rmp,0
adc ZH,rmp
lpm R0,Z+
lpm ZH,Z
mov ZL,R0
; Z points to melody
ldi XH,High(sMelody) ; To SRAM start of music
ldi XL,Low(sMelody)
MusicConvert2:
lpm rmp,Z+ ; Read note
cpi rmp,MelodyEnd ; End of melody?
breq MusicConvertEnd
rcall ConvertNote ; CTC to Y, Delay to R1:R0
lpm rmp,Z ; Read duration multiplier
andi rmp,0x0F ; Lower byte
MusicConvert3:
tst rmp
breq MusicConvert6
lsr R1 ; Divide duration by two
brne MusicConvert5 ; Not zero, without rounding
ror R0 ; MSB zero, rotate with rounding
brcc MusicConvert4 ; A one rotated to carry
inc R0 ; Round up
brne MusicConvert4 ; No carry
inc R1 ; Increase MSB
MusicConvert4:
dec rmp ; Decrease rotation
rjmp MusicConvert3
MusicConvert5:
ror R0 ; MSB larger than 0, rotate without rounding
dec rmp ; Decrease rotation
rjmp MusicConvert3
MusicConvert6:
lpm rmp,Z+ ; Re-read divider
swap rmp ; Upper nibble to lower nibble
andi rmp,0x0F ; Isolate adder
breq MusicConvert8 ; No adding
push ZH ; Save Z for adding
push ZL
mov ZH,R1 ; Copy current count
mov ZL,R0
MusicConvert7:
add R0,ZL ; Add once
adc R1,ZH
dec rmp ; Decrease Adder
brne MusicConvert7 ; Add on
pop ZL ; Restore Z
pop ZH
MusicConvert8:
mov rmp,R0 ; Counter at zero?
or rmp,R1 ; Or MSB
brne MusicConvert9
inc R0 ; Minimum one CTC cycle
MusicConvert9:
st X+,YH ; Write MSB CTC to SRAM
st X+,YL ; Then LSB CTC to SRAM
st X+,R1 ; MSB duration to SRAM
st X+,R0 ; LSB duration to SRAM
cpi XH,High(sMelodyEnd+4) ; End of buffer reached?
brcs MusicConvert2 ; No

cpi XL,Low(sMelodyEnd+4) ; LSB?
brcs MusicConvert2 ; No
MusicConvertEnd:
ldi rmp,0xFF ; End of melody
st X+,rmp
st X+,rmp
clr rmp ; Add 0x0000
st X+,rmp
st X,rmp
;
; Play music in SRAM (again)
MusicPlay:
ldi YH,High(sMelody) ; Point Y to SRAM buffer
ldi YL,Low(sMelody)
ldi rmp,(1<<TOIE0)|(1<<OCIE1A)|(1<<TOIE2) ; Enable int TC1
out TIMSK,rmp ; in Timer int mask
ldi rDurL,1 ; Start timer with one
clr rDurH
ret
;
; Converts a note in rmp
;
Returns CTC 16 bit value in Y
;
Duration 16 bit value in R1:R0
ConvertNote:
push ZH ; Save Z
push ZL
ldi ZH,High(2*NotesTimerTable)
ldi ZL,Low(2*NotesTimerTable)
mov R0,rmp
clr R1
lsl R0 ; Multiply by 4
rol R1
lsl R0
rol R1
add ZL,R0 ; Add to note table
adc ZH,R1
lpm YL,Z+
lpm YH,Z+
lpm R0,Z+
lpm R1,Z
pop ZL
pop ZH
ret
;
; Table of frequencies of music notes
;
For calculating notes from clock
;
fe renamed fee to avoid conflict with FE
;
(Framing Error, UART)!
NoteFrequencyTable:
.equ fA2m=55 ; #0, f = 27.5
.equ fH2m=62 ; #1, f = 30.8677
.equ fC1m=65 ; #2, f = 32.7032
.equ fD1m=73 ; #3, f = 36.7081
.equ fE1m=82 ; #4, f = 41.2034
.equ fF1m=87 ; #5, f = 43.6535
.equ fG1m=98 ; #6, f = 48.9994
.equ fA1m=110 ; #7, f = 55
.equ fH1m=123 ; #8, f = 61.7354
.equ fCm=131 ; #9, f = 65.4064
.equ fDm=147 ; #10, f = 73.4162
.equ fEm=165 ; #11, f = 82.4069
.equ fFm=175 ; #12, f = 87.3071
.equ fGm=196 ; #13, f = 97.9989
.equ fAm=220 ; #14, f = 110

.equ fHm=247 ; #15, f = 123.471
.equ fc=262 ; #16, f = 130.813
.equ fd=294 ; #17, f = 146.832
.equ fee=330 ; #18, f = 164.814
.equ ff=349 ; #19, f = 174.614
.equ fg=392 ; #20, f = 195.998
.equ fa=440 ; #21, f = 220
.equ fh=494 ; #22, f = 246.942
.equ fc1=523 ; #23, f = 261.626
.equ fd1=587 ; #24, f = 293.665
.equ fe1=659 ; #25, f = 329.628
.equ ff1=698 ; #26, f = 349.228
.equ fg1=784 ; #27, f = 391.995
.equ fa1=880 ; #28, f = 440
.equ fh1=988 ; #29, f = 493.883
.equ fc2=1047 ; #30, f = 523.251
.equ fd2=1175 ; #31, f = 587.33
.equ fe2=1319 ; #32, f = 659.255
.equ ff2=1397 ; #33, f = 698.456
.equ fg2=1568 ; #34, f = 783.991
.equ fa2=1760 ; #35, f = 880
.equ fh2=1976 ; #36, f = 987.767
.equ fc3=2093 ; #37, f = 1046.5
.equ fd3=2349 ; #38, f = 1174.66
.equ fe3=2637 ; #39, f = 1318.51
.equ ff3=2794 ; #40, f = 1396.91
.equ fg3=3136 ; #41, f = 1567.98
.equ fa3=3520 ; #42, f = 1760
.equ fh3=3951 ; #43, f = 1975.53
.equ fc4=4186 ; #44, f = 2093
.equ fd4=4699 ; #45, f = 2349.32
.equ fe4=5274 ; #46, f = 2637.02
.equ ff4=5588 ; #47, f = 2793.83
.equ fg4=6272 ; #48, f = 3135.96
.equ fa4=7040 ; #49, f = 3520
.equ fh4=7902 ; #50, f = 3951.07
.equ fc5=8372 ; #51, f = 4186.01
.equ fd5=9397 ; #52, f = 4698.65
.equ fe5=10548 ; #53, f = 5274.05
.equ ff5=11175 ; #54, f = 5587.67
.equ fg5=12544 ; #55, f = 6271.93
.equ fa5=14080 ; #56, f = 7040
.equ fh5=15804 ; #57, f = 7902.13
.equ fc6=16744 ; #58, f = 8372.02
.equ fd6=18795 ; #59, f = 9397.28
.equ fe6=21096 ; #60, f = 10548.08
.equ ff6=22351 ; #61, f = 11175.3
.equ fg6=25088 ; #62, f = 12543.86
.equ fa6=28160 ; #63, f = 14080
.equ fh6=31609 ; #64, f = 15804.26
.equ fc7=33488 ; #65, f = 16744.03
.equ fd7=37589 ; #66, f = 18794.56
.equ fe7=42192 ; #67, f = 21096.16
.equ ff7=44701 ; #68, f = 22350.59
.equ fg7=50175 ; #69, f = 25087.71
;
; Notes table: yields the CTC value for TC1 and
;
the number of CTC cycles for the duration
;
of one second for all music notes
; Adjusted for a 16 bit timer, clock frequency
;
in Hz and prescaler = 1
; Timer table for the music notes 0 to 70
;
First word: Compare value for CTC
;
Second word: Duration = number of CTC cycles

;
= 2*frequency of note
NotesTimerTable:
.dw clock/fA2m-1, 55 ; #0, f=27.5
.dw clock/fH2m-1, 62 ; #1, f=30.8677
.dw clock/fC1m-1, 65 ; #2, f=32.7032
.dw clock/fD1m-1, 73 ; #3, f=36.7081
.dw clock/fE1m-1, 82 ; #4, f=41.2034
.dw clock/fF1m-1, 87 ; #5, f=43.6535
.dw clock/fG1m-1, 98 ; #6, f=48.9994
.dw clock/fA1m-1, 110 ; #7, f=55
.dw clock/fH1m-1, 123 ; #8, f=61.7354
.dw clock/fCm-1, 131 ; #9, f=65.4064
.dw clock/fDm-1, 147 ; #10, f=73.4162
.dw clock/fEm-1, 165 ; #11, f=82.4069
.dw clock/fFm-1, 175 ; #12, f=87.3071
.dw clock/fGm-1, 196 ; #13, f=97.9989
.dw clock/fAm-1, 220 ; #14, f=110
.dw clock/fHm-1, 247 ; #15, f=123.471
.dw clock/fc-1, 262 ; #16, f=130.813
.dw clock/fd-1, 294 ; #17, f=146.832
.dw clock/fee-1, 330 ; #18, f=164.814
.dw clock/ff-1, 349 ; #19, f=174.614
.dw clock/fg-1, 392 ; #20, f=195.998
.dw clock/fa-1, 440 ; #21, f=220
.dw clock/fh-1, 494 ; #22, f=246.942
.dw clock/fc1-1, 523 ; #23, f=261.626
.dw clock/fd1-1, 587 ; #24, f=293.665
.dw clock/fe1-1, 659 ; #25, f=329.628
.dw clock/ff1-1, 698 ; #26, f=349.228
.dw clock/fg1-1, 784 ; #27, f=391.995
.dw clock/fa1-1, 880 ; #28, f=440
.dw clock/fh1-1, 988 ; #29, f=493.883
.dw clock/fc2-1, 1047 ; #30, f=523.251
.dw clock/fd2-1, 1175 ; #31, f=587.33
.dw clock/fe2-1, 1319 ; #32, f=659.255
.dw clock/ff2-1, 1397 ; #33, f=698.456
.dw clock/fg2-1, 1568 ; #34, f=783.991
.dw clock/fa2-1, 1760 ; #35, f=880
.dw clock/fh2-1, 1976 ; #36, f=987.767
.dw clock/fc3-1, 2093 ; #37, f=1046.5
.dw clock/fd3-1, 2349 ; #38, f=1174.66
.dw clock/fe3-1, 2637 ; #39, f=1318.51
.dw clock/ff3-1, 2794 ; #40, f=1396.91
.dw clock/fg3-1, 3136 ; #41, f=1567.98
.dw clock/fa3-1, 3520 ; #42, f=1760
.dw clock/fh3-1, 3951 ; #43, f=1975.53
.dw clock/fc4-1, 4186 ; #44, f=2093
.dw clock/fd4-1, 4699 ; #45, f=2349.32
.dw clock/fe4-1, 5274 ; #46, f=2637.02
.dw clock/ff4-1, 5588 ; #47, f=2793.83
.dw clock/fg4-1, 6272 ; #48, f=3135.96
.dw clock/fa4-1, 7040 ; #49, f=3520
.dw clock/fh4-1, 7902 ; #50, f=3951.07
.dw clock/fc5-1, 8372 ; #51, f=4186.01
.dw clock/fd5-1, 9397 ; #52, f=4698.65
.dw clock/fe5-1, 10548 ; #53, f=5274.05
.dw clock/ff5-1, 11175 ; #54, f=5587.67
.dw clock/fg5-1, 12544 ; #55, f=6271.93
.dw clock/fa5-1, 14080 ; #56, f=7040
.dw clock/fh5-1, 15804 ; #57, f=7902.13
.dw clock/fc6-1, 16744 ; #58, f=8372.02
.dw clock/fd6-1, 18795 ; #59, f=9397.28
.dw clock/fe6-1, 21096 ; #60, f=10548.08
.dw clock/ff6-1, 22351 ; #61, f=11175.3

.dw clock/fg6-1, 25088 ; #62, f=12543.86
.dw clock/fa6-1, 28160 ; #63, f=14080
.dw clock/fh6-1, 31609 ; #64, f=15804.26
.dw clock/fc7-1, 33488 ; #65, f=16744.03
.dw clock/fd7-1, 37589 ; #66, f=18794.56
.dw clock/fe7-1, 42192 ; #67, f=21096.16
.dw clock/ff7-1, 44701 ; #68, f=22350.59
.dw clock/fg7-1, 50175 ; #69, f=25087.71
; Pause
.dw 0xFFFF,clock/0xFFFF ; #70, f = clock/131130
;
; Note names
NotesNameTable:
.equ nA2m = 0 ; A2, f = 27.5 Hz, #0
.equ nH2m = 1 ; H2, f = 30.8677 Hz
.equ nC1m = 2 ; C1, f = 32.7032 Hz
.equ nD1m = 3 ; D1, f = 36.7081 Hz
.equ nE1m = 4 ; E1, f = 41.2034 Hz
.equ nF1m = 5 ; F1, f = 43.6535 Hz
.equ nG1m = 6 ; G1, f = 48.9994 Hz
.equ nA1m = 7 ; A1, f = 55 Hz
.equ nH1m = 8 ; H1, f = 61.7354 Hz
.equ nCm = 9 ; C, f = 65.4064 Hz
.equ nDm = 10 ; D, f = 73.4162 Hz
.equ nEm = 11 ; E, f = 82.4069 Hz
.equ nFm = 12 ; F, f = 87.3071 Hz
.equ nGm = 13 ; G, f = 97.9989 Hz
.equ nAm = 14 ; A, f = 110 Hz
.equ nHm = 15 ; H, f = 123.471 Hz
.equ nc = 16 ; c, f = 130.813 Hz
.equ nd = 17 ; d, f = 146.832 Hz
.equ ne = 18 ; e, f = 164.814 Hz
.equ nf = 19 ; f, f = 174.614 Hz
.equ ng = 20 ; g, f = 195.998 Hz
.equ na = 21 ; a, f = 220 Hz
.equ nh = 22 ; h, f = 246.942 Hz
.equ nc1 = 23 ; c1, f = 261.626 Hz
.equ nd1 = 24 ; d1, f = 293.665 Hz
.equ ne1 = 25 ; e1, f = 329.628 Hz
.equ nf1 = 26 ; f1, f = 349.228 Hz
.equ ng1 = 27 ; g1, f = 391.995 Hz
.equ na1 = 28 ; a1, f = 440 Hz
.equ nh1 = 29 ; h1, f = 493.883 Hz
.equ nc2 = 30 ; c2, f = 523.251 Hz
.equ nd2 = 31 ; d2, f = 587.33 Hz
.equ ne2 = 32 ; e2, f = 659.255 Hz
.equ nf2 = 33 ; f2, f = 698.456 Hz
.equ ng2 = 34 ; g2, f = 783.991 Hz
.equ na2 = 35 ; a2, f = 880 Hz
.equ nh2 = 36 ; h2, f = 987.767 Hz
.equ nc3 = 37 ; c3, f = 1046.5 Hz
.equ nd3 = 38 ; d3, f = 1174.66 Hz
.equ ne3 = 39 ; e3, f = 1318.51 Hz
.equ nf3 = 40 ; f3, f = 1396.91 Hz
.equ ng3 = 41 ; g3, f = 1567.98 Hz
.equ na3 = 42 ; a3, f = 1760 Hz
.equ nh3 = 43 ; h3, f = 1975.53 Hz
.equ nc4 = 44 ; c4, f = 2093 Hz
.equ nd4 = 45 ; d4, f = 2349.32 Hz
.equ ne4 = 46 ; e4, f = 2637.02 Hz
.equ nf4 = 47 ; f4, f = 2793.83 Hz
.equ ng4 = 48 ; g4, f = 3135.96 Hz
.equ na4 = 49 ; a4, f = 3520 Hz
.equ nh4 = 50 ; h4, f = 3951.07 Hz

.equ nc5 = 51 ; c5, f = 4186.01 Hz
.equ nd5 = 52 ; d5, f = 4698.65 Hz
.equ ne5 = 53 ; e5, f = 5274.05 Hz
.equ nf5 = 54 ; f5, f = 5587.67 Hz
.equ ng5 = 55 ; g5, f = 6271.93 Hz
.equ na5 = 56 ; a5, f = 7040 Hz
.equ nh5 = 57 ; h5, f = 7902.13 Hz
.equ nc6 = 58 ; c6, f = 8372.02 Hz
.equ nd6 = 59 ; d6, f = 9397.28 Hz
.equ ne6 = 60 ; e6, f = 10548.08 Hz
.equ nf6 = 61 ; f6, f = 11175.3 Hz
.equ ng6 = 62 ; g6, f = 12543.86 Hz
.equ na6 = 63 ; a6, f = 14080 Hz
.equ nh6 = 64 ; h6, f = 15804.26 Hz
.equ nc7 = 65 ; c7, f = 16744.03 Hz
.equ nd7 = 66 ; d7, f = 18794.56 Hz
.equ ne7 = 67 ; e7, f = 21096.16 Hz
.equ nf7 = 68 ; f7, f = 22350.59 Hz
.equ ng7 = 69 ; g7, f = 25087.71 Hz
; Pause
.equ p = 70 ; Pause, speaker off, #70
; End
.equ MelodyEnd = 0xFF
;
; Melodies
;
Melodies that the clock can play randomly
;
Maximum 16 melodies (0..15)
;
.equ cMaxMelody = 8 ; Number of melodies
;
; Melody table
;
Table entries point to begin of melodies
;
in flash * 2
;
MelodyTab:
.dw 2*Melody0
.if cMaxMelody>=1
.dw 2*Melody1
.endif
.if cMaxMelody>=2
.dw 2*Melody2
.endif
.if cMaxMelody>=3
.dw 2*Melody3
.endif
.if cMaxMelody>=4
.dw 2*Melody4
.endif
.if cMaxMelody>=5
.dw 2*Melody5
.endif
.if cMaxMelody>=6
.dw 2*Melody6
.endif
.if cMaxMelody>=7
.dw 2*Melody7
.endif
.if cMaxMelody>=8
.dw 2*Melody8
.endif
.if cMaxMelody>=9
.dw 2*Melody9
.endif
.if cMaxMelody>=10

.dw 2*Melody10
.endif
.if cMaxMelody>=11
.dw 2*Melody11
.endif
.if cMaxMelody>=12
.dw 2*Melody12
.endif
.if cMaxMelody>=13
.dw 2*Melody13
.endif
.if cMaxMelody>=14
.dw 2*Melody14
.endif
.if cMaxMelody>=15
.dw 2*Melody15
.endif
.if cMaxMelody>=16
.error "Too many melodies"
.endif
;
; Duration dividers for tones and pauses:
.equ d1
= 0
; 16/16
.equ d2
= 1
; 8/16
.equ d4
= 2
; 4/16
.equ d8
= 3
; 2/16
.equ d16 = 4
; 1/16
.equ d32 = 5
; 1/32
.equ d316 = 0x34 ; 3/16
.equ d38 = 0x33 ; 6/16
;
; Melodies
;
Each melody encodes
;
- notes to be played (nx) and one duration byte
;
- pauses (p) to be made and one duration byte
;
- an end signature (MelodyEnd)
;
Melody0: ; Ode an die Freude
;
Freude
schoener
.db nh,d4,p,d8,nh,d4,p,d8,nc1,d4,p,d8,nd1,d4,p,d8
;
Goetterfunken
.db nd1,d4,p,d8,nc1,d4,p,d8,nh,d4,p,d8,na,d4,p,d8
;
Tochter
aus
E.db ng,d4,p,d8,ng,d4,p,d8,na,d4,p,d8,nh,d4,p,d8
;
lysium,
.db nh,d4,p,d8,na,d8,p,d8,na,d4,p,d1
.db MelodyEnd,MelodyEnd
;
Melody1: ; Marmor, Stein und Eisen bricht
;
Weine
nicht,
wenn
der
.db na,d8,p,d8,na,d4,p,d8,na,d4,p,d8,ng,d8,p,d8,na,d4,p,d8
;
Regen
faellt,
.db nc1,d8,p,d8,na,d4,p,d8,na,d38,p,d4
;
dam,
dam,
dam,
dam
.db na,d4,p,d2,nh,d4,p,d2,nd1,d4,p,d8,nc1,d2,p,d2
;
Marmor,
Stein
und
.db na1,d8,p,d8,na1,d4,p,d8,na1,d38,p,d8,na1,d4,p,d8
;
Eisen
bricht,
.db ng1,d8,p,d8,nf1,d4,p,d8,nf1,d38,p,d8
;
aber
unsere
.db ng1,d8,p,d8,ng1,d4,p,d8,ng1,d4,p,d8,nf1,d8,p,d8,ng1,d4,p,d8
;
Liebe
nicht.
.db na1,d4,ne1,d4,nf1,d4,p,d8,ne1,d38,p,d2
;
Alles,
alles
geht

.db
;
.db
;
.db
.db

na1,d4,p,d16,na1,d4,p,d4,na1,d4,p,d16,na1,d4,p,d4,ng1,d8,p,d8
vorbei,
doch
wir
nf1,d4,p,d8,nf1,d38,p,6,ng1,d4,p,d8,ng1,d4,p,d8
sind
uns
treu
nf1,d8,p,d8,ng1,d4,p,d8,na1,d2
MelodyEnd,MelodyEnd

;
Melody2: ; When I'm sixty four
;
When
I
get
older
.db nc,d4,p,d16,nhm,d4,p,d16,nc,d4,p,d16,ne,d4,p,d8,ne,d4,p,d8
;
loosing
my
hair...
.db ne,d4,p,d16,nf,d4,p,d16,ne,d4,p,d16,na,d2,p,d8
;
many
years...
from
now,
.db na,d8,p,d8,nc1,d4,p,d8,na,d38,p,d8,nf,d4,p,d8,nh,d2,p,d2
;
will
you
still
be
send.db ng,d4,p,d16,na,d4,p,d16,na,d4,p,d16,nc1,d4,p,d8,ng,d4,p,d16
;
ing
me
a
valentine...
.db na,d4,p,d16,na,d4,p,d16,nc1,d4,p,d8,ng,d8,p,d8,ng,d4,p,d8,nf,d38,p,d8
;
birthday
greetings,
.db ng,d4,p,d8,na,d4,p,d8,na,d4,p,d8,nh,d4,p,d8
;
bottle
of
wine?
.db nc1,d4,p,d16,nc1,d4,p,d16,nh,d4,p,d16,na,d38,p,d2
;
If
I'd
been
out...
.db nc,d4,p,d16,nhm,d4,p,d16,nc,d4,p,d16,ne,d4,p,d8
;
till
quarter
to
three...
.db nc,d4,p,d8,ne,d4,p,d16,nf,d4,p,d16,ne,d4,p,d16,na,d8,na,d4,p,d2
;
would
you
lock
the
door?
.db nc1,d8,p,d8,nc1,d316,p,d8,nh,d4,p,d8,nf,d4,p,d16,na,d2,p,d2
;
will
you
still
need
me,
.db na,d4,p,d16,nf,d4,p,d16,na,d4,p,d16,nc1,d38,p,d8,nh,d4,p,d8
;
will
you
still
feed
me
.db na,d4,p,d16,nf,d4,p,d16,na,d4,p,d16,ng,d38,p,d16,nf,d4,p,d8
;
when
I'm
sixty
four?
.db nc1,d8,p,d8,nc1,d316,p,d8,nc1,d8,p,d8,nc1,d4,p,d8,na,d2
.db MelodyEnd,MelodyEnd
;
Melody3: ; Smoke on the water
;
Smoke
on
the
water
.db ne1,d2,p,d8,nd1,d4,p,d16,nc1,d4,p,d8,ne1,d4,p,d8,nc1,d4,p,d2
;
A
Fire
in
the
sky
.db ng,d8,p,d8,nh,d4,p,d16,ng,d4,p,d8,nh,d4,p,d16,ng,d4,p,d8,nf,d4,ng,d2
.db MelodyEnd,MelodyEnd
;
Melody4: ; The wall
;
We
don't
need
no
.db nd,d8,p,d8,ne,d4,p,d8,nf,d4,p,d8,ne,d4,p,d8
;
education
.db nd,d4,p,d8,ne,d4,p,d16,nf,d38,p,d8,ne,d4,p,d2
;
We
don't
need
no
.db nd,d8,p,d8,ne,d4,p,d8,nf,d4,p,d8,ne,d4,p,d8
;
thought
control,
.db nd,d4,p,d8,ne,d4,p,d8
;
dark
sarcasm
.db nd,d4,p,d8,ne,d4,p,d16,nf,d4,p,d16,ne,d4,p,d4
;
in
the
classroom
.db nd,d4,p,d8,ne,d38,p,d8,nf,d4,p,d8,ne,d4,p,d2
;
teach
or
leave
them
.db nd,d4,p,d8,ne,d4,p,d16,nf,d38,p,d8,ne,d4,p,d4
;
kids
alone
.db nd,d4,p,d8,ne,d4,p,d16,nf,d2
.db MelodyEnd,MelodyEnd
;
Melody5: ; Emanuela
;
Lass
die
Fin-

.db
;
.db
;
.db
.db

nf,d4,p,d4,nc1,d4,p,d4,nc1,d4,p,d16
ger
von
Enc1,d4,p,d16,na,d4,p,d16,nf,d38,p,d4
manuela
nc1,d4,p,d4,nd1,d8,p,d4,nc1,d4,p,d4,na,d4
MelodyEnd,MelodyEnd

;
Melody6: ; Morse, speed = 80 er minute
; G
e r
h
a r
d
S
c
h
m i d
t
; --. . .-. .... .- .-. -.. ... -.-. .... -- .. -.. ; '.' = 1/16 = d16, '-' = 3/16 = d316, p = 1/16 = d16, ' ' = 3/16 = d316
;
.
.db nc3,d316,p,d16,nc3,d316,p,d16,nc3,d16,p,d316 ; G
;
.
.db nc3,d16,p,d316 ; e
;
.
.
.db nc3,d16,p,d16,nc3,d316,p,d16,nc3,d16,p,d316 ; r
;
.
.
.
.
.db nc3,d16,p,d16,nc3,d16,p,d16,nc3,d16,p,d16,nc3,d16,p,d316 ; h
;
.
.db nc3,d16,p,d16,nc3,d316,p,d316 ; a
;
.
.
.db nc3,d16,p,d16,nc3,d316,p,d16,nc3,d16,p,d316 ; r
;
.
.
.db nc3,d316,p,d16,nc3,d16,p,d16,nc3,d16,p,d1 ; d
;
.
.
.
.db nc3,d16,p,d16,nc3,d16,p,d16,nc3,d16,p,d316 ; S
;
.
.
.db nc3,d316,p,d16,nc3,d16,p,d16,nc3,d316,p,d16,nc3,d16,p,d316 ; c
;
.
.
.
.
.db nc3,d16,p,d16,nc3,d16,p,d16,nc3,d16,p,d16,nc3,d16,p,d316 ; h
;
.db nc3,d316,p,d16,nc3,d316,p,d316 ; m
;
.
.
.db nc3,d16,p,d16,nc3,d16,p,d316 ; i
;
.
.
.db nc3,d316,p,d16,nc3,d16,p,d16,nc3,d16,p,d316 ; d
;
.db nc3,d316,p,d316 ; t
;
.
.
.
.db nc3,d16,p,d16,nc3,d16,p,d16,nc3,d16,p,d16
;
.
.db nc3,d316,p,d16,nc3,d16,p,d16,nc3,d316,p,d1
.db MelodyEnd,MelodyEnd
;
Melody7:
;
Voelker
hoert
.db ne1,d4,ne1,d8,nd1,d4,p,d8,nc1,d2,p,d8
;
die
Signale!
.db ng,d38,p,d8,ne,d8,p,d8,na,d2,p,d8,nf,d4,p,d4
;
Auf
zum
letzt.db nd1,d38,p,d8,nc1,d4,p,d8,nh,d2,p,d8
;
ten
Gefecht!
.db na,d38,p,d8,ng,d8,p,d8,ng,d2
;
Die
Inter.db ng,d4,p,d8,ne1,d4,p,d4,ne1,d8,p,d8
;
nationale
.db ne1,d8,p,d8,nd1,d4,p,d8,ng,d4,p,d8,nc1,d2,p,d8
;
erkaempft
das
.db nh,d8,p,d8,na,d38,p,d8,ng,d8,p,d8
;
Menschenrecht!
.db na,d4,p,d8,nd1,d4,p,d8,nd1,d2
.db MelodyEnd,MelodyEnd
;

Melody8: ; The whole gamut
.db nA2m,d2,p,d16,nH2m,d2,p,d4
.db nC1m,d2,p,d16,nD1m,d2,p,d16,nE1m,d2,p,d16,nF1m,d2,p,d16
.db nG1m,d2,p,d16,nA1m,d2,p,d16,nH1m,d2,p,d4
.db nCm,d2,p,d16,nDm,d2,p,d16,nEm,d2,p,d16,nFm,d2,p,d16
.db nGm,d2,p,d16,nAm,d2,p,d16,nHm,d2,p,d4
.db nc,d2,p,d16,nd,d2,p,d16,ne,d2,p,d16,nf,d2,p,d16
.db ng,d2,p,d16,na,d2,p,d16,nh,d2,p,d4
.db nc1,d2,p,d16,nd1,d2,p,d16,ne1,d2,p,d16,nf1,d2,p,d16
.db ng1,d2,p,d16,na1,d2,p,d16,nh1,d2,p,d4
.db nc2,d2,p,d16,nd2,d2,p,d16,ne2,d2,p,d16,nf2,d2,p,d16
.db ng2,d2,p,d16,na2,d2,p,d16,nh2,d2,p,d4
.db nc3,d2,p,d16,nd3,d2,p,d16,ne3,d2,p,d16,nf3,d2,p,d16
.db ng3,d2,p,d16,na3,d2,p,d16,nh3,d2,p,d4
.db nc4,d2,p,d16,nd4,d2,p,d16,ne4,d2,p,d16,nf4,d2,p,d16
.db ng4,d2,p,d16,na4,d2,p,d16,nh4,d2,p,d4
.db nc5,d2,p,d16,nd5,d2,p,d16,ne5,d2,p,d16,nf5,d2,p,d16
.db ng5,d2,p,d16,na5,d2,p,d16,nh5,d2,p,d4
.db nc6,d2,p,d16,nd6,d2,p,d16,ne6,d2,p,d16,nf6,d2,p,d16
.db ng6,d2,p,d16,na6,d2,p,d16,nh6,d2,p,d4
.db nc7,d2,p,d16,nd7,d2
.db MelodyEnd,MelodyEnd
;
; Note: Update cMaxMelody after adding/removing melodies
;
; End of include file
;

4 The routines for debugging debug_code_en.inc
Link to this code in asm format
;
; *********************************
; * Debug codes for dcf77_m16_v4 *
; * (C)2018 by Gerhard Schmidt
*
; *********************************
;
; Debug the conversion of DCF signals to date/time
; Set received DCF signal to Mo, 31.12.99-23:59
; DCF bytes 7:0 = 4C C8 E3 8D 64 00 00 00
DebugDcfBitConversion:
ldi rmp,0x4C
sts sDcfBits+7,rmp
ldi rmp,0xC8
sts sDcfBits+6,rmp
ldi rmp,0xE3
sts sDcfBits+5,rmp
ldi rmp,0x8D
sts sDcfBits+4,rmp
ldi rmp,0x64
sts sDcfBits+3,rmp
ldi rmp,0x00
sts sDcfBits+2,rmp
sts sDcfBits+1,rmp
sts sDcfBits,rmp
mov rDcfErr,rmp
ldi rmp,58

mov rDcfBitCnt,rmp
call Dcf59Ok ; Debug the conversion routine
DbgMinuteLoop: ; Stop further execution
rjmp DbgMinuteLoop
;
; Debug the ADC by displaying the raw results
;
from the selected channel
DebugAdc:
ldi rmp,debugAdcChannel|(1<<REFS0) ; Select channel
out ADMUX,rmp
ldi ZH,3
ldi ZL,0
rcall LcdPos
ldi rmp,'A'
rcall LcdChar
ldi rmp,'D'
rcall LcdChar
ldi rmp,'C'
rcall LcdChar
ldi rmp,debugAdcChannel+48
rcall LcdChar
ldi rmp,':'
rcall LcdChar
LoopAdc:
ldi rmp,(1<<ADEN)|(1<<ADSC)|(1<<ADIE)|(1<<ADPS2)|(1<<ADPS1)|(ADPS0)
out ADCSRA,rmp ; Start first conversion
WaitAdc:
sbic ADCSRA,ADSC
rjmp WaitAdc
ldi ZH,3
ldi ZL,5
rcall LcdPos
in ZL,ADCL
in ZH,ADCH
rcall LcdHex4
rjmp LoopAdc
;
; Debug the keys attached by displaying
; their current state over and over
DebugKey:
ldi ZH,3
ldi ZL,0
rcall LcdPos
in ZL,pKeyI
ldi rmp,'0'
sbrc ZL,bKeyRI
ldi rmp,'1'
rcall LcdChar
ldi rmp,'0'
sbrc ZL,bKeyYI
ldi rmp,'1'
rcall LcdChar
ldi rmp,'0'
sbrc ZL,bKeyWI
ldi rmp,'1'
rcall LcdChar
rjmp DebugKey
;
; Speaker to 1.000 Hz
DebugSpeaker:
sbi pSpkD,bSpkD ; Output pin active
ldi rmp,High(clock/1000/2-1)
out OCR1AH,rmp
ldi rmp,Low(clock/1000/2-1)

out OCR1AL,rmp
ldi rmp,1<<COM1A0 ; Toggle COM1A
out TCCR1A,rmp
ldi rmp,(1<<WGM12)|(1<<CS10) ; CTC, prescaler = 1
out TCCR1B,rmp
DebugSpeaker1:
rjmp DebugSpeaker1
;
; Init the debugging of DCF signal duration analysis
DebugDcfDur:
ldi ZH,3
ldi ZL,12
call LcdPos
ldi ZH,High(2*DebugDcfText)
ldi ZL,Low(2*DebugDcfText)
jmp LcdText
;
; Display DCF signal duration
DebugDcfDurDispl:
ldi ZH,3
ldi ZL,0
sbic pDcfI,bDcfI
ldi ZL,6
rcall LcdPos
mov R0,rDcfL ; Copy pulses
mov R1,rDcfH
lsl R0 ; Multiply by 5
rol R1
lsl R0
rol R1
add R0,rDcfL
adc R1,rDcfH
mov ZH,R1
mov ZL,R0
jmp LcdDec5 ; Display pulse duration in ms on LCD
;
; Echo DCF bits
DebugDcfBitEcho:
ldi ZH,3
ldi ZL,0
rcall LcdPos
mov rmp,rDcfBitCnt ; Display bit number
rcall LcdHex2
ldi rmp,':'
rcall LcdChar
ldi rmp,' '
rcall LcdChar
lds ZL,sDcfBits+7 ; Load newest byte
ldi ZH,8
Dcf01Ok1:
ldi rmp,'0' ; Display zero
lsl ZL
brcc Dcf01Ok2 ; Display one
ldi rmp,'1' ; Display one
Dcf01Ok2:
rcall LcdChar
dec ZH
brne Dcf01Ok1
ret
;
; Debug the DCF signal display the last 8 bits
DebugDcfSig1:
ldi ZH,3 ; Long version of report on line 4
ldi ZL,0

rcall LcdPos
mov R0,rDcfBitCnt ; Display number of bits
rcall LcdDec2
ldi ZH,High(2*LcdErr6) ; Add number of bit text
ldi ZL,Low(2*LcdErr6)
jmp LcdText
;
; Debug the DCF signal with long error messages
DebugDcfSig2:
mov XH,ZH ; Save Z position in flash
mov XL,ZL
ldi ZH,3
ldi ZL,0
rcall LcdPos
mov ZH,XH ; Restore report text to Z
mov ZL,XL
rjmp LcdText
;
; Display the key state in line 3
KeyState:
ldi ZH,3
ldi ZL,0
rcall LcdPos
ldi rmp,'0'
sbic pKeyI,bKeyRI
ldi rmp,'1'
rcall LcdChar
ldi rmp,'0'
sbic pKeyI,bKeyYI
ldi rmp,'1'
rcall LcdChar
ldi rmp,'0'
sbic pKeyI,bKeyWI
ldi rmp,'1'
rcall LcdChar
ldi rmp,' '
rcall LcdChar
ldi rmp,'f'
rcall LcdChar
ldi rmp,'0'
sbrc rFlag,bKeyAct
ldi rmp,'1'
rcall LcdChar
ldi rmp,' '
rcall LcdChar
ldi rmp,'C'
rcall LcdChar
mov rmp,rKeyCnt
rcall LcdHex2
ldi rmp,' '
rcall LcdChar
ldi rmp,'m'
rcall LcdChar
lds rmp,sKeyMode
subi rmp,-'0'
rcall LcdChar
ldi rmp,' '
rcall LcdChar
ldi rmp,'P'
rcall LcdChar
lds rmp,sKeyPos
subi rmp,-'0'
rcall LcdChar
ret

;
; Set debugging active display, message on line 3
;
DebugActive:
ldi ZH,2
ldi ZL,0
rcall LcdPos
ldi ZH,High(2*DebugActiveText)
ldi ZL,Low(2*DebugActiveText)
rjmp LcdText
;
DebugActiveText:
.db "Debugging activated!",0xFE,0xFE
;
DebugDcfText:
.db "ms hi/lo ",0xFE
;
; Debug the received DCF77 information
;
DebugDcfRcvd:
ldi ZH,3
ldi ZL,0
rcall LcdPos
ldi XH,High(sDcfRcv)
ldi XL,Low(sDcfRcv)
ldi rmp,6
DebugDcfRcvd1:
push rmp
ld rmp,X+
rcall LcdHex2
ldi rmp,' '
rcall LcdChar
pop rmp
dec rmp
brne DebugDcfRcvd1
ret
;
; Test blinking of the LCD
Blinktest:
clr XL
rcall LcdBlink
Blinktest1:
ldi rmp,40
DelayA:
rcall LcdWait50ms
dec rmp
brne DelayA
ldi ZH,High(2*LcdTxtInit1)
ldi ZL,Low(2*LcdTxtInit1)
rcall LcdText
ldi ZH,0
inc XL
mov ZL,XL
rcall LcdPos
rjmp Blinktest1
;
; Debug the music play
DebugMusic:
ldi ZH,3 ; Last line
clr ZL
rcall LcdPos
ldi rmp,'a'
rcall LcdChar
mov ZH,YH

mov ZL,YL
rcall LcdHex4
ldi rmp,' '
rcall LcdChar
ldi rmp,'c'
rcall LcdChar
in ZL,OCR1AL
in ZH,OCR1AH
rcall LcdHex4
ldi rmp,' '
rcall LcdChar
ldi rmp,'d'
rcall LcdChar
mov ZH,rDurH
mov ZL,rDurL
rjmp LcdHex4
;
; Debug the status of TC1
DebugPlayStat:
ldi ZH,3
clr ZL
rcall LcdPos
in ZL,TIMSK
ldi rmp,'i'
sbrc ZL,OCIE1A
ldi rmp,'I'
rcall LcdChar
in ZL,TCCR1A
ldi rmp,'t'
sbrc ZL,COM1A0
ldi rmp,'T'
sbrc ZL,COM1A1
ldi rmp,'C'
rcall LcdChar
ldi rmp,' '
rcall LcdChar
ldi rmp,'S' ; Display stackpointer
rcall LcdChar
ldi rmp,'t'
rcall LcdChar
cli
in R0,SPL
in R1,SPH
sei
rjmp LcdHex4
;
; For date/time debugging
DebugDateTime:
ldi ZH,3
ldi ZL,0
rcall LcdPos
ldi XH,High(sDateTime)
ldi XL,Low(sDateTime)
ldi ZL,7
DebugDateTime1:
ld rmp,X+
rcall LcdHex2
cpi ZL,1
breq DebugDateTime2
ldi rmp,' '
rcall LcdChar
dec ZL
rjmp DebugDateTime1
DebugDateTime2:

ret
;
; End of include file
;

